Centro Federal de Educação, Ciência e Tecnologia
Campus Central - Natal - Rio Grande do Norte
Diretoria de Educação e Tecnologia da Informação

Curso Tecnólogo em Análise e Desenvolvimento de Sistemas
Curso Tecnólogo em Redes de Computadores

Curso de Cálculo Diferencial e Integral

Prof.: Fco. **Assis** de Oliveira

I N T E G R A I S

Neste curso, estudaremos o processo de integração. Abordaremos, inicialmente, os conceitos relacionados com os cálculos de áreas de uma região delimitada entre curvas de funções e os eixos-coordenados. A seguir, procuraremos mostrar a definição de integrais indefinidas, vistas como uma antiderivação. Em seguida, estudaremos as integrais definidas e as propriedades da integração. Mostraremos os métodos de integração. Desenvolveremos implementações da integração no *software* de computação simbólica **Maple**. O sistema principal do **Maple** está preparado para trabalhar com as operações básicas do cálculo, porém devemos carregar para a memória do computador o pacote específico para este fim, ou seja, a biblioteca *student*, o que deve ser executado com os comandos **with(student)** e/ou **with(Student)** [Ver o *help* do Maple].

Introdução

Área sob curvas

No nosso cotidiano, necessitamos calcular áreas de figuras planas. Quase sempre são as áreas de figuras geométricas mais tradicionais, tais como: triângulos, retângulos, quadrados e círculos. Desde os tempos mais antigos, os matemáticos se preocupam com o problema de encontrar tais áreas. Uma das técnicas mais utilizadas é a de aproximação da área de uma figura desconhecida por áreas de figuras conhecidas mais simples. Arquimedes desenvolveu o método da "exaustão" para calcular e aproximar a área de um círculo através de polígonos regulares, aumentando-se o número de lados destes, aproximando-se o máximo possível de um círculo. Assim sendo, fazendo o número do lado do polígono n crescer cada vez mais, ou seja, $n \to \infty$, então a área do polígono tende ao limite da área do círculo.

Podemos proceder de forma idêntica para encontrar áreas de curvas definidas por funções, em um sistema de eixos cartesiano. Desta forma, para calcular-se estas áreas, devemos dividir a região desejada em polígonos cujas áreas possam ser calculadas pelos métodos da geometria elementar.

Vamos considerar o problema de calcular a área de uma região plana delimitada pelo curva de uma função contínua e não-negativa, definida em um intervalo [a ; b], o eixo-x, e as retas $x = a$ e $x = b$, conforme podemos observar na figura a seguir.



Uma abordagem ao problema da área sob esta curva é a utilização do método de exaustão de Arquimedes, considerando:

I - dividimos o intervalo [*a* ; *b*] em *n* subintervalos iguais e em cada um deles construir um retângulo que estende-se do eixo-*x* até interceptar a curva *y* = f(*x*);

II - a soma da área de todos os retângulos pode ser vista como uma aproximação da área sob a curva; à medida que o número *n* de retângulos cresce, essas aproximações ficarão cada vez mais próximas e tenderão à área exata da curva, como um limite.

Desta forma, se A_n representar a área de cada intervalo, então podemos escrever

$$Area = \lim_{n \to \infty} A_n$$

O **Maple** possui os procedimentos *leftbox*, *middlebox* e *rightbox*, os quais fazem parte do "pacote" *student*, que executam a aproximação da curva de uma função em *n* subintervalos [Ver *Help* do **Maple** para os comandos acima], mostrando esta aproximação graficamente, enquanto que os comandos *leftsum*, *middlesum* e *rightsum* avaliam esta aproximação numericamente.

Vamos considerar o gráfico acima de uma função e subdividi-lo em 5 intervalos, conforme mostram as figuras seguintes:





4
3
2
1
0
0.2
0.4
0.6
0.8
1
1.2
1.4
1.6
1.8
2
x
Retângulo ponto Médio



4
3
2
1
0
0.2
0.4
0.6
0.8
1
1.2
1.4
1.6
1.8
2
x
Retângulo à Direita

Observamos que os retângulos são formados a partir do eixo-*x*, com a base inferior neste eixo e a base superior formada em um ponto sobre a curva. A base superior do retângulo pode tocar o ponto à esquerda, no meio ou à direita do subintervalo. Dependendo deste ponto sobre a curva, a aproximação da área será uma aproximação de soma superior, soma inferior ou uma soma aproximada média.

Supondo o extremo esquerdo do intervalo o ponto $x = a$ e o extremo direito o ponto $x = b$, e n o número de subintervalos (retângulos), então a largura de cada subintervalo, se todos forem iguais, será calculado por:

$$\Delta_x = \frac{b - a}{n}$$

Para cada ponto do subintervalo teremos um ponto Δ_{x_i} , onde $i = 1 .. n$. Desta forma, a área de cada retângulo será calculado por:

$$A_i = \Delta_{x_i} \mathrm{f}(x_i)$$

E, por conseguinte, a área total será a soma de todas as áreas A_i. Ou seja:

$$Area = \sum_{i=1}^{n} \Delta_{x_i} \mathrm{f}(x_i)$$

Esta soma é chamada *Soma de Riemann* da função $\mathrm{f}(x)$.

A notação **sigma** $\sum_{i=1}^{n} A_i$ permite expressar a soma acima com muitos termos em uma forma compacta, onde:

Σ ==> símbolo de somatório
$i = 1$ ==> indica que o índice i começa com 1
n ==> indica que o índice i termina em n
A_i ==> expressão para o i-ésimo termo

Exemplo:

Abaixo, vamos calcular a área sob a curva da função $\mathrm{f}(x) = -x^2 + 1$, no intervalo [0 ; 1], desenvolvendo-a, analiticamente, e posteriormente, utilizando os comandos **Maple** acima mencionados.

Inicialmente, vamos considerar, para efeitos de aproximação, uma quantidade de 5 retângulos na subdivisão do intervalo. Assim sendo, temos:

==> $a = 0$;

==> $b = 1$;

==> $n = 5$;

==> $\Delta_x = \dfrac{1-0}{5} = 0,2$;

Considerando que os retângulos tocam a curva no ponto à esquerda da base superior do retângulo, temos:

==> $x_1 = 0$;

==> $x_2 = (0,2)$;

==> $x_3 = (0,4)$;

==> $x_4 = (0,6)$;

==> $x_5 = (0,8)$;

Encontrando os pontos f(x_i):

==> f(x_1) = f(0.0) = 1;

==> f(x_2) = f(0.2) = 0,96;

==> f(x_3) = f(0.4) = 0,84;

==> f(x_4) = f(0.6) = 0,64;

==> f(x_5) = f(0.8) = 0,36;

Encontrando as áreas de cada retângulo:

==> A_1 = 0,2 x 1,00 = 0,200;

==> A_2 = 0,2 x 0,96 = 0,192;

==> A_3 = 0,2 x 0,84 = 0,168;

==> A_4 = 0,2 x 0,64 = 0,128;

==> A_5 = 0,2 x 0,36 = 0,072

Portanto, a área total será:

==> $A = \sum_{i=1}^{5} A_i = A_1 + A_2 + A_3 + A_4 + A_5$

==> A = 0,200 + 0,192 + 0,168 + 0,128 + 0,072 = 0,760

Desenvolvimento com comandos **Maple**:

```
> restart: Digits:=5:
> with(student): with(Student[Calculus1]):
```

```
> f:=x->-x^2+1;
```

$$f := x \rightarrow -x^2 + 1$$

Definindo o intervalo e o número de retângulos:

```
> a:=0: b:=1: n:=5:
```

Criando os retângulos sob a curva $f(x)$:

```
> leftbox(f(x),x=a..b,n,color=blue,shading=turquoise);
```



O comando *leftsum* calcula (avalia) a soma das áreas dos retângulos, traçadas pelo procedimento *leftbox*.

```
> Area:=leftsum(f(x),x=a..b,n);
```

$$Area := \frac{1}{5}\left(\sum_{i=0}^{4}\left(-\frac{i^2}{25}+1\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 0.76000$$

A seguir, vamos aumentar o número de retângulos para 20 unidades:

```
> n:=20:
> leftbox(f(x),x=a..b,n,color=blue,shading=turquoise);
```



O comando *leftsum* calcula (avalia) a soma das áreas dos retângulos traçadas pelo procedimento *leftbox*.

```
> Area:=leftsum(f(x),x=a..b,n);
```

$$Area := \frac{1}{20}\left(\sum_{i=0}^{19}\left(-\frac{i^2}{400}+1\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 0.69125$$

Criando os retângulos sob a curva $f(x)$ e avaliando a área, usando o procedimento *middlebox* e *middlesum*:

```
> n:=5:
```

```
> middlebox(f(x),x=a..b,n,color=blue,shading=turquoise);
```




```
> Area:=middlesum(f(x),x=a..b,n);
```

$$Area := \frac{1}{5}\left(\sum_{i=0}^{4}\left(-\left(\frac{i}{5}+\frac{1}{10}\right)^2+1\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 0.67000$$

A seguir, vamos aumentar o número de retângulos para 20 unidades:

```
> n:=20:
> middlebox(f(x),x=a..b,n,color=blue,shading=turquoise);
```




```
> Area:=middlesum(f(x),x=a..b,n);
```

$$Area := \frac{1}{20}\left(\sum_{i=0}^{19}\left(-\left(\frac{i}{20}+\frac{1}{40}\right)^2+1\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 0.66690$$

Criando os retângulos sob a curva f(x) e avaliando a área, usando o procedimento *rightbox* e right*sum*:

```
> n:=5:
> rightbox(f(x),x=a..b,n,color=blue,shading=turquoise);
```




```
> Area:=rightsum(f(x),x=a..b,n);
```

$$Area := \frac{1}{5}\left(\sum_{i=1}^{5}\left(-\frac{i^2}{25}+1\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 0.56000$$

A seguir, vamos aumentar o número de retângulos para 20 unidades:

```
> n:=20:
> rightbox(f(x),x=a..b,n,color=blue,shading=turquoise);
```




```
> Area:=rightsum(f(x),x=a..b,n);
```

$$Area := \frac{1}{20}\left(\sum_{i=1}^{20}\left(-\frac{i^2}{400}+1\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 0.64125$$

O valor exato da área é calculado (avaliado) pelo **Maple**, usando a integral definida a seguir, o qual será discutida posteriormente.

```
> Area_Exata:=int(f(x),x=a..b);
```

$$Area\_Exata := \frac{2}{3}$$

```
> 'Area_Exata'=evalf(Area_Exata);
```

$$Area\_Exata = 0.66667$$

O pactore *Student[Calculus1]* possui o comando *RiemannSum*, cuja sintaxe resumida é mostrada abaixo:

Somas de Riemann - sintaxe do comando

method = *left*, *lower*, ***midpoint***, *random*, *right*, *upper*, or *procedure*

outline = *true* or ***false***

output = ***value***, *sum*, *plot*, or *animation*

partition = *posint*, *list(algebraic)*, *random[algebraic]*, or *algebraic*. Por default, o intervalo é dividido em 10 subintervalos iguais

pointoptions = *list*

showarea = ***true*** or *false*

showfunction = ***true*** or *false*

showpoints = ***true*** or *false*

A seguir, mostraremos o exemplo anterior avaliado pelo procedimento *RiemannSum*, usando 10 subintervalos.

```
> n:=10:
> RiemannSum(f(x),x=a..b,method=left,output=plot,partition=n
  ,color=blue);
```




```
> RiemannSum(f(x),x=a..b,method=midpoint,output=plot,partiti
  on=n,color=blue);
```




```
> RiemannSum(f(x),x=a..b,method=right,output=plot,partition=
  n,color=blue);
```



Exemplo:
Encontre a área da região limitada pelas curvas dadas, utilizando-se das *Somas de Riemann*:

$\mathrm{f}(x) = x + 3$ e $\mathrm{g}(x) - 5 - x^2$.

Solução com Maple:

```
> restart: with(plots): Digits:=5:
> with(student): with(Student[Calculus1]):
```

Definindo as funções no **Maple**:

```
> f:=x->x+3; g:=x->5-x^2;
```

$$f := x \to x + 3$$

$$g := x \to 5 - x^2$$

Encontrando o intervalo de interseção das curvas e a região delimitada entre as curvas:

```
> solve(f(x)=g(x));
```

$$1, -2$$

Esboçando os gráficos das funções:

```
> graf1:=plot([f(x),g(x)], x = -3..2, y=-1..5, thickness=2,
  color = [red,blue]):
> graf2:=plot({[-2+k/10,t,t=f(-2+k/10)..g(-2+k/10)] $
  k=1..29},thickness=1,color=tan):
> display(graf2,graf1);
```



Observando-se o gráfico anterior, temos que a área *hachurada* corresponde à área delimitada pela parábola e o eixo-*x* menos a área delimitada pela reta e o eixo-*x*, no intervalo entre as raizes, $x_1 = -2$ e $x_2 = 1$.

Definindo os extremos do intervalo e o número de retângulos:

```
> a:=-2: b:=1: n:=10:
```

Criando os retângulos sob a curva $g(x)$ - parábola - e o eixo-x:

```
> middlebox(g(x),x=a..b,n,color=blue,shading=turquoise);
```



Calculando a área sob a curva g(*x*) e o eixo-*x:*

```
> A1:=middlesum(g(x),x=a..b,n,color=blue,shading=turquoise);
```

$$A1 := \frac{3}{10}\left(\sum_{i=0}^{9}\left(5-\left(-\frac{37}{20}+\frac{3\,i}{10}\right)^2\right)\right)$$

```
> 'A1'=evalf(A1);
```

$$A1 = 12.022$$

Criando os retângulos sob a curva f(*x*) - reta - e o eixo-*x* :

```
> middlebox(f(x),x=a..b,n,color=blue,shading=turquoise);
```



Calculando a área sob a curva f(x) e o eixo-x:

```
> A2:=middlesum(f(x),x=a..b,n,color=blue,shading=turquoise);
```

$$A2 := \frac{3}{10}\left(\sum_{i=0}^{9}\left(\frac{23}{20}+\frac{3\,i}{10}\right)\right)$$

```
> 'A2'=evalf(%);
```

$$A2 = 7.5000$$

Calculando a área entre as curvas:

```
> Area:=A1-A2;
```

$$Area := \frac{3}{10}\left(\sum_{i=0}^{9}\left(5-\left(-\frac{37}{20}+\frac{3\,i}{10}\right)^{2}\right)\right)-\frac{3}{10}\left(\sum_{i=0}^{9}\left(\frac{23}{20}+\frac{3\,i}{10}\right)\right)$$

```
> 'Area'=evalf(Area);
```

$$Area = 4.5220$$

Exercícios propostos

I - Encontre a área da região limitada pelas curvas dadas, utilizando as *Somas de Riemann*:

01 - $y = 1 - x^2$ e $y = \frac{1}{3}$

02 - $y = 3 - x^2$ e $y = 3 - x$

03 - $y = \mathbf{e}^x$, $y = 0$, $x = 0$, $x = 1$

04 - $y = \mathbf{e}^{(-x)}$, $y = x + 1$, $x = -1$

05 - $y = \ln(x)$, $y = 0$, $x = 4$

06 - $y = \ln(x)$, $y = 4$, $x = 1$

07 - $y = \mathrm{sen}(x)$, $eixo - x$, $x = [0, 2\pi]$

08 - $y = \mathrm{sen}(x)$, $y = -\mathrm{sen}(x)$, $x = [0, 2\pi]$

09 - $y = \cos(x)$, $eixo - x$, $x = [0, 2\pi]$

10 - $y = \cos(x)$, $y = -\cos(x)$, $x = \left[-\frac{\pi}{2}, \frac{3\pi}{2}\right]$

11 - $y = 4 - x^2$ e $y = x^2 - 4$

12 - $y = |x - 2|$ e $y = 2 - (x - 2)^2$

A Integral Indefinida

Definição:

Uma função $\mathrm{F}(x)$ é uma **antiderivada** (ou **primitiva**) de uma função $\mathrm{f}(x)$, em um dado intervalo, se $\mathrm{F}'(x) = \mathrm{f}(x)$, para cada x pertencente ao intervalo.

Exemplo:

A função $\mathrm{F}(x) = \frac{x^3}{3}$ é a antiderivada (ou primitiva) de $\mathrm{f}(x) = x^2$, no intervalo dos reais, pois

$$F'(x) = \frac{d}{dx}\left(\frac{x^3}{3}\right) = x^2 = f(x)\ .$$

Mas, $F(x) = \frac{x^3}{3}$ não é a única antiderivada de $f(x)$, pois $F(x) = \frac{x^3}{3} + 2$ e $G(x) = \frac{x^3}{3} + 3$ também são antiderivadas de $f(x)$, pois

$$F'(x) = G'(x) = f(x)\ .$$

Teorema 01:
Seja $F(x)$ uma primitiva da função $f(x)$. Então, se C é uma constante qualquer, a função $G(x) = F(x) + C$ também é uma antiderivada de $f(x)$.

Definição:

Se $F(x)$ é uma antiderivada ou primitiva de $f(x)$, a expressão $F(x) + C$ é chamada *integral indefinida* de $f(x)$ e é denotada por

$$\int f(x)\, dx = F(x) + C$$

Ao processo de encontrarmos *antiderivadas* denominamos de **antiderivação** ou **integração**.

Observamos que, se derivarmos uma antiderivada de $f(x)$, voltamos a obter $f(x)$. Assim,

$$\frac{d}{dx}\left[\int f(x)\, dx\right] = f(x)$$

Notação e existência de Integrais Definidas

A simbologia para a integração é expressa conforme abaixo, onde:

$$\text{sinal de integração} ==> \int_a^b f(x)\, dx$$

$f(x)$ ==> é o integrando, ou seja, a função a ser integrada
dx ==> indica a variável de integração
a ==> limite inferior de integração
b ==> limite superior de integração

Propriedades da Integral Indefinida

Teorema 02:

Sejam f(x) e g(x) duas funções definidas em um intervalo pertencente a **R** e aplicadas em **R**, e k uma constante real. Então:

Teorema 02.a:

$$\int k\, \mathrm{f}(x)\, dx = k \int \mathrm{f}(x)\, dx$$

Teorema 02.b:

$$\int [\,\mathrm{f}(x) + \mathrm{g}(x)\,]\, dx = \int \mathrm{f}(x)\, dx + \int \mathrm{g}(x)\, dx$$

O processo de integração requer, essencialmente, um conhecimento de derivadas de funções, pois dada a derivada f(x) de uma certa função F(x), então busca-se encontrar esta função F(x).

Exemplo 01:

Sabemos que $\frac{d}{dx}[\,3\,x\,] = 3$. Então $\int 3\, dx = 3\,x + c$.

Sabemos que $\frac{d}{dx}[\,x^2\,] = 2\,x$. Então $\int 2\,x\, dx = x^2 + c$.

Sabemos que $\frac{d}{dx}[\,\mathrm{sen}(x)\,] = \cos(x)$. Então $\int \cos(x)\, dx = \mathrm{sen}(x) + c$.

Sabemos que $\frac{d}{dx}[\,\mathbf{e}^x\,] = \mathbf{e}^x$. Então $\int \mathbf{e}^x\, dx = \mathbf{e}^x + c$.

Sabemos que $\frac{d}{dx}[\,\ln(x)\,] = \frac{1}{x}$. Então $\int \frac{1}{x}\, dx = \ln(x) + c$.

O **Maple** utiliza o comando **Int**, em sua forma inerte, e **int** para avaliar ou calcular uma integral indefinida [ver no *Help* do **Maple** a sintaxe dos comandos].

Vamos calcular os exemplos acima, fazendo uso do comando **Maple**, conforme sequências dos comandos a seguir:

```
> restart: with(student):
> Int(3,x)=int(3,x);
```

$$\int 3\,dx = 3\,x$$

```
> Int(2*x,x)=int(2*x,x);
```

$$\int 2\,x\,dx = x^2$$

```
> Int(cos(x),x)=int(cos(x),x);
```

$$\int \cos(x)\,dx = \sin(x)$$

```
> Int(exp(x),x)=int(exp(x),x);
```

$$\int \mathbf{e}^x\,dx = \mathbf{e}^x$$

```
> Int(1/x,x)=int(1/x,x);
```

$$\int \frac{1}{x}\,dx = \ln(x)$$

N.B.: Observamos que ao avaliar a integral indefinida o **Maple** não introduz a adição da constante.

As fórmulas básicas de integração podem ser obtidas diretamente das fórmulas de derivação correspondentes. A seguir, apresentamos uma **tabela de integrais** das funções básicas mais importantes.

Tabelas de Integrais Indefinidas

01 - $\int 1\,dx = x + c$

02 - $\int x^n\,dx = \frac{x^{(n+1)}}{n+1} + c \ (n \neq -1)$

03 - $\int \mathbf{e}^x\,du = \mathbf{e}^x + c$

04 - $\int \frac{1}{x}\,dx = \ln(|x|) + c$

05 - $\int a^x\,dx = \frac{a^x}{\ln(a)} + c$

06 - $\int \operatorname{sen}(x)\,dx = -\cos(x) + c$

$$07 - \int \cos(x)\,dx = \operatorname{sen}(x) + c$$

$$08 - \int \sec(x)^2\,dx = \operatorname{tg}(x) + c$$

$$09 - \int \operatorname{cosec}(x)^2\,dx = \operatorname{cotg}(x) + c$$

$$10 - \int \operatorname{tg}(x)\,dx = -\ln(|\cos(u)|) + c = \ln(|\sec(u)| + c)$$

$$11 - \int \frac{1}{\sqrt{1-x^2}}\,dx = \operatorname{arcsen}(x) + c$$

$$12 - \int \frac{1}{1+x^2}\,dx = \operatorname{arctg}(x) + c$$

Exemplo 02: Usando as propriedades das integrais indefinidas e a tabela de integrais, calcule as integrais:

a) $\int [2 + x^3]\,dx$

Solução analítica:

$$\int [2 + x^3]\,dx = \int 2\,dx + \int x^3\,dx = 2\,x + \frac{x^4}{4}$$

Solução com Maple:

```
> Int(2+x^3,x)=int(2+x^3,x);
```

$$\int 2 + x^3\,dx = 2\,x + \frac{1}{4}x^4$$

b) $\int 3\cos(x)\,dx$

Solução analítica:

$$\int 3\cos(x)\,dx = 3\int \cos(x)\,dx + c = 3\,\mathrm{sen}(x) + c$$

Solução com Maple:

```
> Int(3*cos(x),x)=int(3*cos(x),x);
```

$$\int 3\cos(x)\,dx = 3\sin(x)$$

c) $\displaystyle\int \frac{1}{\sqrt{x}} - \mathrm{sen}(x)\,dx$

Solução analítica:

$$\int \frac{1}{\sqrt{x}} - \mathrm{sen}(x)\,dx = \int x^{\left(-\frac{1}{2}\right)}\,dx - \int \mathrm{sen}(x)\,dx + c = \frac{x^{\left(\frac{1}{2}\right)}}{\frac{1}{2}} - (-\cos(x)) + c$$

$$\int \frac{1}{\sqrt{x}} - \mathrm{sen}(x)\,dx = 2\sqrt{x} + \cos(x) + c$$

Solução com Maple:

```
> Int(1/sqrt(x)-sin(x),x)=int(1/sqrt(x)-sin(x),x);
```

$$\int \frac{1}{\sqrt{x}} - \sin(x)\,dx = 2\sqrt{x} + \cos(x)$$

Exercícios propostos

I - Encontre as integrais indefinidas e, em seguida, derive as respostas para verificar os resultados.

01 - $\displaystyle\int x^3\,dx$

02 - $\int \frac{1\,x^2}{2}\,dx$

03 - $\int 3\,x^2 - 2\,dx$

04 - $\int 3\,x^{\left(\frac{1}{2}\right)}\,dx$

05 - $\int \cos(x) + \operatorname{sen}(x)\,dx$

06 - $\int 2\,\mathbf{e}^x - \frac{1}{x}\,dx$

07 - $\int 2^x\,dx$

08 - $\int \mathbf{e}^x - \mathbf{e}^{(-x)}\,dx$

09 - $\int \frac{x^2}{1+x^2}\,dx$

10 - $\int \frac{1+x^2}{x^2}\,dx$

A Integral Definida

A integral definida está associada com os problemas de áreas, ou seja, a região delimitada sob a curva de funções e os eixos cartesiano ou entre curvas de funções.

Teorema 03:
Seja uma função f(x) contínua em um intervalo [a ; b], então f(x) é integrável em [a ; b] e a

área A resultante, computado o sinal, entre o gráfico de $f(x)$ e o intervalo $[\, a\, ;\, b\,]$ é

$$A = \int_a^b f(x)\, dx$$

Em casos mais simples, podemos calcular integrais definidas de funções contínuas usando as fórmulas da Geometria plana para calcular as áreas, considerando os sinais.

Exemplo 01:
Esboçar o gráfico e encontrar a área entre as curvas das funções representadas pelas integrais definidas seguintes e comprove-as usando as fórmulas apropriadas da Geometria.

01 - $\int_1^5 2\, dx$

02 - $\int_{-2}^{2} x + 3\, dx$

03 - $\int_{-1}^{1} \sqrt{1 - x^2}\, dx$

Solução 01:

Solução com Maple:

```
> restart: with(plots): with(plottools):with(student):
> f:=x->2;
```

$$f := x \rightarrow 2$$

```
> graf1:=plot(f(x),x=1..5,y=-1..3,color=blue,thickness=2):
> graf2:=polygon([[1,0],[5,0],[5,2],[0,2]],color=cyan):
> display(graf1,graf2);
```




```
> Int(f(x),x)=int(f(x),x);
```

$$\int 2\,dx = 2\,x$$

```
> A:=Int(f(x),x=1..5)=int(f(x),x=1..5);
```

$$A := \int_1^5 2\,dx = 8$$

Solução analítica:

Observamos que a função é a reta $y = 2$, limitada pelos extremos $x = 1$ e $x = 5$ e o eixo-x. Portanto, forma um retângulo de base igual a 4 unidades e altura 2 unidades. Assim sendo, a área $A = 4\,.\,2 = 8$.

Solução 02:

Solução com Maple:

```
> f:=x->x+3;
```

$$f := x \rightarrow x + 3$$

```
> graf1:=plot(f(x),x=-1..2,y=-2..5,color=blue,thickness=2):
> graf2:=plot(f(x),x=-1..2,y=-2..5,color=cyan,filled=true):
> graf3:=plot([[-1,0],[-1,f(-1)]],color=navy ):
> graf4:=plot([[2,0],[2,f(2)]],color=navy ):
> display(graf1,graf2,graf3,graf4);
```




```
> Int(f(x),x)=int(f(x),x);
```

$$\int x+3\,dx = \frac{1}{2}x^2 + 3\,x$$

```
> A:=Int(f(x),x=-1..2)=int(f(x),x=-1..2);
```

$$A := \int_{-1}^{2} x+3\,dx = \frac{21}{2}$$

Solução analítica:

Observamos que o gráfico do integrando é a reta $y = x + 3$. Por conseguinte, a região entre a reta e o eixo-x forma um trapézio, cuja altura está sob o eixo-x, com dimensão $2 - (-1) = 3$. A base maior tem dimensão de 5 unidades e a base menor tem dimensão de 2 unidade. Logo a área $A = \dfrac{(B+b)\,h}{2} = \dfrac{(5+2)\,3}{2} = \dfrac{21}{2}$.

Solução 03:

Solução com Maple:

```
> f:=x->sqrt(1-x^2);
```

$$f := x \to \sqrt{1-x^2}$$

```
> graf1:=plot(f(x),x=-1..1,y=0..1,color=blue,thickness=2):
> graf2:=plot(f(x),x=-1..1,y=0..1,color=cyan,filled=true):
> display(graf1,graf2,scaling=constrained);
```




```
> Int(f(x),x)=int(f(x),x);
```

$$\int \sqrt{1-x^2}\, dx = \frac{x\sqrt{1-x^2}}{2} + \frac{1}{2}\arcsin(x)$$

```
> A:=Int(f(x),x=-1..1)=int(f(x),x=-1..1);
```

$$A := \int_{-1}^{1} \sqrt{1-x^2}\, dx = \frac{\pi}{2}$$

Solução analítica:

Observamos que o gráfico da função $y = \sqrt{1 - x^2}$ é o semicírculo superior de uma circunferência, de raio unitário, com centro na orgiem. Por conseguinte, a área entre a semicircunferência e o eixo-x será $A = \frac{\pi R^2}{2} = \frac{\pi 1^2}{2} = \frac{\pi}{2}$.

Exemplo 02:

Sejam as funções $f(x) = 2x + 5$ e $g(x) = x^2 + 2$.

a) Encontrar as raizes para os quais as curvas se encontram;

b) traçar um gráfico das funções, em um mesmo sistema de eixos, destacando a região compreendida entre as curvas, no intervalo entre as raizes;

c) determinar a área da região compreendida entre as raizes.

Solução com Maple:

```
> restart: with(plots): with(student):
```

Definindo as funções:

```
> f:=x->2*x+5;
```

$$f := x \to 2x + 5$$

```
> g:=x->x^2+2;
```

$$g := x \to x^2 + 2$$

a) Encontrando as raizes da equação $f(x) = g(x)$:

```
> Raizes={solve(f(x)=g(x),x)};
```

$$Raizes = \{-1, 3\}$$

```
> x1:=-1; x2:=3;
```

$$x1 := -1$$

$$x2 := 3$$

b) Esboçando os gráficos das funções:

```
> graf1:=plot([f(x),g(x)], x = -2..4, y=0..15,thickness=2,
  color = [red,blue]):
> graf2:=plot({[-1+k/10,t,t=f(-1+k/10)..g(-1+k/10)] $
  k=1..39},thickness=1,color=tan):
> display(graf2,graf1);
```



c) Área compreendida entre as curvas, no intervalo entre as raizes $x = -1$ e $x = 3$:

$$A = \int_{-1}^{3} f(x) - g(x)\, dx = \int_{-1}^{3} [2x + 5] - [x^2 + 2]\, dx = \int_{-1}^{3} -x^2 + 2x + 3\, dx$$

```
> A:=Int(f(x)-g(x),x=-1..3)=int(f(x)-g(x),x=-1..3);
```

$$A := \int_{-1}^{3} 2x + 3 - x^2\, dx = \frac{32}{3}$$

Exemplo 03:

Encontrar o valor de d tal que a área compreendida entre as parábolas $y = x^2 - d^2$ e $y = d^2 - x^2$ seja igual a 576.

Solução analítica e com Maple:

Vamos construir um gráfico ilustrativo. Para isto, vamos assumir $d = 3$.
Desta forma, podemos definir as funções $f(x)$ e $g(x)$, a saber:

```
> restart: with(plots): with(student):
> f:=x->x^2-9;
```

$$f := x \rightarrow x^2 - 9$$

```
> g:=x->9-x^2;
```

$$g := x \rightarrow 9 - x^2$$

Pontos de interseção das curvas

```
> Raizes={solve(f(x)=g(x),x)};
```

$$Raizes = \{-3, 3\}$$

Traçado do gráfico, destacando a região compreendida entre as funções $f(x)$ e $g(x)$, no intervalo $x = [-3\,, 3\,]$:

```
> graf1:= plot(f(x),x=-4..4,y=-10..10,thickness=2,color=red):
> graf2:= plot(g(x),x=-4..4,y=-10..10,thickness=2,color=blue):
> graf3:=
  plot({[-3+i*(6/50),t,t=f(-3+i*(6/50))..g(-3+i*(6/50))] $
  i=1..49},thickness=1,color=tan):
> display(graf3,graf2,graf1);
```



Podemos inferir do gráfico acima, que no caso geral, as funções f e g interceptam-se em d e $-d$.

Assim sendo, a área compreendida é dada por: $\int_{-d}^{d} \mathrm{g}(x) - \mathrm{f}(x)\, dx$ =

$$\int_{-d}^{d} [d^2 - x^2] - [x^2 - d^2]\, dx = \int_{-d}^{d} 2\, d^2 - 2\, x^2\, dx \; .$$

Redefinindo as funções $\mathrm{f}(x)$ e $\mathrm{g}(x)$, em função do parâmetro d:

```
> f:=x->x^2-d^2; g:=x->d^2-x^2;
```

$$f := x \rightarrow x^2 - d^2$$
$$g := x \rightarrow d^2 - x^2$$

Calculando a integral no intervalo $[-d, d]$:

```
> A=Int(g(x)-f(x),x=-d..d);
```

$$A = \int_{-d}^{d} 2\,d^2 - 2\,x^2\,dx$$

```
> A:=int(g(x)-f(x),x=-d..d);
```

$$A := \frac{8\,d^3}{3}$$

Resolvendo-se a equação acima para d, e consideração apenas a 1a. raiz, (raiz real), obtemos:

```
> 'd'=solve(A=576,d)[1];
```

$$d = 6$$

Logo, a área entre as curvas $x^2 - d^2$ e $d^2 - x^2$, compreendida no intervalo de $[-6, 6]$ é igual a 576.

Exemplo 04:

Encontrar a área da região limitada pelas equações das curvas $y = x + 6$, $y = -\frac{x}{2}$ e $y = x^3$.

Solução analítica e com o Maple:

Inicialmente, vamos representar graficamente as curvas das equações:

```
> restart: with(plots): with(student):
> f:=x->x+6; g:=x->-x/2; h:=x->x^3;
```

$$f := x \to x + 6$$

$$g := x \to -\frac{1}{2}\,x$$

$$h := x \to x^3$$

```
> graf1:=plot(f(x),x=-5..3,y=-3..9,color=red,thickness=2):
> graf2:=plot(g(x),x=-5..3,y=-3..9,color=brown,thickness=2):
> graf3:=plot(h(x),x=-5..3,y=-3..9,color=magenta,thickness=2):
> graf4:=plot({[-4+k/10,t,t=f(-4+k/10)..g(-4+k/10)] $
  k=1..40},thickness=1,color=pink):
> graf5:=plot({[0+k/10,t,t=f(0+k/10)..h(0+k/10)] $
  k=0..19},thickness=1,color=tan):
> display(graf1,graf2,graf3,graf4,graf5);
```



Observando-se o gráfico acima, temos que a região hachurada entre as curvas apresenta duas regiões: uma delimitada entre as equações da reta, limitada pelo intervalo [-4 ; 0] e outra entre a equação da reta $y = x + 6$ e a curva $y = x^3$, no intervalo [0 ; 2].

Assim sendo, temos:

```
> a:=-4: b:=0: c:=2:
> Area1:=Int('f(x)-g(x)',x=a..b)=int(f(x)-g(x),x=a..b);
```

$$Area1 := \int_{-4}^{0} f(x) - g(x)\, dx = 12$$

```
> Area2:=Int('f(x)-h(x)',x=b..c)=int(f(x)-h(x),x=b..c);
```

$$Area2 := \int_{0}^{2} f(x) - h(x)\, dx = 10$$

```
> Area:=Area1+Area2;
```

$$Area := \int_{-4}^{0} \frac{3\,x}{2} + 6\, dx + \int_{0}^{2} x + 6 - x^3\, dx = 22$$

Propriedades da Integral Definida

Seja uma função $f(x)$ contínua em um intervalo $[\,a\,;\,b\,]$, então $f(x)$ é integrável em $[\,a\,;\,b\,]$. Considerando que $a < b$, portanto o limite superior de integração de uma integral definida é maior do que o limite inferior. Assim sendo, vamos considerar dois casos especiais em que os limites de integração são iguais ou o limite inferior é maior do que o superior. Desta forma, segue as definições especiais.

Definição 01

Se a pertencer ao domínio de $f(x)$, definimos

$$\int_{a}^{a} f(x)\, dx = 0$$

Definição 02

Se $f(x)$ for integrável em $[\,a\,;\,b\,]$, definimos

$$\int_{b}^{a} f(x)\, dx = -\int_{a}^{b} f(x)\, dx$$

Teorema 04

Se $f(x)$ e $g(x)$ forem integráveis em um intervalo $[\,a\,;\,b\,]$ e k uma constante real, então

$$\int_{a}^{b} k\, f(x)\, dx = k \int_{a}^{b} f(x)\, dx$$

$$\int_{a}^{b} [\,f(x) + g(x)\,]\, dx = \int_{a}^{b} f(x)\, dx + \int_{a}^{b} g(x)\, dx$$

$$\int_{a}^{b} [\,f(x) - g(x)\,]\, dx = \int_{a}^{b} f(x)\, dx - \int_{a}^{b} g(x)\, dx$$

Teorema 05

Se $f(x)$ for integrável em um intervalo fechado contendo três pontos a, b e c, em quaisquer ordem, então

$$\int_a^b f(x)\,dx = \int_a^c f(x)\,dx + \int_c^b f(x)\,dx$$

Teorema 06
Se $f(x)$ for integrável em um intervalo [a ; b] e $0 \le f(x)$ para todo x em [a ; b], então

$$0 \le \int_a^b f(x)\,dx$$

Teorema 07
Se $f(x)$ for integrável em um intervalo [a ; b] e $g(x) \le f(x)$ para todo x em [a ; b], então

$$\int_a^b g(x)\,dx \le \int_a^b f(x)\,dx$$

Exercícios propostos

I - Esboce a curva e a região cuja área com sinal é representada pela integral definida e calcule esta integral. Comprove o resultado usando uma fórmula apropriada da Geometria, quando possível.

01 - $f(x) = x$

a) $\int_0^3 f(x)\,dx$

b) $\int_{-2}^{-1} f(x)\,dx$

c) $\int_{-5}^{5} f(x)\,dx$

02 - $f(x) = x^2 - 1$

a) $\int_1^2 f(x)\,dx$

b) $\int_{-2}^{-1} f(x)\, dx$

c) $\int_{-1}^{1} f(x)\, dx$

d) $\int_{-2}^{2} f(x)\, dx$

e) $\int_{-3}^{3} f(x)\, dx$

03 - $f(x) = \cos(x)$

a) $\int_{0}^{\frac{\pi}{2}} f(x)\, dx$

b) $\int_{0}^{\pi} f(x)\, dx$

c) $\int_{-\pi}^{\pi} f(x)\, dx$

04 - $f(x) = |x - 1|$

a) $\int_{1}^{3} f(x)\, dx$

b) $\int_{-3}^{1} f(x)\, dx$

c) $\int_{-3}^{3} f(x)\, dx$

05 - $f(x) = \sqrt{4 - x^2}$

a) $\int_{0}^{2} f(x)\, dx$

b) $\int_{-2}^{0} f(x)\, dx$

c) $\int_{-2}^{2} f(x)\, dx$

II - Considere a função $f(x) = x + \sqrt{1 - x^2}$. Determine:
a) a área da região positiva da curva
b) a área da região negativa da curva
c) a área geométrica total da curva
d) avalie a área dada pela integral definida da função

III - Considere as inequações $x^2 + y^2 \leq 1$ e $1 \leq |x + y|$. Determine a área compreendida entre as curvas que resulta da simultaneidade das inequações, em um mesmo sistema de plano cartesiano. Desenvolva a solução:
a) usando procedimentos da Geometria
b) usando integral definida

Sugestão: Desenvolva, graficamente, o sistema de inequações fazendo uso do comando gráfico **Maple** *implicitplot*.

Teoremas Fundamentais

Valor Médio de uma Função

Se desejarmos calcualr a média de um conjunto de *n* números x_1, x_2, x_3, **...** , x_n devemos somar todos esses valores e dividir esta soma por *n*, o qual podemos representar da forma seguinte:

$$x_{médio} = \frac{\sum_{i=1}^{n} x_i}{n}$$

Mas, se desejarmos calcular a média de uma função contínua em um intervalo [a ; b]?

Se esta função for constante, ou seja, $y = \mathrm{f}(x) = k$, em um intervalo $[\ a\ ;\ b\ ]$, então a média é o próprio valor da constante k. Se a constante k for positiva, então formará um retângulo com altura k e largura $b - a$. A média da função pode ser vista, geometricamente, como a área deste retângulo dividido por $b - a$. A figura seguinte mostra esta situação.



Qual será a média da função $\mathrm{f}(x) = 2\,x$, no intervalo $[\ 0\ ;\ 4]$?
Podemos proceder, analogamente, calculando a área sob a curva e dividindo-a pela largura do intervalo, conforme observamos no gráfico a seguir.



Neste exemplo, não precisamos recorrer à Soma de Riemann para avaliarmos a área sob a curva, pois podemos calcular a área do triângulo pela fórmula da Geometria, ou seja:

$$A = \frac{B\,h}{2}$$

Como a base B pode ser vista como a largura do intervalo $b - a$, então $B = 4 - 0 = 4$, e a altura o valor da função aplicada no ponto $b = 4$, ou seja, $h = \mathrm{f}(4) - 0 = 8 - 0 = 8$; assim sendo, a área $A = \frac{4.\,8}{2}$ $= 16$. Portanto, o valor médio da função será:

$$Valor\_Médio = \frac{Área}{b - a}$$

$$Valor\_Médio = \frac{16}{4 - 0} = 4$$

Vamos determinar a média da função $\mathrm{f}(x) = \mathrm{sen}(x)$, no intervalo $[\ 0\ ;\ \pi\ ]$.

Inicialmente, traçamos o gráfico da função no intervalo indicado.

```
> restart: with(plots): with(student): Digits:=5:
> f:=x->sin(x);
```

$$f := x \to \sin(x)$$

```
> graf1:=plot(f(x),x=0..Pi,color=blue,thickness=2):
> graf2:=plot(f(x),x=0..Pi,color=turquoise,filled=true):
> display(graf1,graf2);
```



Vamos estimar a área da função $f(x) = \text{sen}(x)$ no intervalo $[\ 0, \pi\ ]$, usando a *Soma de Riemann*, dividindo a curva em 5 intervalos e aplicando comandos **Maple** correspondente, ou seja:

```
> a:=0: b:=Pi: n:=5:
> middlebox(f(x),x=a..b,n,color=blue,shading=turquoise);
```




```
> Area=middlesum(f(x),x=a..b,n,color=blue,shading=turquoise)
;
```

$$Area = \frac{1}{5}\pi\left(\sum_{i=0}^{4}\sin\left(\frac{\left(i+\frac{1}{2}\right)\pi}{5}\right)\right)$$

```
> Area:=evalf(rhs(%));
```

$$Area := 2.0334$$

```
> Média_Função='Area'/('Pi-0');
```

$$Média\_Função = \frac{Area}{\pi}$$

```
> Média_Função=evalf(Area/Pi);
```

$$Média\_Função = 0.64725$$

Vamos repetir o procedimeto, aumentando-se o número de retângulos para 20, conforme sequência de comandos a seguir:

```
> a:=0: b:=Pi: n:=20:
> middlebox(f(x),x=a..b,n,color=blue,shading=turquoise);
```




```
> Area=middlesum(f(x),x=a..b,n,color=blue,shading=turquoise)
;
```

$$2.0334 = \frac{1}{20}\pi\left(\sum_{i=0}^{19}\sin\left(\frac{\left(i+\frac{1}{2}\right)\pi}{20}\right)\right)$$

```
> Area:=evalf(rhs(%));
```

$$Area := 2.0022$$

```
> Média_Função=evalf(Area/Pi);
```

$$Média\_Função = 0.63732$$

O valor exato pode ser obtido, ao invés de aplicarmos a *Soma de Riemann*, utilizando-se a área calculada com integrais definidas, conforme a seguir:

```
> Area:=Int(sin(x),x=0..Pi)=int(sin(x),x=0..Pi);
```

$$Area := \int_0^{\pi} \sin(x)\,dx = 2$$

```
> Média_Função=int(sin(x),x=0..Pi)/Pi;
```

$$Média\_Função = \frac{2}{\pi}$$

```
> evalf(%);
```

$$\mathit{Média\_Função} = 0.63662$$

Considerando os exemplos acima explanados, podemos enunciar o teorema seguinte:

Teorema 08 - Teorema do Valor Médio

Se $f(x)$ for uma função contínua em um intervalo [a ; b], então seu **valor médio** neste intervalo, também chamado **média da função** será:

$$\mathrm{Med}(f) = \frac{1 \int_a^b f(x)\, dx}{b - a}$$

Uma outra maneira de enunciarmos o teorema anterior é:

Se $f(x)$ for uma função contínua em um intervalo [a ; b], existe um ponto *c* entre *a* e *b*, tal que

$$\int_a^b f(x)\, dx = (b - a)\, f(c)$$

A interpretação geométrica do Teorema do Valor Médio fornece-nos que a área sob a curva $y = f(x)$, entre *a* e *b*, é igual a área de um retângulo de base $[b - a]$ e altura $f(c)$, conforme figura a seguir.



Exemplo 01:

Determinar a média da função $f(x) = \sqrt{4 - x^2}$, no intervalo $x = [-2 ; 2]$.

Inicialmente, traçamos o gráfico da função no intervalo indicado.

```
> restart: with(plots): with(student): Digits:=5:
> f:=x->sqrt(4-x^2);
```

$$f := x \to \sqrt{4 - x^2}$$

```
> a:=-2: b:=2:
> graf1:=plot(f(x),x=a..b,color=blue,thickness=2):
> graf2:=plot(f(x),x=a..b,color=turquoise,filled=true):
> graf3:=plot(Pi/2,x=a..b,color=red,linestyle=3,thickness=2)
  :
> display(graf1,graf2,graf3,scaling=constrained);
```




```
> Media:='(1/(b-a))'*Int('f(x),x=a..b');
```

$$Media := \frac{1}{b-a}\int_a^b f(x)\,dx$$

```
> Media:=(1/(b-a))*Int(f(x),x=a..b)=int(f(x),x=a..b)/(b-a);
```

$$Media := \frac{1}{4}\int_{-2}^{2}\sqrt{4-x^2}\,dx = \frac{\pi}{2}$$

Exemplo 02:

Determinar a média da função $f(x) = 4 - x^2$, no intervalo $x = [-2 ; 2]$.

Inicialmente, traçamos o gráfico da função no intervalo indicado.

```
> restart: with(plots): with(student): Digits:=5:
> f:=x->4-x^2;
```

$$f := x \to 4 - x^2$$

```
> a:=-2: b:=2:
> graf1:=plot(f(x),x=a..b,color=blue,thickness=2):
> graf2:=plot(f(x),x=a..b,color=turquoise,filled=true):
```

```
> graf3:=plot(8/3,x=a..b,color=red,linestyle=3,thickness=2):
> display(graf1,graf2,graf3,scaling=constrained);
```




```
> Media:='(1/(b-a))'*Int('f(x),x=a..b');
```

$$Media := \frac{1}{b-a}\int_a^b f(x)\,dx$$

```
> Media:=(1/(b-a))*Int(f(x),x=a..b)=int(f(x),x=a..b)/(b-a);
```

$$Media := \frac{1}{4}\int_{-2}^{2} 4 - x^2\,dx = \frac{8}{3}$$

Exemplo 03:

Determinar a média e o ponto médio da função $f(x) = -x^2$, no intervalo $x = [-1\ ;\ 3\ ]$.

Inicialmente, traçamos o gráfico da função no intervalo indicado.

```
> restart: with(plots): with(student): Digits:=5:
> f:=x->-x^2;
```

$$f := x \rightarrow -x^2$$

```
> a:=-1: b:=3:
> graf1:=plot(f(x),x=a..b,y=-8..1,color=blue,thickness=2):
```

```
> graf2:=plot(f(x),x=a..b,y=-8..1,color=turquoise,filled=true):
> graf3:=plot(-7/3,x=a..b,color=red,linestyle=2,thickness=2):
> graf4:=plot([[sqrt(21)/3,0],[sqrt(21)/3,f(sqrt(21)/3)]],color=red,linestyle=2,thickness=2):
> display(graf1,graf2,graf3,graf4,scaling=constrained);
```




```
> Media:='(1/(b-a))'*Int('f(x),x=a..b');
```

$$Media := \frac{1}{b-a}\int_{a}^{b} \mathrm{f}(x)\,dx$$

```
> Media:=(1/(b-a))*Int(f(x),x=a..b)=int(f(x),x=a..b)/(b-a);
```

$$Media := \frac{1}{4}\int_{-1}^{3} -x^2\,dx = \frac{-7}{3}$$

```
> x[med]:=solve(f(x)=int(f(x)/(b-a),x=a..b))[2];
```

$$x_{med} := \frac{\sqrt{21}}{3}$$

Teorema Fundamental do Cálculo

Teorema 09 - Teorema Fundamental do Cálculo - Parte I
Se $f(x)$ for uma função contínua em um intervalo $[\,a\,;\,b]$ e se $F(x)$ for uma primitiva de $f(x)$ nesse intervalo, então

$$\int_a^b f(x)\,dx = F(b) - F(a)\,.$$

N.B.: Este teorema nos garante avaliar uma integral definida sem termos a necessidade de aplicarmos as *Somas de Riemann*.

Exemplo 01:

a) Calcular a integral definida $\int_1^3 2\,x + 1\,dx$.

Solução analítica:

Sabemos que $F(x) = x^2 + x$ é uma primitiva de $f(x) = 2\,x + 1$. Por conseguinte:

$\int_1^3 2\,x + 1\,dx = x^2 + x$. Como desejamos avaliar a integral definida no intervalo [1 ; 3], então temos que
$F(x) = x^2 + x = F(3) - F(1) = [\,3^2 + 3\,]$ - $[\,1^3 + 1\,]$ $= 9 + 3 - 1 - 1$ = 10.

Solução com Maple:

```
> restart:
> F:=Int(2*x+1,x=1..3);
```

$$F := \int_1^3 2\,x + 1\,dx$$

```
> value(F); # avalia a integral
```

$$10$$

b) Calcular a integral definida $\int_0^{\frac{\pi}{2}} \cos(x)\,dx$.

Solução analítica:

Sabemos que $F(x) = \mathrm{sen}(x)$ é uma primitiva de $f(x) = \cos(x)$. Por conseguinte:

$\int_0^{\frac{\pi}{2}} \cos(x)\, dx = \operatorname{sen}(x)$. Como desejamos avaliar a integral definida no intervalo $[\, 0\, ;\, \frac{\pi}{2}\,]$, então temos que

$F(x) = \operatorname{sen}(x) = F\left(\frac{\pi}{2}\right) - F(0) = \operatorname{sen}\left(\frac{\pi}{2}\right) - \operatorname{sen}(0) \; = 1 - 0 \; = 1.$

Solução com Maple:

```
> restart:
> F:=Int(cos(x),x=0..Pi/2);
```

$$F := \int_0^{\frac{\pi}{2}} \cos(x)\, dx$$

```
> value(F);
```

$$1$$

Teorema 09 - Teorema Fundamental do Cálculo -Parte II

Se $f(x)$ for uma função contínua em um intervalo $[\, a\, ;\, b]$ então $F(x) = \int_a^x f(t)\, dt$ é contínua em $[\, a\, ;\, b]$ e derivável em $[\, a\, ;\, b]$ e sua derivada é $f(x)$. Ou seja:

$$F'(x) = \frac{d}{dx}\left[\int_a^x f(t)\, dt\right] = f(x)$$

Exemplo 02:

a) Determinar $\frac{d}{dx}\left[\int_a^x \cos(t)\, dt\right]$

Solução analítica:

$\frac{d}{dx}\left[\int_a^x \cos(t)\, dt\right] = \cos(x)$, aplicando-se diretamente o teorema fundamental do cálculo - parte II.

Solução com Maple:

```
> restart:
> f:=Diff(Int(cos(t),t=a..x),x);
```

$$f := \frac{\partial}{\partial x}\left(\int_a^x \cos(t)\, dt\right)$$

```
> value(f);
```

$$\cos(x)$$

b) Determinar $\frac{d}{dx}\left[\int_0^x \frac{1}{1+t^2}\,dt\right]$

Solução analítica:

$\frac{d}{dx}\left[\int_0^x \frac{1}{1+t^2}\,dt\right] = \frac{1}{1+x^2}$, aplicando-se diretamente o teorema fundamental do cálculo - parte II.

Solução com Maple:

```
> restart:
> f:=Diff(Int(1/(1+t^2),t=0..x),x);
```

$$f := \frac{\partial}{\partial x}\left(\int_0^x \frac{1}{1+t^2}\,dt\right)$$

```
> value(f);
```

$$\frac{1}{1+x^2}$$

c) Determine $\frac{dy}{dx}$, se $y = \int_1^{x^2} \cos(t)\,dt$.

Solução analítica:

Neste caso, temos que efetuar uma mudança de variável, pois o limite superior da integral não é mais *x*, porém x^2 .

Fazendo $u = x^2$, temos que $\frac{du}{dx} = 2\,x$.

Assim sendo, devemos aplicar a Regra da Cadeia para encontrarmos $\frac{dy}{dx}$. Ou seja: $\frac{dy}{dx} = \frac{dy}{du} \cdot \frac{du}{dx}$

Desta forma, temos: y = $\int_{1}^{u} \cos(t)\, dt$. Aplicando-se o Teorema Fundamental do Cálculo, temos: $\frac{dy}{du} = \cos(u)$.

Portanto, $\frac{dy}{dx} = \frac{dy}{du} \cdot \frac{du}{dx} = \cos(u) \cdot 2\,x = \cos(x^2) \cdot 2\,x = 2\,x\cos(x^2)$.

Solução com Maple:

```
> restart:
> f:=Diff(Int(cos(t),t=0..x^2),x);
```

$$f := \frac{\partial}{\partial x}\left(\int_{0}^{x^2} \cos(t)\, dt\right)$$

```
> value(f);
```

$$2\cos(x^2)\,x$$

```
>
```

Os dois teoremas fundamentais do cálculo, acima enunciados, fornece-nos que a *diferenciação* e a *integração* são processos inversos, pois uma operação desfaz o efeito da outra. Uma vez que:

$$\int_{a}^{x} f\,'(t)\, dt = f(x) - f(a)$$

e se o valor de $f(a)$ for conhecido, então a função pode ser obtida a partir de sua derivada por integração.

Por outro lado,

$$\frac{d}{dx}\left[\int_{a}^{x} f(t)\, dt\right] = f(x)$$

significa que a função pode ser obtida a partir de sua integral por diferenciação.

Exercícios Propostos

I - Calcule as integrais definidas nos itens abaixo:

01 - $\int_{-2}^{2} 2\,x + 5\; dx$

02 - $\int_{0}^{1} x^2 + \sqrt{x}\; dx$

03 - $\int_{0}^{\pi} \operatorname{sen}(x)\; dx$

04 - $\int_{0}^{\pi} \cos(x)\; dx$

05 - $\int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \cos(x)\; dx$

06 - $\int_{-2}^{2} |x|\; dx$

07 - $\int_{-1}^{1} \frac{1}{x} - \mathbf{e}^{(-x)}\; dx$

08 - $\int_{0}^{1} \frac{4}{1 + x^2}\; dx$

09 - $\int_{0}^{1} \frac{x}{\sqrt{1 + x^2}}\; dx$

10 - $\int_{-1}^{0} \pi^{(x-1)}\; dx$

II - Determine $\frac{dy}{dx}$ nos itens a seguir:

01 - $y = \int_{0}^{x} \sqrt{1+t^2}\, dt$

02 - $y = \int_{\sqrt{x}}^{0} \operatorname{sen}(t^2)\, dt$

03 - $y = \int_{0}^{\operatorname{sen}(x)} \frac{1}{\sqrt{1-t^2}}\, dt \,, |x| < \frac{\pi}{2}$

04 - $y = \int_{0}^{e^{(x^2)}} \frac{1}{\sqrt{t}}\, dt$

Métodos de Integração

Nesta seção, abordaremos técnicas para encontrarmos primitivas de funções que não sejam claramente reconhecidas como derivadas das funções que estamos desejando integrar. Ao longo do tempo foram compiladadas e desenvolvidas tabelas de integração, as quais incorporaram as habilidades e experiências de muitos matemáticos e cientistas envolvidos com o cálculo diferencial e integral, no desenvolvimento dessas tabelas, as quais podemos encontrar, facilmente, em várias bibliografias relacionadas com o tema. Recentemente, temos a incorporação dos programas de computação algébrica ou simbólica, tais como o *Derive*, *Mathematica*, *Maple*, entre outros, que são capazes de calcular integrais de difícil resolução, porém ainda se faz necessário desenvolvermos habilidades gerais para encontrar as primitivas, através de métodos de transformação, convertendo integrais não-conhecidas em integrais conhecidas, como os métodos da substituição, por partes e frações parciais, que serão analisados a seguir, com o suporte do *software Maple*.

Método da Substituição

O método da substituição consiste em efetuarmos uma mudança de variável e é um processo análogo à Regra da Cadeia da derivação.

Teorema 9 - Regra da Substituição

Se $u = \mathrm{g}(x)$ é uma função derivável cuja imagem é um intervalo I e $\mathrm{f}(x)$ é contínua neste intervalo, então

$$\int [\,\mathrm{f}(\mathrm{g}(x))\,]\,[\,\mathrm{g}'(x)\,]\,dx = \int \mathrm{f}(u)\,du$$

A regra da substituição fornece-nos o seguinte método para calcularmos a integral:

1- Substitua $u = \mathrm{g}(x)$ e $du = \mathrm{g}'(x)dx$ para obter a integral $\int \mathrm{f}(u)\,du$

2 - Integre em relação a *u*.

3 - Substitua *u* por g(*x*) no resultado final.

Exemplo 01:

Encontrar a integral $\displaystyle\int \frac{2\,x}{1+x^2}\,dx$.

Solução analítica:

Fazemos $u = 1 + x^2$. Então $du = 2\,dx$. Assim sendo,

$$\int \frac{2\,x}{1+x^2}\,dx = \int \frac{1}{u}\,du$$

$$\int \frac{2\,x}{1+x^2}\,dx = \ln(|u|) + c$$

$$\int \frac{2\,x}{1+x^2}\,dx = \ln(1 + x^2) + c$$

Solução com Maple:

O **Maple** possui o comando *changevar* que é usado na resolução da integração por substituição, o qual faz parte do pacote *student* [ver no *Help* a sintaxe do comando *changevar*].

Primeiro, vamos definir uma expressão **p1** para a integral indefinida, e então podemos efetuar a substituição desejada.

```
> restart: with(student):
> p1:=Int(2*x/(1+x^2),x);
```

$$p1 := \int \frac{2\,x}{1+x^2}\,dx$$

Vamos definir uma expressão **p2** para a mudança de variável. O terceiro argumento do comando **changevar** é necessário, pois há duas variáveis no problema (*x* e *u*), e devemos dizer ao **Maple** quais delas é para ser a nova variável de integração.

```
> p2:=changevar(u=1+x^2,p1,u);
```

$$p2 := \int \frac{1}{u}\,du$$

O uso do comando **simplify** na linha seguinte é opcional; em alguns exemplos ele pode ou não ser necessário.

```
> p3:=simplify(p2);
```

$$p3 := \int \frac{1}{u}\,du$$

A expressão de **p3** é uma integral conhecida - inverso de *u*, que fornece o logarítmo de *u* - e assim nós podemos avaliá-la. O comando **value** avaliará uma expressão que não foi desenvolvida (calculada).

```
> p4:=value(p3);
```

$$p4 := \ln(u)$$

Finalmente, devemos escrever a resposta em termos da variável x, usando a equação da substituição $u = x^2 + 1$. Primeiro, encontramos uma anti-derivada, e então o caso geral.

```
> p5:=subs(u=x^2+1,p4);
```

$$p5 := \ln(1+x^2)$$

A solução geral pode ser facilmente escrita:

```
> p1=p5+c;
```

$$\int \frac{2\,x}{1+x^2}\,dx = \ln(1+x^2)+c$$

Verificação:

Podemos verificar a resposta com o comando **int**.

```
> Int(2*x/(1+x^2),x)=int(2*x/(1+x^2),x);
```

$$\int \frac{2\,x}{1+x^2}\,dx = \ln(1+x^2)$$

Devemos ter o hábito de verificarmos o resultado pela diferenciação da antiderivada. Observamos que **p5** é uma expressão, assim devemos usar o comando **diff**.

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}\ln(1+x^2) = \frac{2\,x}{1+x^2}$$

Exemplo 02:

Encontrar a integral $\int x^2\,\mathbf{e}^{(x^3)}\,dx$.

Solução analítica e com o Maple:

```
> restart: with(student):
> p1:=Int(x^2*exp(x^3),x);
```

$$p1 := \int x^2\,\mathbf{e}^{(x^3)}\,dx$$

Fazendo-se $u = x^3$. Então, $du = 3\,x^2\,dx$.

```
> p2:=changevar(u=x^3,p1,u);
```

$$p2 := \int \frac{1}{3}\,\mathbf{e}^u\,du$$

```
> p3:=simplify(p2);
```

$$p3 := \frac{1}{3}\int \mathbf{e}^u\,du$$

```
> p4:=value(p3);
```

$$p4 := \frac{1}{3}\,\mathbf{e}^u$$

```
> p5:=subs(u=x^3,p4);
```

$$p5 := \frac{1}{3}\,\mathbf{e}^{(x^3)}$$

```
> p1=p5+c;
```

$$\int x^2\, e^{(x^3)}\, dx = \frac{1}{3} e^{(x^3)} + c$$

Verificação:

```
> p1=int(x^2*exp(x^3),x);
```

$$\int x^2\, e^{(x^3)}\, dx = \frac{1}{3} e^{(x^3)}$$

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}\left(\frac{1}{3} e^{(x^3)}\right) = x^2\, e^{(x^3)}$$

Exemplo 03:

Encontre a integral $\int \mathrm{sen}(x+1)\, dx$.

Solução analítica e com o Maple:

```
> restart: with(student):
> p1:=Int(sin(x+1),x);
```

$$p1 := \int \sin(x+1)\, dx$$

Fazendo-se $u = x + 1$. Então, $du = dx$.

```
> p2:=changevar(u=x+1,p1,u);
```

$$p2 := \int \sin(u)\, du$$

```
> p3:=simplify(p2);
```

$$p3 := \int \sin(u)\, du$$

```
> p4:=value(p3);
```

$$p4 := -\cos(u)$$

```
> p5:=subs(u=x+1,p4);
```

$$p5 := -\cos(x+1)$$

```
> p1=p5+c;
```

$$\int \sin(x+1)\, dx = -\cos(x+1) + c$$

Verificação:

```
> p1=int(sin(x+1),x);
```

$$\int \sin(x+1)\, dx = -\cos(x+1)$$

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}(-\cos(x+1)) = \sin(x+1)$$

Exemplo 04:

Encontre a integral $\int \operatorname{sen}(x)^2 \cos(x)\, dx$.

Solução analítica e com o Maple:

```
> restart: with(student):
> p1:=Int(sin(x)^2*cos(x),x);
```

$$p1 := \int \sin(x)^2 \cos(x)\, dx$$

Fazendo-se $u = \operatorname{sen}(x)$. Então, $du = \cos(x)\, dx$.

```
> p2:=changevar(u=sin(x),p1,u);
```

$$p2 := \int u^2\, du$$

```
> p3:=simplify(p2);
```

$$p3 := \int u^2\, du$$

```
> p4:=value(p3);
```

$$p4 := \frac{u^3}{3}$$

```
> p5:=subs(u=sin(x),p4);
```

$$p5 := \frac{1}{3}\sin(x)^3$$

```
> p1=p5+c;
```

$$\int \sin(x)^2 \cos(x)\, dx = \frac{1}{3}\sin(x)^3 + c$$

Verificação:

```
> p1=int(sin(x)^2*cos(x),x);
```

$$\int \sin(x)^2 \cos(x)\, dx = \frac{1}{3}\sin(x)^3$$

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}\left(\frac{1}{3}\sin(x)^3\right) = \sin(x)^2 \cos(x)$$

Exemplo 05:

Encontre a integral $\int \frac{x}{\sqrt{1-x^4}}\,dx$.

Solução analítica e com o Maple:

```
> restart: with(student):
> p1:=Int(x/sqrt(1-x^4),x);
```

$$p1 := \int \frac{x}{\sqrt{1-x^4}}\,dx$$

Fazendo-se $u = x^2$. Então, $du = 2\,dx$.

```
> p2:=changevar(u=x^2,p1,u);
```

$$p2 := \int \frac{1}{2\sqrt{1-u^2}}\,du$$

```
> p3:=simplify(p2);
```

$$p3 := \frac{1}{2}\int \frac{1}{\sqrt{1-u^2}}\,du$$

```
> p4:=value(p3);
```

$$p4 := \frac{1}{2}\arcsin(u)$$

```
> p5:=subs(u=x^2,p4);
```

$$p5 := \frac{1}{2}\arcsin(x^2)$$

```
> p1=p5+c;
```

$$\int \frac{x}{\sqrt{1-x^4}}\,dx = \frac{1}{2}\arcsin(x^2) + c$$

Verificação:

```
> p1=int(x/sqrt(1-x^4),x);
```

$$\int \frac{x}{\sqrt{1-x^4}}\,dx = \frac{1}{2}\arcsin(x^2)$$

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}\left(\frac{1}{2}\arcsin(x^2)\right) = \frac{x}{\sqrt{1-x^4}}$$

Exemplo 06:

Encontre a integral $\int \frac{1}{e^x + e^{(-x)}}\, dx$.

Solução analítica e com o Maple:

```
> restart: with(student):
> p1:=Int(1/(exp(x)+exp(-x)),x);
```

$$p1 := \int \frac{1}{e^x + e^{(-x)}}\, dx$$

Multiplicando-se o numerador e o denominador por e^x , temos:

```
> p1a:=Int(exp(x)/(exp(2*x)+1),x);
```

$$p1a := \int \frac{e^x}{e^{(2x)} + 1}\, dx$$

Fazendo-se $u = e^x$. Então, $du = e^x\, dx$.

```
> p2:=changevar(u=exp(x),p1a,u);
```

$$p2 := \int \frac{1}{u^2 + 1}\, du$$

```
> p3:=simplify(p2);
```

$$p3 := \int \frac{1}{u^2 + 1}\, du$$

```
> p4:=value(p3);
```

$$p4 := \arctan(u)$$

```
> p5:=subs(u=exp(x),p4);
```

$$p5 := \arctan(e^x)$$

```
> p1=p5+c;
```

$$\int \frac{1}{e^x + e^{(-x)}}\, dx = \arctan(e^x) + c$$

Verificação:

```
> p1=int(exp(x)/(exp(2*x)+1),x);
```

$$\int \frac{1}{e^x + e^{(-x)}}\, dx = \arctan(e^x)$$

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx} \arctan(e^x) = \frac{e^x}{1 + (e^x)^2}$$

Exemplo 07:

Encontre a integral $\displaystyle\int \frac{\ln(x^2)}{x}\, dx$.

Solução analítica e com o Maple:

```
> restart: with(student):
> p1:=Int(ln(x^2)/x,x);
```

$$p1 := \int \frac{\ln(x^2)}{x}\, dx$$

Desenvolvendo a expressão do logarítmo, usando a propriedade da potenciação, temos:

```
> p1a:=Int(2*ln(x)/x,x);
```

$$p1a := \int \frac{2 \ln(x)}{x}\, dx$$

Fazendo-se $u = \ln(x)$. Então, $du = \dfrac{dx}{x}$.

```
> p2:=changevar(u=ln(x),p1a,u);
```

$$p2 := \int 2\, u\, du$$

```
> p3:=simplify(p2);
```

$$p3 := 2 \int u\, du$$

```
> p4:=value(p3);
```

$$p4 := u^2$$

```
> p5:=subs(u=ln(x),p4);
```

$$p5 := \ln(x)^2$$

```
> p1=p5+c;
```

$$\int \frac{\ln(x^2)}{x}\,dx = \ln(x)^2 + c$$

Verificação:

```
> p1=int(2*ln(x)/x,x);
```

$$\int \frac{\ln(x^2)}{x}\,dx = \ln(x)^2$$

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}(\ln(x)^2) = \frac{2\ln(x)}{x}$$

Teorema 10: Integração por Substituição em Integrais Definidas

Se g´(x) é contínua no intervalo [a ; b] e f(x) é contínua na imagem de g(x), então

$$\int [\,f(g(x))\,]\,[\,g'(x)\,]\,dx = \int_{g(a)}^{g(b)} f(u)\,du$$

O teorema acima nos indica que os limites de integração mudam quando a variável de integração é mudada por substituição.

Exemplo 08:

Calcular a área sob a curva $y = \text{sen}(x)^2$ e o eixo-x , no intervalo [0 ; 2 π] .

Solução analítica e com o Maple:

```
> restart: with(student): with(plots):
```

Inicialmente, vamos definir a função e traçar o seu gráfico, no intervalo dado:

```
> f:=x->sin(x)^2;
```

$$f := x \to \sin(x)^2$$

```
> graf1:=plot(f(x),x=0..2*Pi,color=blue,thickness=2):
```

```
> graf2:=plot(f(x),x=0..2*Pi,filled=true,color=turquoise):
> display(graf2,graf1);
```



Encontrando a integral da função $\mathrm{f}(x) = \mathrm{sen}(x)^2$.

```
> p0:=Int(sin(x)^2,x=0..2*Pi);
```

$$p0 := \int_0^{2\pi} \sin(x)^2 \, dx$$

Neste caso, temos duas opções a saber:

I - Calcularmos a integral indefinida, pelo método da substituição, e aplicarmos depois os limites originais de *x*.
II - Efetuarmos a mudança de variável e aplicarmos os limites transformados, conforme teorema anterior.

Aplicando-se a identidade trigonométrica, temos que $\mathrm{sen}(x)^2 = \dfrac{1 - \cos(2\,x)}{2}$.

Logo, a integral $\int \sin(x)^2\, dx = \int \frac{1-\cos(2\,x)}{2}\, dx$.

```
> p1:=Int((1-cos(2*x))/2,x);
```

$$p1 := \int \frac{1}{2} - \frac{1}{2}\cos(2\,x)\, dx$$

Fazendo-se $u = 2\,x$. Então, $du = 2\,dx$.

```
> p2:=changevar(u=2*x,p1,u);
```

$$p2 := \int \frac{1}{4} - \frac{1}{4}\cos(u)\, du$$

```
> p3:=simplify(p2);
```

$$p3 := \frac{1}{4}\int 1 - \cos(u)\, du$$

```
> p4:=value(p3);
```

$$p4 := \frac{u}{4} - \frac{1}{4}\sin(u)$$

```
> p5:=subs(u=2*x,p4);
```

$$p5 := \frac{x}{2} - \frac{1}{4}\sin(2\,x)$$

Desta forma, temos que:

$\int_0^{2\,\pi} \sin(x)^2\, dx = \int_0^{2\,\pi} \frac{1}{2} - \frac{1\cos(2\,x)}{2}\, dx = \frac{1\,x}{2} - \frac{1\sin(2\,x)}{4}$, com x pertence ao intervalo [$0\,;2\,\pi$].

Aplicando-se os extremos originais de x na expressão acima, obtemos:

$$Area = \mathrm{p5}(2\,\pi) - \mathrm{p5}(0) = \left[\frac{2\,\pi}{2} - \frac{\mathrm{sen}(4\,\pi)}{4}\right] - \left[\frac{0}{2} - \frac{\mathrm{sen}(0)}{4}\right] = \pi - 0 = \pi\ .$$

Com comandos **Maple**, temos:

```
> Area:=subs(x=2*Pi,p5)-subs(x=0,p5);
```

$$Area := \pi - \frac{1}{4}\sin(4\,\pi) + \frac{1}{4}\sin(0)$$

```
> 'Area'=simplify(Area);
```

$$Area = \pi$$

Efetuando-se a mudança de variável e aplicando-se os limites transformados:

```
> p1:=Int((1-cos(2*x))/2,x=0..2*Pi);
```

$$p1 := \int_0^{2\pi} \frac{1}{2} - \frac{1}{2}\cos(2x)\, dx$$

Fazendo-se $u = 2x$. Então, $du = 2\, dx$. Temos ainda que: $u(0) = 0$ e $u(2\pi) = 4\pi$.

```
> p2:=changevar(u=2*x,p1,u);
```

$$p2 := \int_0^{4\pi} \frac{1}{4} - \frac{1}{4}\cos(u)\, du$$

```
> p3:=simplify(p2);
```

$$p3 := \frac{1}{4}\int_0^{4\pi} 1 - \cos(u)\, du$$

```
> p4:=value(p3);
```

$$p4 := \pi$$

```
> 'Area'=p4;
```

$$Area = \pi$$

Exercícios propostos

Calcular as integrais usando o método da substituição:

01 - $\int 2x(x^2+1)\, dx$

02 - $\int \frac{\operatorname{sen}(\sqrt{x})}{\sqrt{x}}\, dx$

03 - $\int \sec(2x-1)^2\, dx$

04 - $\int \frac{1}{x\ln(x)}\, dx$

05 - $\displaystyle\int \frac{\mathrm{sen}(3\,x)}{1+\cos(3\,x)}\,dx$

06 - $\displaystyle\int \frac{1}{(1-4\,x^2)^{\left(\frac{1}{2}\right)}}\,dx$

07 - $\displaystyle\int x^2\,\mathbf{e}^{(-2\,x^3)}\,dx$

08 - $\displaystyle\int \frac{\mathrm{sen}(x)}{1+\cos(x)^2}\,dx$

Método por Partes

Teorema 11 - Regra por Partes

A integração por partes é um método para simplificar integrais na forma do produto de funções, isto é:

$$\int \mathrm{f}(x)\,\mathrm{g}(x)\,dx$$

Sejam $u = \mathrm{f}(x)$ e $v = \mathrm{g}(x)$ funções contínuas deriváveis em um intervalo I, então a integração por partes é expressa pela fórmula seguinte:

$$\int [\,\mathrm{f}(x)\,\mathrm{g}'(x)\,]\,dx = \mathrm{f}(x)\,\mathrm{g}(x) - \int [\,\mathrm{f}\,'(x)\,\mathrm{g}(x)\,]\,dx.$$

Ou ainda,

$$\int u\,dv = u\,v - \int v\,du$$

Exemplo 01:

Encontrar a integral $\int x\cos(x)\,dx$.

Solução analítica:

Fazemos $u = x$ e $dv = \cos(x)\,dx$. Então $du = dx$ e $v = \text{sen}(x)$. Assim sendo,

$$\int x\cos(x)\,dx = x\,\text{sen}(x) - \int \text{sen}(x)\,dx$$

$$\int x\cos(x)\,dx = x\,\text{sen}(x) - [-\cos(x)]$$

$$\int x\cos(x)\,dx = x\,\text{sen}(x) + \cos(x) + c$$

Solução com Maple:

O **Maple** possui o comando *intparts* que é usado na resolução da integração por partes, o qual faz parte do pacote *student* [ver no *Help* a sintaxe do comando *intparts*].

O comando *intparts* tem dois argumentos: o primeiro é a expressão a ser antidiferenciada, e o segundo é a escolha para *u*, o pedaço que é para ser diferenciado. Nós não necessitamos especificar *dv* , pois faz parte do integrando.

Encontrando $\int x\cos(x)\,dx$. Podemos observar a escolha apropriada para $u = x$.

```
> restart: with(student):
> p1:=Int(x*cos(x),x);
```

$$p1 := \int x\cos(x)\,dx$$

```
> p2:=intparts(p1,x);
```

$$p2 := x\sin(x) - \int \sin(x)\,dx$$

```
> p3:=simplify(p2);
```

$$p3 := x\sin(x) - \int \sin(x)\,dx$$

Nós ainda temos uma integral a resolver, mas é bastante mais simples do que a integral original a ser calculada, e assim usaremos o **Maple** para fazer isto:

```
> p4:=value(p3);
```

$$p4 := x\sin(x) + \cos(x)$$

```
> p1=p4+c;
```

$$\int x\cos(x)\,dx = x\sin(x) + \cos(x) + c$$

Verificação:

Podemos verificar a resposta com o comando **int**.

```
> Int(x*cos(x),x)=int(x*cos(x),x);
```

$$\int x \cos(x)\, dx = x \sin(x) + \cos(x)$$

Devemos ter o hábito de verificarmos o resultado pela diferenciação da antiderivada. Observamos que **p4** é uma expressão, logo devemos usar o comando **diff**.

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}\left(x \sin(x) + \cos(x)\right) = x \cos(x)$$

Exemplo 02:

Encontrar a integral $\int x\, \mathbf{e}^{(-2\, x)}\, dx$.

Solução analítica e com Maple:

Fazendo $u = v$, temos:

```
> p1:=Int(x*exp(-2*x),x);
```

$$p1 := \int x\, \mathbf{e}^{(-2\, x)}\, dx$$

```
> p2:=intparts(p1,x);
```

$$p2 := -\frac{1}{2} x\, \mathbf{e}^{(-2\, x)} - \int -\frac{1}{2} \mathbf{e}^{(-2\, x)}\, dx$$

```
> p3:=simplify(p2);
```

$$p3 := -\frac{1}{2} x\, \mathbf{e}^{(-2\, x)} + \frac{1}{2} \int \mathbf{e}^{(-2\, x)}\, dx$$

Ainda temos uma integral a resolver, mas é bastante mais simples do que a integral original a ser calculada, e desta forma, usaremos o comando **Maple** para avaliar a integral:

```
> p4:=value(p3);
```

$$p4 := -\frac{1}{2} x\, \mathbf{e}^{(-2\, x)} - \frac{1}{4} \mathbf{e}^{(-2\, x)}$$

```
> p1=p4+c;
```

$$\int x\, \mathbf{e}^{(-2\, x)}\, dx = -\frac{1}{2} x\, \mathbf{e}^{(-2\, x)} - \frac{1}{4} \mathbf{e}^{(-2\, x)} + c$$

Verificação:
Podemos verificar a resposta com o comando **int**.

```
> Int(x*cos(x),x)=int(x*cos(x),x);
```

$$\int x\cos(x)\,dx = x\sin(x) + \cos(x)$$

Verificando o resultado pela diferenciação da antiderivada. Observamos que **p4** é uma expressão, logo devemos usar o comando **diff**.

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}\left(-\frac{1}{2}x\,\mathbf{e}^{(-2x)} - \frac{1}{4}\mathbf{e}^{(-2x)}\right) = x\,\mathbf{e}^{(-2x)}$$

Exemplo 03:

Encontrar a integral $\int \ln(x)\,dx$.

Solução analítica e com Maple:

Há alguns bons "truques" que podemos fazer com o comando **intparts**. Por exemplo, um uso padrão do método por partes é o cálculo de $\int \ln(x)\,dx$, escrevendo-o como $\int 1\ln(x)\,dx$ e escolhendo $u = \ln(x)$.

```
> restart: with(student):
> p1:=Int(ln(x),x);
```

$$p1 := \int \ln(x)\,dx$$

```
> p2:=intparts(p1,ln(x));
```

$$p2 := \ln(x)\,x - \int 1\,dx$$

```
> p3:=value(p2);
```

$$p3 := \ln(x)\,x - x$$

```
> p1=p3+C;
```

$$\int \ln(x)\,dx = \ln(x)\,x - x + C$$

Efetuando a verificação por diferenciação:

```
> Diff(p3,x)=diff(p3,x);
```

$$\frac{d}{dx}(\ln(x)\,x - x) = \ln(x)$$

Exemplo 04:

Encontrar a integral $\int x^2\,\mathbf{e}^x\,dx$.

Solução analítica e com Maple:

Em alguns momentos, precisamos utilizar a integração por partes mais de uma vez. Neste exemplo, vamos fazer $u = x^2$ e $dv = \mathbf{e}^x\,dx$. Assim, $du = 2\,x\,dx$ e $v = \mathbf{e}^x$.

```
> restart: with(student):
> p1:=Int(x^2*exp(x),x);
```

$$p1 := \int x^2\,\mathbf{e}^x\,dx$$

```
> p2:=intparts(p1,x^2);
```

$$p2 := x^2\,\mathbf{e}^x - \int 2\,x\,\mathbf{e}^x\,dx$$

Observamos que a nova integral é semelhante à original, porém mais simples, pois o termo em x é do primeiro grau. Portanto, realizaremos uma nova integração por partes, assumindo $u = x$.

```
> p3:=intparts(p2,x);
```

$$p3 := x^2\,\mathbf{e}^x - 2\,x\,\mathbf{e}^x + \int 2\,\mathbf{e}^x\,dx$$

```
> p4:=value(p3);
```

$$p4 := x^2\,\mathbf{e}^x - 2\,x\,\mathbf{e}^x + 2\,\mathbf{e}^x$$

```
> p1=p4+c;
```

$$\int x^2\,\mathbf{e}^x\,dx = x^2\,\mathbf{e}^x - 2\,x\,\mathbf{e}^x + 2\,\mathbf{e}^x + c$$

Efetuando a verificação por diferenciação:

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}(x^2\,\mathbf{e}^x - 2\,x\,\mathbf{e}^x + 2\,\mathbf{e}^x) = x^2\,\mathbf{e}^x$$

Exemplo 05:

Encontrar a integral $\int \mathbf{e}^x \cos(x)\,dx$.

Solução analítica e com Maple:

Em alguns momentos, a integral que desejamos calcular reaparece na sua forma original ou em forma semelhante.
Neste exemplo, vamos fazer $u = \mathbf{e}^x$ e $dv = \cos(x)\,dx$. Assim, $du = \mathbf{e}^x\,dx$ e $v = \text{sen}(x)$.

```
> restart: with(student):
> p1:=Int(exp(x)*cos(x),x);
```

$$p1 := \int \mathbf{e}^x \cos(x)\,dx$$

```
> p2:=intparts(p1,exp(x));
```

$$p2 := \mathbf{e}^x \sin(x) - \int \mathbf{e}^x \sin(x)\,dx$$

Observamos que a nova integral é semelhante à original, porém com o novo integrando sendo a função $\text{sen}(x)$. Portanto, realizaremos uma nova integração por partes, assumindo $u = \mathbf{e}^x$ e $dv = \text{sen}(x)\,dx$. Assim, $du = \mathbf{e}^x\,dx$ e $v = -\cos(x)$.

```
> p3:=intparts(p2,exp(x));
```

$$p3 := \mathbf{e}^x \sin(x) + \mathbf{e}^x \cos(x) + \int -\mathbf{e}^x \cos(x)\,dx$$

Notamos que a integral original reaparece no lado direito da expressão de **p3**. Logo, com uma simples operação algébrica encontramos o valor da integração procurada.

```
> p4:=value(p3);
```

$$p4 := \frac{1}{2}\mathbf{e}^x \sin(x) + \frac{1}{2}\mathbf{e}^x \cos(x)$$

```
> p1=p4+c;
```

$$\int \mathbf{e}^x \cos(x)\,dx = \frac{1}{2}\mathbf{e}^x \sin(x) + \frac{1}{2}\mathbf{e}^x \cos(x) + c$$

Efetuando a verificação por diferenciação:

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}\left(\frac{1}{2}\mathbf{e}^x \sin(x) + \frac{1}{2}\mathbf{e}^x \cos(x)\right) = \mathbf{e}^x \cos(x)$$

Teorema 11: Integração por Partes em Integrais Definidas

A fórmula para calcularmos uma integral definida usando o método por partes é a mesma da integral indefinida, com aplicação dos extremos nas funções $f(x)$ e $g(x)$ e na nova integral, ou seja:

$$\int_a^b f(x)\, g'(x)\, dx = f(x)\, g(x) - \int_a^b f'(x)\, g(x)\, dx$$

Exemplo 06:

Calcular a área sob a curva $y = x\,\mathbf{e}^{(-x)}$ e o eixo-x , no intervalo [0 ; 4] .

Solução analítica e com o Maple:

```
> restart: with(student): with(plots):
```

Inicialmente, vamos definir a função e traçar o seu gráfico, no intervalo dado:

```
> f:=x->x*exp(-x);
```

$$f := x \to x\,\mathbf{e}^{(-x)}$$

```
> graf1:=plot(f(x),x=0..4,y=-0.5..0.5,color=blue,thickness=2
  ):
> graf2:=plot(f(x),x=0..4,y=-0.5..0.5,filled=true,color=turq
  uoise):
> display(graf2,graf1);
```



Fazendo $u = x$, $dv = \mathbf{e}^{(-x)}$, logo $du = dx$ e $v = -\mathbf{e}^{(-x)}$. Então:

```
> p0:=Int(f(x),x=0..4);
```

$$p0 := \int_0^4 x\, \mathbf{e}^{(-x)}\, dx$$

```
> p1:=Int(f(x),x);
```

$$p1 := \int x\, \mathbf{e}^{(-x)}\, dx$$

```
> p2:=intparts(p1,x);
```

$$p2 := -x\, \mathbf{e}^{(-x)} - \int -\mathbf{e}^{(-x)}\, dx$$

Analogamente ao caso do cálculo de integrais definidas por substituição, podemos aplicar os limites de integração no cálculo de $\int -\mathbf{e}^{(-x)}\, dx$ ou após a sua avaliação, conforme segue abaixo:

```
> p3:=value(p2);
```

$$p3 := -x\, e^{(-x)} - e^{(-x)}$$

Neste ponto, vamos proceder a substituição dos limites de integração:

```
> Area:=subs(x=4,p3)-subs(x=0,p3);
```

$$Area := -5\, e^{(-4)} + e^{0}$$

Portanto a área avaliada é de:

```
> 'Area'=evalf(Area);
```

$$Area = 0.9084218056$$

Exercícios propostos

I - Calcular as integrais usando integração por partes

01 - $\displaystyle\int x \operatorname{sen}\left(\frac{x}{2}\right) dx$

02 - $\displaystyle\int x^2 \cos(x)\, dx$

03 - $\displaystyle\int x^3\, e^x\, dx$

04 - $\displaystyle\int (x^2 - 3\,x)\, e^x\, dx$

05 - $\displaystyle\int \operatorname{arcsen}\left(\frac{x}{2}\right) dx$

06 - $\displaystyle\int e^x \operatorname{sen}(x)\, dx$

07 - $\displaystyle\int_0^{\frac{\pi}{2}} x^2 \operatorname{sen}(2\,x)\, dx$

08 - $\displaystyle\int_1^{e} x^3 \ln(x)\, dx$

II - I - Calcular as integrais usando uma integração por substituição e depois por partes.

01 - $\int \text{sen}(\ln(x))\, dx$

02 - $\int_0^1 x(\sqrt{1-x})\, dx$

III - Determinar a área da região delimitada pela curva $y = x\, \text{sen}(x)$ e pelo eixo-x, nos intervalos indicados nos itens a seguir:
a) [0 ; π]
b) [π ; 2 π]
c) [2 π ; 3 π]
d) Encontrar a área geométrica total
e) Encontrar a área líquida total, considerando os sinais

Sugestão: Traçar o gráfico da função no intervalo [2 π ; 3 π] .

Método por Frações Parciais

O método de resolução de integração por frações parciais consiste em escrevermos uma função $f(x)$ como uma função racional $\dfrac{p(x)}{q(x)}$, onde $p(x)$ e $q(x)$ são polinômios e podermos decompor esta fração parcial como uma soma de frações mais simples.

A decomposição de uma função racional $f(x) = \dfrac{p(x)}{q(x)}$ em frações mais simples é dada por:

$$f(x) = \frac{A_1}{x - a_1} + \frac{A_2}{x - a_2} + \dots + \frac{A_n}{x - a_n} \;,$$

onde $A_1, A_2, \dots, A_n$ são constantes a serem determinadas e $x - a_1, x - a_2, \dots, x - a_n$ são fatores lineares ou quadráticos do polinômio denominador $q(x)$.

Por exemplo, vamos considerar a função $f(x) = \dfrac{2x - 1}{x^2 - 5x + 6}$. Esta função racional pode, facilmente, ser decomposta na forma seguinte:

$$\frac{2x-1}{x^2-5x+6} = \frac{2x-1}{(x-2)(x-3)} = \frac{A}{x-2} + \frac{B}{x-3}$$. Nesta situação, devemos encontar as constantes A e B.

Reduzindo-se ao mesmo denominador a expressão acima, obtemos:

$$2x-1 = A(x-3) + B(x-2) = Ax - 3A + Bx - 2B = (A+B)x - 3A - 2B .$$

Comparando-se os termos de mesma potência em *x*, obtemos o sistema de equações lineares seguinte:

$$A + B = 2$$
$$-3A - 2B = -1$$

Resolvendo-se o sistema acima, obtemos: $A = 3$ e $B = -1$.

Logo, $f(x) = \frac{2x-1}{x^2-5x+6} = \frac{3}{x-2} + \frac{-1}{x-3} = \frac{3}{x-2} - \frac{1}{x-3}$

A expressão acima poderá facilmente ser integrada por um dos métodos estudados anteriormente.

O **Maple** tem uma sub-rotina para converter uma função racional própria em frações parciais. Isto pode ser feito com o comando **convert** junto com a opção **parfrac**. Após a decomposição da função *polinomial primitiva* em *frações parciais* , podemos empregar os métodos de integração vistos anteriormente.

Exemplo 01:

Encontrar a integral $\int \frac{1}{x^2-1}\,dx$, usando a decomposição em frações parciais.

Primeiro, vamos definir uma expressão **p1** para a integral indefinida, e então depois encontrarmos as frações parciais para o polinômio

```
> restart: with(student):
> p1:=Int(1/(x^2-1),x);
```

$$p1 := \int \frac{1}{x^2-1}\,dx$$

Definindo a expressão **p2** para o polinômio original.

```
> p2:=1/(x^2-1);
```

$$p2 := \frac{1}{x^2 - 1}$$

Usando os comandos **convert** junto com a opção **parfrac**, decompomos o polinômio em frações parciais:

```
> p3:=convert(p2,parfrac,x);
```

$$p3 := -\frac{1}{2\,(x+1)} + \frac{1}{2\,(x-1)}$$

A expressão de **p3** é uma expressão que apresenta uma integral simplificada, e assim nós podemos avaliá-la.

```
> p4:=p1=Int(p3,x);
```

$$p4 := \int \frac{1}{x^2 - 1}\,dx = \int -\frac{1}{2\,(x+1)} + \frac{1}{2\,(x-1)}\,dx$$

```
> p5:=value(rhs(p4));
```

$$p5 := -\frac{1}{2}\ln(x+1) + \frac{1}{2}\ln(x-1)$$

Ou desta outra forma:

```
> p5:=int(p3,x);
```

$$p5 := -\frac{1}{2}\ln(x+1) + \frac{1}{2}\ln(x-1)$$

A solução geral pode ser facilmente escrita:

```
> p1=p5+c;
```

$$\int \frac{1}{x^2 - 1}\,dx = -\frac{1}{2}\ln(x+1) + \frac{1}{2}\ln(x-1) + c$$

Podemos verificar a resposta pela diferenciação da antiderivada. Observamos que **p4** é uma expressão, assim devemos usar o comando **diff**.

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}\left(-\frac{1}{2}\ln(x+1) + \frac{1}{2}\ln(x-1)\right) = -\frac{1}{2\,(x+1)} + \frac{1}{2\,(x-1)}$$

Utilizando-se do comando **factor**, **normal**, **simplify** podemos encontrar a expressão original, a saber:

```
> factor(rhs(%));
```

$$\frac{1}{(x-1)\,(x+1)}$$

```
> simplify(%);
```

$$\frac{1}{x^2-1}$$

Exemplo 02:

Resolver a integral $\displaystyle\int \frac{x^2+x+1}{x^2-1}\,dx$.

Neste caso, o grau do numerador é igual ao do denominador, então primeiro efetuamos a divisão indicada no integrando. Podemos realizar esta operação com o auxílio do **Maple**, através de comandos para manipulação de polinômios, a saber: **quo** (quociente), **rem** (resto), **factor** (fatoração) , **normal** (normalização), **simplify** (simplificação), ou simplesmente usarmos o comando **convert** com a opção **parfrac**, conforme a seguir:

```
> restart: with(student):
> p1:=Int((x^2+x+1)/(x^2-1),x);
```

$$p1 := \int \frac{x^2+x+1}{x^2-1}\,dx$$

```
> p2:=(x^2+x+1)/(x^2-1);
```

$$p2 := \frac{x^2+x+1}{x^2-1}$$

```
> p3:=convert(p2,parfrac,x);
```

$$p3 := 1 - \frac{1}{2\,(1+x)} + \frac{3}{2\,(x-1)}$$

A expressão de **p3** é uma expressão simplificada que apresenta uma integral simplificada, e assim nós podemos avaliá-la.

```
> p4:=p1=Int(p3,x);
```

$$p4 := \int \frac{x^2+x+1}{x^2-1}\,dx = \int 1 - \frac{1}{2\,(1+x)} + \frac{3}{2\,(x-1)}\,dx$$

```
> p5:=value(rhs(p4));
```

$$p5 := x - \frac{1}{2}\ln(1+x) + \frac{3}{2}\ln(x-1)$$

Verificando o cálculo acima:

```
> Diff(p5,x)=diff(p5,x);
```

$$\frac{d}{dx}\left(x-\frac{1}{2}\ln(1+x)+\frac{3}{2}\ln(x-1)\right)=1-\frac{1}{2\,(1+x)}+\frac{3}{2\,(x-1)}$$

```
> normal(rhs(%));
```

$$\frac{x^2+x+1}{(x-1)\,(1+x)}$$

```
> simplify(%);
```

$$\frac{x^2+x+1}{x^2-1}$$

Exemplo 03:
Usando o método descrito acima encontrrar as seguintes antiderivadas e verificá-las pela diferenciação.

a) $\displaystyle\int\frac{x+1}{(x-1)^2\,(x-2)}\,dx$

Solução com Maple:

```
> restart: with(plots):
> p:=(x+1)/((x-1)^2*(x-2));
```

$$p:=\frac{x+1}{(x-1)^2\,(x-2)}$$

```
> p1:=Int(p,x);
```

$$p1:=\int\frac{x+1}{(x-1)^2\,(x-2)}\,dx$$

```
> p2:=convert(p,parfrac,x);
```

$$p2:=-\frac{3}{x-1}+\frac{3}{x-2}-\frac{2}{(x-1)^2}$$

```
> p3:=Int(p,x)=Int(p2,x);
```

$$p3:=\int\frac{x+1}{(x-1)^2\,(x-2)}\,dx=\int-\frac{3}{x-1}+\frac{3}{x-2}-\frac{2}{(x-1)^2}\,dx$$

```
> p4:=value(rhs(p3));
```

$$p4:=-3\ln(x-1)+3\ln(x-2)+\frac{2}{x-1}$$

Verificação:

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}\left(-3\ln(x-1)+3\ln(x-2)+\frac{2}{x-1}\right)=-\frac{3}{x-1}+\frac{3}{x-2}-\frac{2}{(x-1)^2}$$

```
> normal(rhs(%));
```

$$\frac{x+1}{(x-1)^2\,(x-2)}$$

b) $\displaystyle\int \frac{x^5+x^4-8}{x^3-4\,x}\,dx$

```
> restart: with(student):
> p:=(x^5+x^4-8)/( x^3-4*x);
```

$$p := \frac{x^5+x^4-8}{x^3-4\,x}$$

```
> p1:=Int(p,x);
```

$$p1 := \int \frac{x^5+x^4-8}{x^3-4\,x}\,dx$$

```
> p2:=convert(p,parfrac,x);
```

$$p2 := x^2+x+4-\frac{3}{x+2}+\frac{2}{x}+\frac{5}{x-2}$$

```
> p3:=Int(p,x)=Int(p2,x);
```

$$p3 := \int \frac{x^5+x^4-8}{x^3-4\,x}\,dx = \int x^2+x+4-\frac{3}{x+2}+\frac{2}{x}+\frac{5}{x-2}\,dx$$

```
> p4:=value(rhs(p3));
```

$$p4 := \frac{x^3}{3}+\frac{x^2}{2}+4\,x-3\ln(x+2)+2\ln(x)+5\ln(x-2)$$

Verificação:

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}\left(\frac{x^3}{3}+\frac{x^2}{2}+4\,x-3\ln(x+2)+2\ln(x)+5\ln(x-2)\right)=$$
$$x^2+x+4-\frac{3}{x+2}+\frac{2}{x}+\frac{5}{x-2}$$

```
> normal(rhs(%));
```

$$\frac{x^5+x^4-8}{x\,(x-2)\,(x+2)}$$

c) $\displaystyle\int \frac{x}{(x-1)(x^2+1)}\,dx$

```
> restart: with(student):
> p:=x/((x-1)*(x^2+1));
```

$$p := \frac{x}{(x-1)(x^2+1)}$$

```
> p1:=Int(p,x);
```

$$p1 := \int \frac{x}{(x-1)(x^2+1)}\,dx$$

```
> p2:=convert(p,parfrac,x);
```

$$p2 := \frac{-x+1}{2\,(x^2+1)} + \frac{1}{2\,(x-1)}$$

```
> p3:=Int(p,x)=Int(p2,x);
```

$$p3 := \int \frac{x}{(x-1)(x^2+1)}\,dx = \int \frac{-x+1}{2\,(x^2+1)} + \frac{1}{2\,(x-1)}\,dx$$

```
> p4:=value(rhs(p3));
```

$$p4 := -\frac{1}{4}\ln(x^2+1) + \frac{1}{2}\arctan(x) + \frac{1}{2}\ln(x-1)$$

Verificação:

```
> Diff(p4,x)=diff(p4,x);
```

$$\frac{d}{dx}\left(-\frac{1}{4}\ln(x^2+1) + \frac{1}{2}\arctan(x) + \frac{1}{2}\ln(x-1)\right) =$$

$$-\frac{x}{2\,(x^2+1)} + \frac{1}{2\,(x^2+1)} + \frac{1}{2\,(x-1)}$$

```
> normal(rhs(%));
```

$$\frac{x}{(x-1)(x^2+1)}$$

Exercícios propostos

I - Calcular as integrais usando o método de frações parciais

01 - $\displaystyle\int \frac{1}{x^2-1}\,dx$

02 - $\displaystyle\int \frac{1}{1-x^2}\,dx$

03 - $\displaystyle\int \frac{1}{(x^2-1)^2}\,dx$

04 - $\displaystyle\int \frac{1}{x^2+2\,x}\,dx$

05 - $\displaystyle\int \frac{x+3}{2\,x^3-8\,x}\,dx$

06 - $\displaystyle\int_0^1 \frac{1}{(x+1)\,(x^2+1)}\,dx$

07 - $\displaystyle\int \frac{2\,x+2}{(x^2+1)\,(x-1)^3}\,dx$

08 - $\displaystyle\int \frac{2\,x^3-2\,x^2+1}{x^2-x}\,dx$

09 - $\displaystyle\int \frac{x^4}{x^2-1}\,dx$

10 - $\displaystyle\int \frac{x^4+x^2-1}{x^3+x}\,dx$