Centro Federal de Educação, Ciência e Tecnologia
Campus Central - Natal - Rio Grande do Norte
Diretoria de Educação e Tecnologia da Informação

Curso Tecnólogo em Análise e Desenvolvimento de Sistemas
Curso Tecnólogo em Redes de Computadores

Curso de Cálculo Diferencial e Integral

Prof.: Fco. **Assis** de Oliveira

D E R I V A D A S

Neste curso, estudaremos a derivada. Abordaremos, inicialmente, os conceitos relacionados com a inclinação de uma curva em um ponto e taxas de variação. A seguir, procuraremos mostrar a definição de derivada e as suas propriedades, cálculos de derivadas de funções e aplicações práticas da derivada em diversos ramos do conhecimento, notadamente na Física e Engenharia. Desenvolveremos implementações das derivadas no *software* de computação simbólica Maple.

O sistema principal do **Maple** está preparado para trabalhar com as operações básicas do cálculo, porém devemos carregar para a memória do computador o pacote específico para este fim, ou seja, a biblioteca student, o que deve ser executado com os comandos **with(student)** e/ou with(Student) [Ver o *help* do Maple].

Introdução

Reta Tangente

Uma reta é tangente a um círculo se o encontrar precisamente em um ponto perpendicularmente ao seu raio. Este conceito geométrico pode ser visto na figura a seguir.



A idéia acima não pode ser generalizada quando tratamos com curvas de funções. As figuras seguintes mostram retas que parecem ser retas tangentes, porém não apresentam as caracteristícas de definição destas retas. A primeira figura mostra uma reta que intercepta a curva da função em mais de um ponto; a segunda figura, apesar de interceptar a curva em um único ponto, obviamente não a consideramos uma reta tangente.





Desta forma, precisamos definir precisamente o conceito de reta tangente a uma curva. Inicialmente, vamos introduzir o conceito de inclinação de uma reta e o conceito de retas secantes a curvas.

A figura seguinte mostra os conceitos de inclinação de uma reta e uma reta secante sobre uma curva.
Seja f(x) a curva de uma função parabólica definida em um intervalo aberto e dois pontos P($x1$; $y1$) e **Q**(x2 ; y2) sobre esta curva. Seja S(x) a reta secante que passa pelos dois pontos P e Q.

Vamos considerar o triângulo retângulo PMQ. A inclinação ou coeficiente angular da reta será calculada por:

$$m_S = \tan(A) = \frac{y2 - y1}{x2 - x1} = \frac{dy}{dx} \,.$$

Vamos manter o ponto **P** fixo e fazer o ponto **Q** deslocar-se sobre a curva até atingir o ponto **P**. À medida que o ponto **Q** aproxima-se de **P**, temos que a inclinação da reta secante vai alterando-se, até atingir um *valor limite*, quando o valor *de* $x2$ apromisa-se do valor de $x1$. Chamamos este limite de inclinação da reta tangente da curva no ponto **P** (ver animação de secante mais a frente).



Definição:

Seja f(x) uma função definida no conjunto dos reais em um intervalo aberto. Seja **P** ($x1$; y

1) um ponto sobre a curva e **Q** ($x2$; $y2$) um segundo ponto sobre a curva da função. A inclinação da reta tangente à curva no ponto **P** é dada por

$$m_{x1} = \lim_{Q \to P} \frac{dy}{dx} = \lim_{x2 \to x1} \frac{f(x2) - f(x1)}{x2 - x1}$$, quando este limite existe.

Fazendo $x2 = x1 + dx$ podemos reescrever o limite acima na forma:

$$m_{x1} = \lim_{dx \to 0} \frac{f(x1 + dx) - f(x1)}{dx}$$

A figura a seguir ilustra o limite quando $Q \to P$, mostrando que a inclinação da reta secante vai aproximando-se da inclinação da reta tangente à medida que o ponto **Q** aproxima-se do ponto **P**. As retas secantes estão indicadas em linhas tracejadas vermelhas, enquanto que a reta tangente corresponde à linha cheia na cor magenta.



Conhecendo-se a inclinação da reta tangente à curva no ponto **P**, podemos encontrar a equação da reta tangente à curva neste ponto **P**.

Equação da Reta Tangente

Consideremos que uma função $f(x)$ seja contínua em um ponto **P** de abscissa $x = c$. A **reta tangente** à curva desta função no ponto **P** $(c ; f(c))$ é:

a) a reta que passa pelo ponto **P**, cujo coeficiente angular ou inclinação m_c, seja dada por:

$$m_c = \lim_{dx \to 0} \frac{f(c + dx) - f(c)}{dx} \text{ , se o limite existir;}$$

neste caso, temos a equação seguinte para a reta tangente:

$y - f(c) = m_c (x - c)$, ou seja:

$y = f(c) + m_c (x - c)$.

b) a reta $x = c$, se

$$\lim_{dx \to 0} \frac{f(c + dx) - f(c)}{dx} = \infty \text{ ou } -\infty \text{ .}$$

Caso os dois itens anteriores não forem verdadeiros, então não existirá reta tangente à curva no ponto **P**.

N.B.: No caso b), dizemos que a equação da reta tangente é uma **tangente vertical** .

Exemplo:
Vamos definir a função $f(x)$ usada nos gráfricos acima, conforme comandos **Maple** a seguir::

```
> restart: with(plots): with(student):
> f:=x->-(x-3)^2/2+3;
```

$$f := x \to -\frac{1}{2}(x - 3)^2 + 3$$

```
> f:=unapply(expand(f(x)),x);
```

$$f := x \to -\frac{1}{2}x^2 + 3\,x - \frac{3}{2}$$

Calculando o coeficiente angular ou declividade da reta secante, assumindo as abscissas $x_1 = 1$ para o ponto **P** e $x_2 = 4$ para o ponto **Q**:

```
> m[s]:=(f(4)-f(1))/(4-1);
```

$$m_s := \frac{1}{2}$$

Encontrando a equação para a reta secante que passa pelos pontos **P** e **Q**:

```
> s:=unapply(f(1)+m[s]*(x-1),x);
```

$$s := x \to \frac{1}{2} + \frac{1}{2}x$$

```
> 'retasec(x)'=sort(s(x),x);
```

$$\mathrm{retasec}(x) = \frac{x}{2} + \frac{1}{2}$$

Plotando o gráfico das curvas f(x) e da reta secante:

```
> plot([f(x),s(x)],x=-1..6,y=-1..5,color=[blue,red],thicknes
  s=2);
```



Abaixo, vamos traçar uma sequência de 6 retas secantes, começando na reta que passa pelos pontos **P** e **Q**, até atingir a reta tangente à curva no ponto $x = 1$. Para isto, vamos definir a equação da reta secante em sua forma geral, dependendo das coordenadas x_1 e x_2 e o ponto de tangência, neste caso $x = c = 1$. O controle do número de retas secantes é efetuado através da variável *k*, em uma estrutura lógica de repetição em uma sequência. Vamos, também, definir a equação da reta tangente no ponto $x = c = 1$.

```
> c:=1:
> retasec:=(x1,x2)->f(x1)+((f(x2)-f(x1))/(x2-x1))*(x-x1);
```

$$retasec := (x1, x2) \rightarrow f(x1) + \frac{(f(x2) - f(x1))(x - x1)}{x2 - x1}$$

```
> m[tg]:=limit( (f(c+dx)-f(c))/dx,dx=0 );
```

$$m_{tg} := 2$$

```
> retatan:=x->f(c)+m[tg]*(x-c);
```

$$retatan := x \rightarrow f(c) + m_{tg}(x - c)$$

```
> retatan:=unapply(f(c)+m[tg]*(x-c),x);
```

$$retatan := x \rightarrow -1 + 2x$$

```
> 'retatan(x)'=sort(retatan(x),x);
```

$$\mathrm{retatan}(x) = 2x - 1$$

Definindo gráficos e plotando-os:

```
> graf1:=plot(f(x), x= -1..6,y=-1..6, color=blue,
  thickness=2):
> graf2:=plot({retasec(c,c+k/2) $ k = 1..6 }, x= -1..6,
  color=red):
> graf3:=plot(retatan(x),x=-1..5, color=magenta,
  thickness=2):
> display(graf1,graf2,graf3);
```



A equação para a reta tangente será dada por:

$$y = f(c) + m_c\,(x - c)\,.$$

Considerando que $c = 1$ e $m = \lim_{dx \to 0} \frac{f(c+dx) - f(c)}{dx}$, fazendo uso de comandos **Maple**, então temos:

```
> c:=1; 'f(c)'=f(c);
```

$$c := 1$$

$$f(c) = 1$$

```
> m:=Limit(('f(c+dx)'-'f(c)')/dx,dx=0)=limit((f(c+dx)-f(c))/
  dx,dx=0);
```

$$m := \lim_{dx \to 0} \frac{f(c+dx) - f(c)}{dx} = 2$$

Portanto, podemos escrever:

$$y = f(c) + m\,(x - c) = 1 + 2\,(x - 1)$$

$y = 2x - 1$.

Animações de Retas Secante e Tangente à Curva

Exemplo 1: Animação da Reta Secante à Curva
Abaixo, mostramos uma animação de retas secantes para uma curva definida pela função $f(x)$.
Os comandos seguintes **Maple** criam uma animação para retas secantes.

No procedimento, *n* representa o número de "*frames*", isto é, a quantidade de quadros por segundo, *f* representa a função $f(x)$, *c* o ponto de tangência e *h* o incremento para o ponto da secante.

```
> restart: with(plots): with(student):
> f:=x->-(x-3)^2/2+3;
```

$$f := x \to -\frac{1}{2}(x-3)^2 + 3$$

```
> f:=unapply(expand(f(x)),x);
```

$$f := x \to -\frac{1}{2}x^2 + 3x - \frac{3}{2}$$

```
> AnimaSecante := proc( f, c, h )
     local n, Background, Mover, Secante;
  n:=40:
  Background:=plots[display](plots[pointplot]({[c,f(c)],[
  c,0]},symbol=circle),color=blue,
  plot(f(x),x=-1..6,y=-1..6),color=blue,thickness=2):
  Mover:=plots[display](seq(
  plots[pointplot]({[c+(n-i)/(n/h),f(c+(n-i)/(n/h))],[
  c+(n-i)/(n/h),0]},symbol=circle,color=red),i=0..n-1),in
  sequence=true):
  Secante:=plots[display](seq(plot((f(c+(n-i)/(n/h))-f(c)
  )/(n-i)*(n/h)*(x-c)+f(c),x=-1..6,y=-1..6,color=blue,thi
  ckness=2),i=0..n-1),insequence=true):
  plots[display](Secante,Background,Mover);
  end proc:
```

Para executar a animação selecione o gráfico e use o botão de animação na barra de ferramentas. Explore os recursos dos botões de animação para visualisar a animação passo a passo e de forma contínua. No procedimento, alterando-se a parâmetro *c*, podemos aumentar ou diminuir a quantidade de frames. Tente usar $c = 10$ e $c = 60$ e observe os resultados.

```
> AnimaSecante(f,1,3);
```



N.B.: Código do procedimento adaptado do *site Maplesoft Inc.*, referenciado no texto.

Exemplo 2: Animação da Reta Tangente à Curva

No exemplo abaixo, vamos mostrar uma animação da reta tangente a uma curva. Para isto, utilizaremos a função $f(x) = x^2$. A função $g(x) = 2\,x$ é calculada como a derivada da curva $f(x)$, através do operador **D**(f), atribuindo o resultado a uma função $g(x)$. Este operador e cálculo será discutido posteriormente, quando da definição formal de derivadas e suas propriedades.
As retas tangentes serão posicionadas nos valores inteiros de *x*, com variação de *x* no intervalo de [-5 ; 5]. O controle da animação é efetuado usando-se o comando de repetição **for** ... **do**.
Após a execução dos comandos e gerado o gráfico, com um clique (*click*) sobre o gráfico, é aberto a barra de ferramentas de animação, o qual poderá ser escolhido formas alternativas de apresentação da animação.

```
> restart:
> with(plots): with(student):
```

Definindo função e calculando a função derivada:

```
> f:=x->x^2;
```

$$f := x \to x^2$$

```
> g:=D(f);
```

$$g := x \to 2\,x$$

Estruturando a animação da tangente à curva:

```
> for i from -5 to 5 do
      fp[i]:=
  plot([f(x),f(i)+g(i)*(x-i)],x=-5..5,y=-5..25,color=[red
  ,blue],xtickmarks=[-5,-4,-3,-2,-1,0,1,2,3,4,5],ytickmar
  ks=[-5,0,1,4,9,16,25],thickness=2):

  pp[i]:=pointplot({[i,f(i)]},symbol=circle,color=red):

  xp[i]:=pointplot({[i,0],[i,f(i)]},connect=true,linestyl
  e=3,color=red):
      pic[i]:=display({fp[i],pp[i],xp[i]}):
  od:
```

```
> display([seq(pic[j],j=-5..5)],insequence=true,title=`A
  linha azul é a tangente à curva no ponto vermelho`);
```



Exercícios propostos

I - Encontre a equação da reta tangente para as funções nos pontos indicados e traçe os gráficos das funções e das retas tangentes, em um mesmo sistema de eixos, nos itens abaixo:

a) $f(x) = 3\,x - x^2$, $x = 2$

b) $g(x) = x^3 - \sqrt{x}$, $x = 1$

c) $h(x) = 1 + 3\cos(x)$, $x = \pi/6$

d) $k(x) = \dfrac{3 + x}{1 - x^2}$, $x = 0$

Definição de Derivadas

No estudo das retas tangentes apresentado anteriormente vimos que

$$m = \lim_{dx \to 0} \frac{\mathrm{f}(c + dx) - \mathrm{f}(c)}{dx}$$

é a inclinação da reta tangente à curva $\mathrm{f}(x)$, no ponto $x = c$, se o limite existe. Este limite é tão importante que denominamos de **derivada** de *f* em *c*, o qual denotamos por:

$$\mathrm{f}\,'(c) = \lim_{dx \to 0} \frac{\mathrm{f}(c + dx) - \mathrm{f}(c)}{dx},$$

onde f '(x) (lê-se f linha de x) pode ser visto como uma função cuja entrada é o ponto c e a saída (imagem) é o número f '(c).

A expressão acima, também pode ser escrita como

$$\mathrm{f}\,'(x_1) = \lim_{x_2 \to x_1} \frac{\mathrm{f}(x_2) - \mathrm{f}(x_1)}{x_2 - x_1}$$

Definição

Definimos a função f ' expressa por

$$\mathrm{f}\,'(x) = \lim_{h \to 0} \frac{\mathrm{f}(x + h) - \mathrm{f}(x)}{h}$$

como a **derivada de f em relação a *x***. O domínio de f ' consiste de todos os valores de x do domínio de f para os quais o limite existe.

Analogamente, podemos, alternativamente, expressar a derivada como

$$\mathrm{f}\,'(x) = \lim_{z \to x} \frac{\mathrm{f}(z) - \mathrm{f}(x)}{z - x}$$

Dizemos que uma função é derivável quando existe a derivada em todos os pontos do seu domínio.

Ao processo para calcular uma **derivada** denominamos de ***derivação***.

A seguir, apresentamos outras notações para derivadas:

a) $\mathrm{D}_x\mathrm{f}(x)$ - lê-se derivada de $\mathrm{f}(x)$ em relação a x.

b) $D_x y$ - lê-se derivada de y em relação a x.

c) $\frac{dy}{dx}$ - lê-se derivada de y em relação a x.

d) $\frac{d}{dx}$ f(x) - lê-se derivada de f(x) em relação a x.

Exemplos

A seguir, apresentamos alguns exemplos de cálculos de derivadas, utilizando-se de sua definição.

Exemplo 01

Encontre a derivada da função y = f(x) = 2 x + 3, fazendo uso de sua definição.

Solução analítica:

$$f'(x) = \lim_{h \to 0} \frac{f(x+h) - f(x)}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{[2(x+h)+3] - [2x+3]}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{2x + 2h + 3 - 2x - 3}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{2h}{h} = \lim_{h \to 0} 2$$

f '(x) = 2.

Exemplo 02

Encontre a derivada em relação a x, de $f(x) = x^2 - 1$ e use-a para determinar a equação da reta tangente à curva no ponto $x = 2$.

Solução analítica:

$$f'(x) = \lim_{h \to 0} \frac{f(x+h) - f(x)}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{[(x+h)^2 - 1] - [x^2 - 1]}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{[x^2 + 2xh + h^2 - 1] - [x^2 - 1]}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{[x^2 + 2xh + h^2 - 1] - x^2 + 1}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{x^2 + 2xh + h^2 - 1 - x^2 + 1}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{2xh + h^2}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{h(2x + h)}{h}$$

$$f'(x) = \lim_{h \to 0} [2x + h]$$

$$f'(x) = \lim_{h \to 0} 2x$$

$$f'(x) = 2x$$

Desta forma, a inclinação da reta tangente à curva $y = x^2 - 1$, no ponto $x = 2$, é m = f '(2) = $2 \cdot 2 = 4$.

A imagem da função no ponto $x = 2$ é $f(2) = 2^2 - 1 = 3$.

Como a equação para a reta tangente é dada por $y = f(c) + m_c\,(x - c)$, então temos:

$y = 3 + 4\,(x - 2) = 4x - 5.$

Em geral, se f '(x) está definida em $x = c$, então a expressão para encontrar a equação da reta tangente é dada por:

$$\boldsymbol{y = f(c) + f'(c) \cdot (x - c)}$$

Utilizando-se do **Maple**, vamos analisar graficamente a curva e a reta tangente no ponto $x = 2$:

```
> restart: with(plots): with(plottools):
> f:=x->x^2-1;
```

$$f := x \to x^2 - 1$$

```
> g:=x->4*x-5;
```

$$g := x \to 4\,x - 5$$

```
> graf1:=plot(f(x),x=-4..4,y=-2..8,color=blue,thickness=2
,legend="f (x) = x^2 - 1"):
> graf2:=plot(g(x),x=-4..4,y=-2..8,color=red,thickness=1,
legend="tangente = 4x - 5"):
> graf3:=line([2,0],[2,f(2)],color=black,linestyle=2,thic
kness=1):
> graf4:=line([0,f(2)],[2,f(2)],color=black,linestyle=2,t
hickness=1):
> graf5:=textplot([4.2,7,"f'(x) = 4x - 5"],color=red):
> graf6:=textplot([-1.6,7,"f (x) = x^2 - 1"],color=blue):
> display(graf2,graf1,graf3,graf4);
```



Exemplo 03

Encontre a derivada da função y = $f(x) = \dfrac{x-2}{x+3}$, fazendo uso de sua definição.

Solução analítica:

$$f'(x) = \lim_{h \to 0} \frac{f(x+h) - f(x)}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{\dfrac{[x+h]-2}{[x+h]+3} - \dfrac{x-2}{x+3}}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{\dfrac{x+h-2}{x+h+3} - \dfrac{x-2}{x+3}}{h}$$

$$f'(x) = \lim_{h \to 0} \frac{[(x+h-2)(x+3) - (x-2)(x+h+3)]}{[(x+h+3)(x+3)]\,h}$$

$$f'(x) = \lim_{h \to 0} \frac{[x^2 + 3x + hx + 3h - 2x - 6] - [x^2 + hx + 3x - 2x - 2h - 6]}{[(x+h+3)(x+3)]\,h}$$

$$f'(x) = \lim_{h \to 0} \frac{5h}{[(x+h+3)(x+3)]\,h}$$

$$f'(x) = \lim_{h \to 0} \frac{5}{[(x+h+3)(x+3)]}$$

$$f'(x) = \frac{5}{(x+3)^2}$$

Exemplo 04

Encontre a derivada de $f(x) = \sqrt{x}$, para $0 < x$, fazendo uso da definição.

Neste exemplo, vamos fazer uso da expressão alternativa para cálculo da derivada. Logo, temos:

$$f'(x) = \lim_{z \to x} \frac{f(z) - f(x)}{z - x}$$

$$f'(x) = \lim_{z \to x} \frac{\sqrt{z} - \sqrt{x}}{z - x}$$

$$f'(x) = \lim_{z \to x} \frac{\sqrt{z} - \sqrt{x}}{z - x}$$

$$f'(x) = \lim_{z \to x} \frac{\sqrt{z} - \sqrt{x}}{(\sqrt{z} - \sqrt{x})\,(\sqrt{z} + \sqrt{x})}$$

$$f'(x) = \lim_{z \to x} \frac{1}{\sqrt{z} + \sqrt{x}}$$

$$f'(x) = \frac{1}{2\sqrt{x}}$$

Comandos Maple para Derivadas

O sistema principal do **Maple** está preparado para trabalhar com as operações básicas do cálculo, porém devemos carregar para a memória do computador o pacote específico para este fim, ou seja, a biblioteca student, o que deve ser executado com os comandos **with(student)** e/ou **with(Student)**. [Ver *help* do **Maple**].

É fácil escrever um procedimento **Maple** que calcula a derivada de uma função usando a definição. O comando **unapply** é usado para converter uma expressão para uma função nos comandos da estrutura de procedimento, conforme exemplo a seguir:

```
> restart: with(student):
> derivada:=proc(f) unapply(limit((f(x+h)-f(x))/h,h=0),x)
  end:
```

Vamos definir a função $f(x) = x^2 - 1$ e o cálculo da derivada atribuí-lo a uma função $g(x)$.

```
> f:=x-> x^2-1;
```

$$f := x \to x^2 - 1$$

```
> g:=derivada(f);
```

$$g := x \to 2\,x$$

Podemos calcular, também, de forma direta a derivada de uma certa função na chamada do procedimento definido anteriormente, a saber:

```
> h:=derivada(sin);
```

$$h := x \to \cos(x)$$

Neste exemplo, foi calculado a derivada da função seno(x), a qual é uma função fundamental implícita do **Maple**.

N.B.: As derivadas de funções fundamentais serão analisadas em tópicos posteriores.

O **Maple** possui dois comandos internos para calcular ou avaliar a derivada de uma função ou expressão algébrica: **D(f)** ou **diff(f(*x*), *x*)** .

O comando **diff(f(*x*) , *x*)**, quando escrito no formato **Diff(f (*x*) , *x*)** produz o resultado inerte, ou seja, não avalia ou calcula a derivada, apenas apresenta a notação da derivada: $\frac{d}{dx}\mathrm{f}(x)$.

O comando **D(f)** diferencia uma função, e a derivada de uma função é também uma função. Este comando pode ser usado para avaliar a derivada num ponto, conforme exemplos seguinte:

```
> g:=x->x^3;
```

$$g := x \to x^3$$

```
> h:=D(g);
```

$$h := x \to 3\,x^2$$

```
> D(g)(2);
```

$$12$$

```
> h(2);
```

$$12$$

Conforme visto anteriormente, o **Maple** possui um outro comando que é usual na prática da Matemática, o qual é equivalente à notação $\frac{d}{dx}$. O comando **diff** diferencia uma expressão e tem como resultado uma outra expressão. O comando **diff** necessita especificar a variável independente para a derivação. Abaixo, mostramos exemplos:

Derivada da expressão: $2\,x^3$, em relação à x

```
> diff(2*x^3,x);
```

$$6\,x^2$$

Derivada da expressão: $2\,x^2 + x$, em relação à x

```
> diff(2*x^2+x,x);
```

$$4\,x+1$$

Derivada da expressão: $2\,x^2+x$, em relação à y

```
> diff(2*x^2+x,y);
```

$$0$$

N.B.: Lembramos que a derivada de uma função é uma outra função, e a derivada de uma expressão é uma outra expressão, e que elas são computadas com os operadores **D** e **diff**, respectivamente. Isto é um bom exemplo do **Maple,** o qual leva-nos a pensar claramente. A distinção entre diferenciação de funções e expressões não foi inventada pelos programadores do **Maple**: ela é um fato real, e é refletida no fato que nós temos ambas as notações (') e $\frac{d}{dx}$ para a diferenciação, mas ela é usualmente encoberta ou ignorada na maioria dos livros textos.

Exemplo 01:

Fazendo uso de comandos **Maple**, encontrar as equações das retas tangente e normal (perpendicular) à curva da função $f(x)=x^2-1$, no ponto $x=-2$.

```
> restart:
> with(plots): with(student):
```

Definindo a função no **Maple**:

```
> f:=x->x^2-1;
```

$$f:=x\to x^2-1$$

Encontrando a função derivada de $f(x)$:

```
> g:=(D(f));
```

$$g:=x\to 2\,x$$

Escrevendo a equação da reta tangente à curva no pont $x=-2$:

```
> reta_tan:=f(-2)+g(-2)*(x+2);
```

$$reta\_tan:=-5-4\,x$$

```
> 'reta_tan(x)'=sort(reta_tan,x);
```

$$\mathrm{reta\_tan}(x)=-4\,x-5$$

Determinando o coeficiente angular da reta normal à reta tangente

```
> n:=(-1)/(-4);
```

$$n:=\frac{1}{4}$$

Escrevendo a equação da reta normal à curva no ponto $x=-2$:

```
> reta_normal:=f(-2)+n*(x+2);
```

$$reta\_normal := \frac{7}{2} + \frac{x}{4}$$

```
> 'reta_normal(x)'=sort(reta_normal,x);
```

$$\text{reta\_normal}(x) = \frac{x}{4} + \frac{7}{2}$$

Traçando os gráficos das curvas:

```
> plot([reta_tan,reta_normal,f(x)],x=-3..3,y=-2..6,scaling=c
  onstrained,color=[red,tan,blue],legend=["Reta
  Tangente","Reta Normal","Função f ( x ) "]);
```



Exemplo 02:

Fazendo uso de comandos **Maple**, encontrar as equações das retas tangentes horizontais à curva da função $x^4 - 2x^2 + 2$. Traçar os gráficos das curvas.

```
> restart: with(plots): with(plottools): with(student):
> f:=x->x^4-2*x^2+2;
```

$$f := x \to x^4 - 2x^2 + 2$$

```
> plot(f(x),x=-3..3,y=-1..5);
```



Encontrando a equação da função derivada:

```
> g:=unapply(diff(f(x),x),x);
```

$$g := x \rightarrow 4\,x^3 - 4\,x$$

A resolução da equação g(x) = 0 fornece-nos os pontos de tangências horizontais (retas paralelas ao eixo-x)

```
> solve(g(x)=0);
```

$$0, 1, -1$$

Portanto, as retas tangentes horizontais "tangenciam" à curva nos pontos $x_1 = -1$, $x_2 = 0$ e $x_3 = 1$.
Aplicando estes pontos à função f(x) temos as equações das retas horizontais:

```
> y[1]:=f(-1); y[2]:=f(0); y[3]:=f(1);
```

$$y_1 := 1$$

$$y_2 := 2$$

$$y_3 := 1$$

```
> graf1:=plot(f(x),x=-3..3,y=-1..5):
> graf2:=line([-2,1],[2,1],color=black,linestyle=1):
> graf3:=line([-1,2],[1,2],color=black,linestyle=1):
> display(graf1,graf2,graf3);
```



Animação da Reta Tangente à Curva

Equação da Reta Tangente à Curva

Vamos considerar uma função polinomial dada pela expressão $\mathrm{f}(x) = x^3 - 5\,x^2 + x + 10$.

Demonstraremos o significado da reta tangente a uma curva através de uma animação, em um intervalo fechado de [-5 ; 5]. Neste exemplo, fazemos uso dos comandos **Maple** associados à derivação de funções.

```
> restart: with(plots): with(student):
> f:=x->x^3-5*x^2+x+10;
```

$$f := x \rightarrow x^3 - 5\,x^2 + x + 10$$

```
> T:=(x,a)->f(a)+(x - a)*D(f)(a):
> display(plot(f(x), x=-2..5, y=-8..12, thickness = 3,
  color = red),
          animate(T(x,t), x=-2..5, t=-1..5, view=-8..12,
```

```
color = blue, frames=60));
```



Determinando a equação da reta tangente para a função $f(x)$ nos pontos $x = -1$ e $x = 1$.

Para $x = -1$ temos:

```
> f(-1);
```

$$3$$

Assim, temos o ponto **A**(-1,3).

A declividade da reta será dada pela derivada da função no ponto $x = -1$.

A derivada da função $f(x)$ será uma função $g(x)$, expressa por:

```
> g:=D(f);
```

$$g := x \to 3\,x^2 - 10\,x + 1$$

O valor da derivada no ponto nos dá:

```
> m:=g(-1);
```

$$m := 14$$

A equação da reta tangente será calculada pela expressão: $y = y_0 + m\,(x - x_0)$, onde (x_0 , y_0) é o ponto dado e m é a declividade da reta. Assim, temos:

```
> reta_tangente=f(-1)+m*(x-(-1));
```

$$reta\_tangente = 17 + 14\,x$$

Para o ponto $x = 1$, temos:

```
> reta_tangente=f(1)+g(1)*(x-1);
```

$$reta\_tangente = 13 - 6\,x$$

Procedimento showtangent

O **Maple** possui o procedimento *showtangent*, constituinte do "pacote" *student*, que traça a tangente de uma curva, em um ponto especificado. A seguir, mostramos aplicação deste comando.

Seja a função $\mathrm{f}(x) = x^2 - 1$ definida no **Exemplo 01** acima. Vamos traçar a curva da função e traçar a reta tangente à curva no ponto $x = 2$.

```
> restart: with(plots): with(student):
> f:=x->x^2-1;
```

$$f := x \to x^2 - 1$$

```
> showtangent(f(x),x=2,x=-3..3,y=-1..10,thickness=[1,2,1]
,color=[red,blue]);
```



Exercícios propostos

I - Fazendo uso da definição de derivadas, encontre a derivada para as funções seguintes:

01 - $f(x) = 1 - x^2$

02 - $f(x) = 2x^3 - x$

03 - $f(x) = \dfrac{1 - x}{2x}$

04 - $f(x) = \dfrac{x - 1}{2 - x}$

II - Use a definição de derivadas para encontrar $f'(x)$, e encontre a equação da reta tangente ao gráfico da função no ponto $x = c$ especificado.

01 - $f(x) = x^3$, no ponto $c = 0$.

02 - $f(x) = \dfrac{1}{x}$, nos pontos $c = 3$ e $c = \dfrac{1}{3}$.

03 - $f(x) = \sqrt{x + 1}$, no ponto $c = 8$.

III - Usando a definição alternativa para derivadas $f'(x) = \lim\limits_{z \to x} \dfrac{f(z) - f(x)}{z - x}$ encontre a derivada das funções seguintes:

01 - $f(x) = \dfrac{1}{x + 2}$

02 - $f(x) = \dfrac{1}{(x - 1)^2}$

03 - $f(x) = \dfrac{x}{x - 1}$

04 - $f(x) = 1 + \sqrt{x}$

IV - Encontre as equações das retas tangente e normal à curva da função $f(x) = x^2 - 2x + 1$, no ponto $x = -2$. Esboce os gráficos das curvas.

V - Encontre a equação da reta tangente à curva da função $f(x) = x^3 - 1$, que seja perpendicular à reta $y = -x$. Esboce os gráficos das curvas.

Regras de Diferenciação

Nas seções anteriores mostramos a derivada de uma função como um limite e usamos esse limite para encontrar algumas derivadas simples. Nesta seção, mostraremos algumas regras ou teoremas, os quais nos possibilitarão encontrar derivadas de funções ou expressões de uma forma mais fácil. Utilizaremos a notação $\frac{d}{dx}\, f(x)$ para expressarmos a derivada de uma função $f(x)$, e quando necessário e apropriado faremos uso dos recursos computacionais do **Maple**.

Teorema 01 - Derivada de uma constante

Seja $f(x) = c$. A derivada de uma função constante é 0 (zero), isto é, se c for um número real qualquer, então

$$\frac{d}{dx}\, f(x) = \frac{d}{dx}\, [\, c\,] = 0$$

Exemplo 01:

Seja $f(x) = -5$.

Então $\frac{d}{dx}\, f(x) = \frac{d}{dx}\, (-5) = 0$

Usando o **Maple**:

```
> Diff(-5,x)=diff(-5,x);
```

$$\frac{d}{dx}\, (-5) = 0$$

Teorema 02 - Derivadas de potências inteiras de *n*

Seja n um inteiro positivo e $f(x) = x^n$. A derivada de x elevado a uma potência inteira é o produto do expoente inteiro por x elevado à potência inteira subtraída de uma unidade. Ou seja:

$$\frac{d}{dx}\, f(x) = \frac{d}{dx}\, [\, x^n\,] = n\, x^{(n-1)}$$

Exemplo 02:

Seja $f(x) = x^3$.

Então $\frac{d}{dx} f(x) = \frac{d}{dx}(x^3) = 3\,x^2$.

Usando o **Maple**:

```
> Diff(x^3,x)=diff(x^3,x);
```

$$\frac{d}{dx}(x^3) = 3\,x^2$$

Embora o **Teorema 02 - potência inteira de *n*** seja aplicado à potências inteira de *n*, pode-se ser mostrado que ele também vale para potências inteiras negativas e potências de qualquer expoente real. Vejamos o exemplo seguinte:

Exemplo 02a:

Seja $f(x) = x^{(-3)}$.

Então $\frac{d}{dx} f(x) = \frac{d}{dx}[x^{(-3)}] = (-3)\,x^{(-3-1)} = (-3)\,x^{(-4)} = -\frac{3}{x^4}$.

Usando o **Maple**:

```
> Diff(x^(-3),x)=diff(x^(-3),x);
```

$$\frac{d}{dx}\left(\frac{1}{x^3}\right) = -\frac{3}{x^4}$$

Teorema 03 - Derivada do produto de uma constante por uma função

Sejam *c* uma constante real e f(*x*) uma função qualquer diferenciável em *x*. A derivada do produto da constante pela função é o produto da constante pela derivada da função. Ou seja:

$$\frac{d}{dx}[c\,f(x)] = c\left[\frac{d}{dx} f(x)\right]$$

Exemplo 03:

Seja $f(x) = 3\,x^2$.

Então $\frac{d}{dx} f(x) = 3\left[\frac{d}{dx}(x^2)\right] = 3\,.\,2\,.\,\mathrm{x} = 6\,.\,x$

Usando o **Maple**:

```
> Diff(3*x^2,x)=diff(3*x^2,x);
```

$$\frac{d}{dx}(3\,x^2) = 6\,x$$

Teorema 04: Derivada da soma (ou diferença) de duas funções

Sejam $f(x)$ e $g(x)$ duas funções deriváveis em x e $h(x) = f(x) + g(x)$. Então a derivada da soma (ou diferença) de duas funções é a soma (ou diferença) das derivadas das funções. Ou seja:

$$\frac{d}{dx}h(x) = \frac{d}{dx}[\,f(x) + g(x)\,] = \left[\frac{d}{dx}f(x)\right] + \left[\frac{d}{dx}g(x)\right]$$

$$\frac{d}{dx}h(x) = \frac{d}{dx}[\,f(x) + g(x)\,] = \left[\frac{d}{dx}f(x)\right] + \left[\frac{d}{dx}g(x)\right]$$

Se $u_1, u_2, u_3, \ldots, u_n$ são funções deriváveis em x, então $u_1 + u_2 + u_3 + , \ldots , + u_n$ também será, e:

$$\frac{d}{dx}[\,u_1 + u_2 + u_3 + , \ldots , + u_n\,] = \left[\frac{d}{dx}[u_1]\right] + \left[\frac{d}{dx}[u_2]\right] + \ldots + \left[\frac{d}{dx}[u_n]\right].$$

Exemplo 04: Derivada de um polinômio

Seja $p(x) = 2\,x^3 - \frac{x^2}{2} + 1$.

Então $\frac{d}{dx}p(x) = \frac{d}{dx}\left[2\,x^3 - \frac{x^2}{2} + 1\right] = \left[\frac{d}{dx}[2\,x^3]\right] - \left[\frac{d}{dx}\left[\frac{x^2}{2}\right]\right] + \left[\frac{d}{dx}1\right] = 6\,x^2 - x.$

Usando o **Maple**:

```
> Diff(2*x^3-x^2/2+1,x)=diff(2*x^3-x^2/2+1,x);
```

$$\frac{d}{dx}\left(2\,x^3 - \frac{1}{2}x^2 + 1\right) = 6\,x^2 - x$$

Teorema 05: Derivada do produto de duas funções

Sejam $f(x)$ e $g(x)$ duas funções deriváveis em x e $h(x) = f(x)\,.\,g(x)$. Então a derivada do produto de duas funções é o produto da primeira função pela derivada da segunda função somado com o produto da segunda função pela derivada da primeira função. Ou seja:

$$\frac{d}{dx}h(x) = \frac{d}{dx}[\,f(x)\,g(x)\,] = f(x)\left[\frac{d}{dx}g(x)\right] + \left[\frac{d}{dx}f(x)\right]g(x)$$

Sejam duas funções f e g deriváveis em x. Em notação de linha, então temos:

$$[f \cdot g]' = f \cdot g' + g \cdot f'.$$

N.B.: Observe que a derivada do produto de duas funções <u>não</u> é o produto das derivadas destas funções.

Exemplo 05:

Sejam os polinômios $f(x) = x^2 - 1$ e $g(x) = x^2 + 1$.

Então $\frac{d}{dx}[f\,g] = [x^2 - 1]\left[\frac{d}{dx}[x^2 + 1]\right] + [x^2 + 1]\left[\frac{d}{dx}[x^2 - 1]\right]$.

$$\frac{d}{dx}[f\,g] = [x^2 - 1]\,[2\,x] + [x^2 + 1]\,[2\,x].$$

$$\frac{d}{dx}[f\,g] = 2\,x^3 - 2\,x + 2\,x^3 + 2\,x.$$

$$\frac{d}{dx}[f\,g] = 4\,x^3.$$

Sabemos que $f(x)\,g(x) = (x^2 - 1)\,(x^2 + 1) = x^4 - 1$. Aplicando-se o teorema anterior ao resultado obtemos:

$$\frac{d}{dx}[f\,g] = \frac{d}{dx}[x^4 - 1] = 4\,x^3.$$

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->x^2-1; g:=x->x^2+1;
```

$$f := x \to x^2 - 1$$

$$g := x \to x^2 + 1$$

```
> Diff('f(x)'*'g(x)',x)=diff(f(x)*g(x),x);
```

$$\frac{d}{dx}(f(x)\,g(x)) = 2\,x\,(x^2 + 1) + 2\,(x^2 - 1)\,x$$

```
> Diff('f(x)'*'g(x)',x)=expand(rhs(%));
```

$$\frac{d}{dx}(f(x)\,g(x)) = 4\,x^3$$

Teorema 06: Derivada do quociente de duas funções

Sejam $f(x)$ e $g(x)$ duas funções deriváveis em x e $h(x) = \frac{f(x)}{g(x)}$, com $g(x) \neq 0$. Então a derivada do quociente das funções é o denominador vezes a derivada do numerador menos o numerador vezes a derivada do denominador, divididos pelo quadrado do denominador. Ou seja:

$$\frac{d}{dx} h(x) = \frac{d}{dx}\left[\frac{f(x)}{g(x)}\right] = \frac{g(x)\left[\frac{d}{dx} f(x)\right] - f(x)\left[\frac{d}{dx} g(x)\right]}{[g(x)]^2}$$

Sejam duas funções f e g deriváveis em x. Em notação de linha, então temos:

$$[f/g]' = [g \cdot f' - f \cdot g']/g^2$$

N.B.: Observe que a derivada do quociente de duas funções não é o quociente das derivadas destas funções.

Exemplo 06:

Sejam os polinômios $f(x) = x$ e $g(x) = x + 1$.

Então $\frac{d}{dx}\left[\frac{f}{g}\right] = \frac{(x+1)\left[\frac{d}{dx}[x]\right] - x\left[\frac{d}{dx}[x+1]\right]}{(x+1)^2}$.

$$\frac{d}{dx}\left[\frac{f}{g}\right] = \frac{(x+1)\,1 - x\,[1]}{(x+1)^2} .$$

$$\frac{d}{dx}\left[\frac{f}{g}\right] = \frac{x+1-x}{(x+1)^2} .$$

$$\frac{d}{dx}\left[\frac{f}{g}\right] = \frac{1}{(x+1)^2} .$$

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->x; g:=x->x+1;
```

$$f := x \to x$$

$$g := x \to x + 1$$

```
> Diff('f(x)'/'g(x)',x)=diff(f(x)/g(x),x);
```

$$\frac{d}{dx}\left(\frac{f(x)}{g(x)}\right)=\frac{1}{x+1}-\frac{x}{(x+1)^2}$$

```
> Diff('f(x)'/'g(x)',x)=expand(rhs(%));
```

$$\frac{d}{dx}\left(\frac{f(x)}{g(x)}\right)=\frac{1}{x+1}-\frac{x}{(x+1)^2}$$

```
> Diff('f(x)'/'g(x)',x)=normal(rhs(%));
```

$$\frac{d}{dx}\left(\frac{f(x)}{g(x)}\right)=\frac{1}{(x+1)^2}$$

Teorema 07 - Derivadas de segunda ordem e ordem superior

Se $f(x)$ é uma função derivável, então a sua derivada também é uma função. Se $f'(x)$ for uma função derivável, então poderemos derivá-la e obtemos, desta forma, uma nova função, denotada por $f''(x)$. A esta função chamamos de derivada segunda ou derivada de ordem 2 de $f'(x)$.

Abaixo, apresentamos a notação para esta derivada:

$$f''(x)=\frac{d^2y}{dx^2}=\frac{d}{dx}\left[\frac{dy}{dx}\right]=\frac{d}{dx}[y']=y''=Dx^2\,f(x)$$

A derivada de ordem n (ou enésima derivada) será representada, conforme segue:

$$y^{[n]}=\frac{d}{dx}[y^{(n-1)}]=\frac{d^n y}{dx^n}=\ Dx^n[y].$$

Exemplo 07:

O exemplo seguinte mostra a operação de derivação de ordem 4 para a função y = $f(x)=x^3-x^2+1$.

==> primeira ordem: $y'=3x^2-2x$;
==> segunda ordem: $y''=6x-2$;
==> terceira ordem: $y'''=6$;
==> quarta ordem: $y^{[4]}=0$.

As demais derivadas de ordem sucessivas serão todas iguais a zero.

O **Maple** utiliza a seguinte sintaxe **diff(f(x), x $n)**, para avaliar uma derivada de ordem *n*, ou seja: $\left[\frac{d^n}{dx^n}[f(x)]\right]$.

A seguir, podemos comprovar a avaliação do exemplo acima:

```
> y:=x^3-x^2+1;
```

$$y:=x^3-x^2+1$$

```
> Diff(y,x $3)=diff(y,x $3);
```

$$\frac{d^3}{dx^3}(x^3 - x^2 + 1) = 6$$

Outro exemplo, usando uma função:

```
> f:=x->x^4-x^3+x^2-1;
```

$$f := x \to x^4 - x^3 + x^2 - 1$$

```
> Diff('f(x)',x $3)=diff(f(x),x $3);
```

$$\frac{d^3}{dx^3}\mathrm{f}(x) = 24\,x - 6$$

Analogamente, o operador **D**, também avalia derivadas de ordem superior. A sintaxe seguinte é usada para avaliar as derivadas sucessivas de uma função: **(D@@n)(f)** ou **D[1$n](f)**.

A seguir, comprovamos a derivada de ordem 3 para a função f(x) anterior:

```
> (D@@3)(f); # avalia a derivada terceira e atribui o resultado
  como função
```

$$x \to 24\,x - 6$$

```
> D[1$3](f); # forma alternativa ao comando acima
```

$$x \to 24\,x - 6$$

```
> D(D(D(f))); # forma alternativa aos comandos acima
```

$$x \to 24\,x - 6$$

Teorema 08 - Derivada da função exponencial $\mathbf{e}^x$

Seja $\mathrm{f}(x) = \mathbf{e}^x$ a função exponencial, então a derivada da função é a própria função. Ou seja:

$$\frac{d}{dx}(\mathbf{e}^x) = \mathbf{e}^x$$

Exemplo 08:

Calcular a derivada da função $\mathrm{f}(x) = 2\,\mathbf{e}^x$.

Temos que $\frac{d}{dx}[2\,\mathbf{e}^x] = 2\left[\frac{d}{dx}(\mathbf{e}^x)\right] = 2\,\mathbf{e}^x$.

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->2*exp(x);
```

$$f := x \to 2\,\mathbf{e}^x$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\,\mathrm{f}(x) = 2\,\mathbf{e}^{x}$$

N.B.: A função exponencial é a única função em que a derivada é a própria função.

Teorema 09 - Derivada da função logarítmica ln(x)
Seja $\mathrm{f}(x) = \ln(x)$ a função logarítmica natural, então a derivada da função é o inverso de x.. Ou seja:

$$\frac{d}{dx}\left[\ln(x)\right] = \frac{1}{x}$$

Exemplo 09:
Calcular a derivada da função $\mathrm{f}(x) = 2\ln(x)$.

Temos que $\frac{d}{dx}\left[2\ln(x)\right] = 2\left[\frac{d}{dx}\left[\ln(x)\right]\right] = \frac{2}{x}$.

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->2*ln(x);
```

$$f := x \to 2\ln(x)$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\,\mathrm{f}(x) = \frac{2}{x}$$

Teorema 10 - Derivada das funções trigonométricas

Teorema 10.1 - Derivada da função seno
Seja $\mathrm{f}(x) = \mathrm{sen}(x)$ a função seno; então a sua derivada é a função cosseno. Ou seja:

$$\frac{d}{dx}\left[\mathrm{sen}(x)\right] = \cos(x)$$

Exemplo 10a:
Calcular a derivada da função $\mathrm{f}(x) = x^2 - \mathrm{sen}(x)$.

Temos que $\frac{d}{dx}[x^2 - \text{sen}(x)] = \left[\frac{d}{dx}[x^2]\right] - \left[\frac{d}{dx}[\text{sen}(x)]\right] = 2\,x - \cos(x)$.

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->x^2-sin(x);
```

$$f := x \to x^2 - \sin(x)$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\,\mathrm{f}(x) = 2\,x - \cos(x)$$

Exemplo 10b:

Calcular a derivada da função $\mathrm{f}(x) = \mathbf{e}^x\,\text{sen}(x)$.

Temos que $\frac{d}{dx}[\mathbf{e}^x\,\text{sen}(x)] = \left[\frac{d}{dx}(\mathbf{e}^x)\right]\text{sen}(x) + \mathbf{e}^x\left[\frac{d}{dx}\,\text{sen}(x)\right] = \mathbf{e}^x\,\text{sen}(x) + \cos(x)\,\mathbf{e}^x =$
$\mathbf{e}^x\,[\text{sen}(x) + \cos(x)]$.

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->exp(x)*sin(x);
```

$$f := x \to \mathbf{e}^x \sin(x)$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\,\mathrm{f}(x) = \mathbf{e}^x \sin(x) + \mathbf{e}^x \cos(x)$$

Teorema 10.2 - Derivada da função cosseno

Seja $\mathrm{f}(x) = \cos(x)$ a função cosseno; então a sua derivada é o oposto da função seno. Ou seja:

$$\frac{d}{dx}[\cos(x)] = -\text{sen}(x)$$

Exemplo 10c:

Calcular a derivada da função $\mathrm{f}(x) = x^2 - \cos(x)$.

Temos que $\frac{d}{dx}[x^2 - \cos(x)] = \left[\frac{d}{dx}[x^2]\right] - \left[\frac{d}{dx}[\cos(x)]\right] = 2\,x + \mathrm{sen}(x)$.

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->x^2-cos(x);
```

$$f := x \to x^2 - \cos(x)$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\mathrm{f}(x) = 2\,x + \sin(x)$$

Exemplo 10d:

Calcular a derivada da função $\mathrm{f}(x) = \mathrm{tg}(x)$.

Temos que $\mathrm{tg}(x) = \frac{\mathrm{sen}(x)}{\cos(x)}$. Assim $\frac{d}{dx}[\mathrm{tg}(x)] = \frac{d}{dx}\left[\frac{\mathrm{sen}(x)}{\cos(x)}\right]$

$$\frac{d}{dx}[\mathrm{tg}(x)] = \frac{\left[\frac{d}{dx}[\mathrm{sen}(x)]\right]\cos(x) - \mathrm{sen}(x)\left[\frac{d}{dx}[\cos(x)]\right]}{\cos(x)^2} = \frac{\cos(x)\cos(x) - \mathrm{sen}(x)\,(-\mathrm{sen}(x))}{\cos(x)^2}$$

$$\frac{d}{dx}[\mathrm{tg}(x)] = \frac{\cos(x)^2 + \mathrm{sen}(x)^2}{\cos(x)^2} = \frac{\cos(x)^2}{\cos(x)^2} + \frac{\mathrm{sen}(x)^2}{\cos(x)^2} = 1 + \mathrm{tg}(x)^2$$

Usando o **Maple**:

```
> restart: with(student):
```

```
> f:=x->tan(x);
```

$$f := x \to \tan(x)$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\mathrm{f}(x) = 1 + \tan(x)^2$$

Teorema 10.3 - Derivada de outras funções trigonométricas

Teorema 10.3.1 - Derivada da função tangente

$$\frac{d}{dx}[\,\mathrm{tg}(x)\,] = [\,\sec(x)\,]^2 = 1 + [\,\mathrm{tg}(x)\,]^2$$

Teorema 10.3.2 - Derivada da função secante

$$\frac{d}{dx}[\,\sec(x)\,] = [\,\sec(x)\,]\,[\,\mathrm{tg}(x)\,]$$

Teorema 10.3.3 - Derivada da função cotangente

$$\frac{d}{dx}[\,\mathrm{cotg}(x)\,] = -[\,\mathrm{cosec}(x)\,]^2 = -[\,1 + [\,\mathrm{cotg}(x)\,]^2\,]$$

Teorema 10.3.4 - Derivada da função cossecante

$$\frac{d}{dx}[\,\mathrm{cosec}(x)\,] = -[\,\mathrm{cosec}(x)\,]\,[\,\mathrm{cotg}(x)\,]$$

Exercícios propostos

I - Usando as regras de derivação, calcule as derivadas para cada item, usando: a) multiplicação dos polinômios e aplicando as regras; b) aplicando a regra do produto. Observe que os itens a) e b) devem apresentar o mesmo resultado.

01 - $\mathrm{f}(x) = (x + 1)\,(2\,x - 1)$

02 - $\mathrm{f}(x) = (x + 1)\,(x^2 - x + 1)$

03 - $\mathrm{f}(x) = (2\,x + 1)\,(3\,x^2 + 6)$

II - Encontre a derivada , no ponto $x = 1$, para as curvas:

01 - $y = \dfrac{1}{5\,x - 3}$

02 - $y = \dfrac{2\,x - 1}{x + 3}$

03 - $y = \dfrac{2 - x^2}{x - 2}$

III - Considere a função $f(x) = \dfrac{x}{x^2+1}$. Usando regras de derivação, encontre a função derivada. Com apoio computacional do **Maple**, trace os gráficos das funções $f(x)$ e sua derivada. Estime o valor da derivada nos pontos $x = 0$, $x = 1$ e $x = -1$. Determine as equações das retas tangentes nestes pontos. Refaça o gráfico da função $f(x)$ e as retas tangentes, em um mesmo sistema de coordenadas.

IV - Encontre as equações das retas tangentes à curva $y = \dfrac{x-1}{x+1}$ que sejam paralelas à reta $y = x$.

V - Encontre as derivadas das funções seguintes:

01 - $y = \dfrac{1}{x} + 2\,\mathrm{sen}(x)$

02 - $y = \dfrac{\cos(x)}{1+\mathrm{sen}(x)}$

03 - $y = \dfrac{\mathrm{sen}(x)}{1+\cos(x)}$

04 - $y = 2 - x^2\,\mathrm{sen}(x)$

05 - $y = \dfrac{x}{\ln(x)}$

06 - $y = \dfrac{\ln(x)}{x}$

VI - Para a função $f(x) = \mathrm{sen}(x)\cos(x)$, encontre as derivadas até a quarta ordem. Comprove os resultados com os comandos **Maple** apropriados.

Regra da Cadeia

Nesta seção, desenvolveremos estudos para expressar a derivada de funções compostas a partir das derivadas das funções primitivas. O processo desenvolvido permitirá obter com mais facilidade derivadas de funções complexas usando as derivadas de funções mais simples e as regras vistas anteriormente. Em algumas situações, é muito trabalhoso, quando não impossível, fazermos uso apenas da definição de derivadas e as regras básicas para encontrarmos as derivadas de funções que apresentem composições em sua formulação. Assim sendo, a regra da cadeia vem possibilitarmo-nos encontrar essas derivadas.

Sejam as funções $f(x)$ e $g(x)$ deriváveis em todo o seu domínio. Seja $h(x)$ uma função composta entre estas funções, tais que $y = h(x) = (f \circ g)(x) = f(g(x))$. Desta forma, para qualquer x temos que $g(x)$ está no domínio de $f(x)$.

Vamos considerar, como exemplo, a função $y = h(x) = (x^2 - 1)^3$. Neste caso, para avaliarmos a derivada y', teríamos que, ou aplicarmos a definição de derivadas, ou desenvolvermos o binômio $x^2 - 1$ à terceira potência, o que pode ser trabalhoso em várias situações, e aplicarmos as regras vistas anteriormente.

Podemos pensar a função $h(x)$ como a composição de duas outras funções, tais como $u = g(x) = x^2 - 1$ e $f(x) = x^3$. Logo $y = h(x)$, pode ser expressa como $h(u) = u^3$.

Porém, as derivadas das funções $f(x)$ e $g(x)$ são fáceis de encontrarmos, e expressas por $f'(x) = 3x^2$ e $g'(x) = 2x$. A questão é sabermos como essas derivadas relacionam-se?

Vamos analisar a função $h(x)$ expressa como segue: $h(x) = (x^2 - 1)(x^2 - 1)(x^2 - 1)$. Da regra de derivadas para o produto de funções, se $h(x) = f(x)\,g(x)\,k(x)$, então $h' = f \cdot g \cdot k' + f \cdot g' \cdot k + f' \cdot g \cdot k$.

Assim, obtemos:

$$h'(x) = (x^2 - 1)(x^2 - 1)\,D_x(x^2 - 1) + (x^2 - 1)\,D_x(x^2 - 1)(x^2 - 1) + D_x(x^2 - 1)(x^2 - 1)(x^2 - 1)$$

$$h'(x) = (x^2 - 1)(x^2 - 1)[2x] + (x^2 - 1)[2x](x^2 - 1) + [2x](x^2 - 1)(x^2 - 1)$$

$$h'(x) = 3\,[(x^2 - 1)^2]\,[2x]$$

Observamos que $h'(x) = f'(g(x)) \cdot g'(x)$.

Vamos analisar uma outra situação-exemplo para a função $h(x) = \text{sen}(2x)$.
Podemos, facilmente, compor esta função como segue: $h(x) = f(g(x))$, onde $f(x) = \text{sen}(x)$ e $g(x) = 2x$.

Sabemos das identidades trigonométricas que $\text{sen}(2x) = 2\,\text{sen}(x)\cos(x)$ e $\cos(2x) = \cos(x)^2 - \text{sen}(x)^2$, assim segue:

$$h'(x) = D_x(\text{sen}(2x))$$

$$h'(x) = D_x(2\,\text{sen}(x)\cos(x))$$

$$h'(x) = 2\,[\,D_x(\operatorname{sen}(x)\cos(x))\,]$$

$$h'(x) = 2\,[\,\operatorname{sen}(x)\,D_x(\cos(x)) + D_x(\operatorname{sen}(x)\cos(x))\,]$$

$$h'(x) = 2\,[\,\operatorname{sen}(x)\,(-\operatorname{sen}(x)) + \cos(x)\cos(x)\,]$$

$$h'(x) = 2\,[\,-\operatorname{sen}(x)^2\,] + \cos(x)^2$$

$$h'(x) = 2\,[\,\cos(x)^2 - \operatorname{sen}(x)^2\,]$$

$$h'(x) = 2\,[\,\cos(2\,x)\,]$$

Analogamente ao exemplo anterior, podemos concluir que $h'(x) = f'(g(x))\,.\,g'(x)$.

Proposição - Teorema 11 (Regra da Cadeia): Se $y = h(x) = f(u)$ e $u = g(x)$, e as derivadas $\frac{dy}{du}$ e $\frac{du}{dx}$ existem, então a função composta tem derivada, e sua derivada é expressa por:

$$\frac{dy}{dx} = \left[\frac{dy}{du}\right]\left[\frac{du}{dx}\right]$$

ou

$$h'(x) = f'(u)\,.\,g'(x)$$

ou

$$h'(x) = f'(g(x))\,.\,g'(x)$$

A seguir, comprovamos os exemplos acima mencionados fazendo uso do **Maple**:

Situação Exemplo 01

```
> restart: with(student):
> f:=x->x^3; g:=x->x^2-1;
```

$$f := x \to x^3$$

$$g := x \to x^2 - 1$$

```
> h:=unapply((f@g)(x),x);
```

$$h := x \to (x^2 - 1)^3$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\,f(x) = 3\,x^2$$

```
> Diff('g(x)',x)=diff(g(x),x);
```

$$\frac{d}{dx}\,\mathrm{g}(x) = 2\,x$$

```
> Diff('h(x)',x)=diff(h(x),x);
```

$$\frac{d}{dx}\,\mathrm{h}(x) = 6\,(x^2-1)^2\,x$$

Desenvolvendo (expandindo) os termos do polinômio:

```
> expand(rhs(%));
```

$$6\,x^5 - 12\,x^3 + 6\,x$$

Vamos calcular a derivada da função $\mathrm{h}(x) = (x^2-1)^3$, primeiro, desenvolvendo o polinômio e depois calculando a sua derivada.

```
> k:=(x^2-1)^3;
```

$$k := (x^2-1)^3$$

```
> k:=expand(k);
```

$$k := x^6 - 3\,x^4 + 3\,x^2 - 1$$

```
> diff(k,x);
```

$$6\,x^5 - 12\,x^3 + 6\,x$$

As avaliações do **Maple** acima mostram que os resultados são os mesmos, o que vem comprovar a proposição/teorema da regra da cadeia.

Situação Exemplo 02:

```
> restart: with(student):
> f:=x->sin(x); g:=x->2*x;
```

$$f := x \rightarrow \sin(x)$$

$$g := x \rightarrow 2\,x$$

```
> h:=unapply((f@g)(x),x);
```

$$h := x \rightarrow \sin(2\,x)$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}\,\mathrm{f}(x) = \cos(x)$$

```
> Diff('g(x)',x)=diff(g(x),x);
```

$$\frac{d}{dx}\,\mathrm{g}(x) = 2$$

```
> Diff('h(x)',x)=diff(h(x),x);
```

$$\frac{d}{dx}\,\mathrm{h}(x) = 2\cos(2\,x)$$

Teorema 12 - Regra da potência

Se $u = f(x)$ é uma função derivável e n é uma potência inteira não nula, então

$$\frac{d}{dx}[f(x)^n] = n\,[f(x)^{(n-1)}]\,.\,f'(x)$$

Se r é um número racional, então

$$\frac{d}{dx}[f(x)^r] = r\,[f(x)^{(r-1)}]\,.\,f'(x)$$

Exemplo 03:

Seja a função $f(x) = 5\sqrt{x^2-1}$, determine $f'(x)$.

Podemos escrever que $f(x) = 5\,g(x)$

$$f(x) = 5\sqrt{x^2-1} = 5\,(x^2-1)^{\left(\frac{1}{2}\right)}$$

$$f'(x) = \left[[5]\left[\frac{1}{2}\right][x^2-1]^{\left(\frac{1}{2}-1\right)}\right][2x]$$

$$f'(x) = \left[\frac{5}{2}\right][x^2-1]^{\left(-\frac{1}{2}\right)}\,[2x]$$

$$f'(x) = \frac{5}{\sqrt{x^2-1}}$$

Usando o **Maple**, temos:

```
> restart: with(student):
> f:=x->5*sqrt(x^2-1);
```

$$f := x \to 5\sqrt{x^2-1}$$

```
> Diff('f(x)',x)=diff(f(x),x);
```

$$\frac{d}{dx}f(x) = \frac{5x}{\sqrt{x^2-1}}$$

Exercícios propostos

I - Calcule as derivadas das funções.

$$01 - f(x) = \frac{(2x^4 + 3x^{(-3)})^3}{2}$$

$$02 - f(x) = \frac{(a x^2 + b x)^3}{a}$$

$$03 - f(x) = \frac{2x}{\sqrt{2x-1}}$$

$$04 - f(x) = \frac{\mathbf{e}^{(2-x)}}{5}$$

$$05 - f(x) = 2\,\mathbf{e}^{\sqrt{x}}$$

$$06 - f(x) = 3^{(2x^3 - 3x)}$$

$$07 - f(x) = 3\log_2(2x+3)$$

$$08 - f(x) = \ln\left(\frac{1-x}{1+x}\right)$$

$$09 - f(x) = (1+2x)^{(x^2)}$$

$$10 - f(x) = \mathrm{sen}(3x-1)$$

$$11 - f(x) = 2\,\mathrm{sen}(x^2)\cos(2x)$$

$$12 - f(x) = \mathbf{e}^{(2x)}\,\mathrm{sen}(2x)$$

$$13 - f(x) = \begin{cases} 1-x & x \le 0 \\ \mathbf{e}^{(-x)} & 0 < x \end{cases}$$

$$14 - f(x) = \mathbf{e}^{|2x-1|}$$

$$15 - f(x) = \ln(|2x-1|)$$

Aplicações de Derivadas

Funções Crescentes, Decrescentes e Constantes

Muitas informações a respeito do comportamento de uma função podem ser obtidas a partir do seu gráfico.
Através dele, analisando-se este gráfico da esquerda para a direita, podemos ter uma visão do crescimento ou decrescimento da função, dos valores máximos ou mínimos que ela assume, comportamento para valores muito grandes ou pequenos, pontos de descontinuidade, concavidade da curva, pontos de inflexões, de eventuais simetrias, etc. A derivada de uma função é utilizada para analisarmos, algebricamente, estes aspectos.

Função Crescente - Definição

Seja uma função $f(x)$ definida em um intervalo e sejam dois pontos x_1 e x_2 pertencentes a este intervalo.
Para todo x_1 e x_2 pertencentes a este intervalo e $x_1 < x_2$, e se tivermos $f(x_1) < f(x_2)$, então essa função é <u>crescente</u> neste domínio intervalo.

Os comandos **Maple** a seguir mostram, graficamente, o conceito para funções crescentes.
Vamos considerar a função $y = f(x) = x^3 - 2x^2 + 3x + 1$, definida em **R+**.

```
> restart:with(plots):with(plottools):
> f:=x->x^3-2*x^2+3*x+1:
> graf1:=plot(f(x),x=-0.4..4,y=-2..25,thickness=2,color=blue
  ,tickmarks=[0,0],titlefont=[COURIER,BOLD,14],title="Função
  Crescente"):
> graf2a:=line([1,0],[1,f(1)],linestyle=3,color=black):
> graf2b:=line([3,0],[3,f(3)],linestyle=3,color=black):
> graf3a:=line([0,f(1)],[1,f(1)],linestyle=3,color=black):
> graf3b:=line([0,f(3)],[3,f(3)],linestyle=3,color=black):
> graf4:=textplot([[1,-0.8,"x1"],[3,-0.8,"x2"],[3.8,-0.8,"x"
  ]],color=black):
> graf5:=textplot([[-0.2,f(1),"f(x1)"],[-0.2,f(3),"f(x2)"],[
  -0.2,24,"y"]],color=black):
> display(graf5,graf4,graf3b,graf3a,graf2b,graf2a,graf1);
```



***Exemplo* 01:** Considere a função $y = 2\,x + 1$.

Análise gráfica da função com o **Maple**:

```
> restart: with(plots):
> f:=x->2*x+1;
```

$$f := x \to 2\,x + 1$$

```
> plot(f(x),x=-5..5,y=-5..5,titlefont=[COURIER,BOLD,14],title="Função Crescente");
```



Observamos, no gráfico acima, que para todo $x_1 < x_2$, temos que $f(x_1) < f(x_2)$.

Função Decrescente - Definição

Seja uma função $f(x)$ definida em um intervalo e sejam dois pontos x_1 e x_2 pertencentes a este intervalo.

Para todo x_1 e x_2 pertencentes a este intervalo e $x_1 < x_2$, e se tivermos $f(x_2) < f(x_1)$, então essa função é decrescente neste domínio intervalo.

Os comandos **Maple** a seguir mostram, graficamente, o conceito para funções decrescentes.

Vamos considerar a função $y = f(x) = -x^3 - 2x^2 - 3x + 1$, definida em **R-**.

```
> restart:with(plots):with(plottools):
> f:=x->-x^3-2*x^2-3*x+1:
> graf1:=plot(f(x),x=-4..0.5,y=-2..25,thickness=2,color=blue
  ,tickmarks=[0,0],titlefont=[COURIER,BOLD,14],title="Função
  Decrescente"):
> graf2a:=line([-1,0],[-1,f(-1)],linestyle=3,color=black):
> graf2b:=line([-3,0],[-3,f(-3)],linestyle=3,color=black):
> graf3a:=line([0,f(-1)],[-1,f(-1)],linestyle=3,color=black)
  :
> graf3b:=line([0,f(-3)],[-3,f(-3)],linestyle=3,color=black)
```

```
:
> graf4:=textplot([[-1,-0.8,"x1"],[-3,-0.8,"x2"],[0.4,-0.8,"x"]],color=black):
> graf5:=textplot([[0.3,f(-1),"f(x1)"],[0.3,f(-3),"f(x2)"],[0.2,24,"y"]],color=black):
> display(graf5,graf4,graf3b,graf3a,graf2b,graf2a,graf1);
```



***Exemplo* 02:** Considere a função $y = -2\,x + 1$.

Análise gráfica da função com o **Maple**:

```
> restart:
> f:=x->-2*x+1;
```

$$f := x \to -2\,x + 1$$

```
> plot(f(x),x=-5..5,y=-5..5,titlefont=[COURIER,BOLD,14],titl
```

```
e="Função Decrescente");
```



Observamos, no gráfico acima, que para todo $x_1 < x_2$, temos que $f(x_2) < f(x_1)$.

Função Constante - Definição

Seja uma função $f(x)$ definida em um intervalo e sejam dois pontos x_1 e x_2 pertencentes a este intervalo.

Para todo x_1 e x_2 pertencentes a este intervalo e $x_1 = x_2$, e se tivermos $f(x_1) = f(x_2)$, então essa função é *constante* neste domínio intervalo.

Os comandos **Maple** a seguir mostram, graficamente, o conceito para funções constantes. Vamos considerar a função $y = f(x) = k$, definida em **R+**.

```
> restart:with(plots):with(plottools):
> f:=x->3: # assumindo k=3
> graf1:=plot(f(x),x=-4..4,y=-2..6,thickness=2,color=blue,ti
  ckmarks=[0,0],titlefont=[COURIER,BOLD,14],title="Função
  Constante"):
> graf2a:=line([1,0],[1,f(1)],linestyle=3,color=black):
> graf2b:=line([3,0],[3,f(3)],linestyle=3,color=black):
> graf3a:=line([0,f(1)],[1,f(1)],linestyle=3,color=black):
> graf3b:=line([0,f(3)],[3,f(3)],linestyle=3,color=black):
```

```
> graf4:=textplot([[1,-0.3,"x1"],[3,-0.3,"x2"],[3.8,-0.8,"x"]],color=black):
> graf5:=textplot([[1,f(1)+0.5,"f(x1)"],[3,f(3)+0.5,"f(x2)"],[-0.2,24,"y"]],color=black):
> graf6:=textplot([-0.3,f(3)+0.3,"k"],color=black):
> display(graf6,graf5,graf4,graf3b,graf3a,graf2b,graf2a,graf1);
```



Teorema 13: Seja uma função contínua f(*x*) defninida em um intervalo fechado [*a* ; *b*] e derivável no intervalo aberto (*a* ; *b*).

I - Se f '(*x*) > 0 para todo valor de *x* no intervalo (*a* ; *b*) , então f(*x*) é crescente neste intervalo;

II - Se f '(*x*) < 0 para todo valor de *x* no intervalo (*a* ; *b*) , então f(*x*) é decrescente neste intervalo;

III - Se f '(x) = 0 para todo valor de x no intervalo (a ; b) , então f(x) é constante neste intervalo;

Exemplo 03: Determine os intervalos para os quais a função $f(x) = x^2 - 2x - 3$ é crescente e decrescente.

Solução: Vamos traçar o gráfico da função com apoio do **Maple** e analisarmos as regiões de crescimento e decrescimento.

```
> restart: with(plots): with(student):
> f:=x->x^2-2*x-3;
```

$$f := x \rightarrow x^2 - 2x - 3$$

```
> plot(f(x),x=-3..5,y=-5..3,color=blue,thickness=2);
```



Observamos, no gráfico acima, que as raizes da função são os valores $x_1 = -1$ e $x_2 = 3$.
Ainda, observamos que o vértice da parábola é o valor $x_v = 1$.
O gráfico acima, também nos informa que a função é decrescente para valores de $x < 1$ e crescente para valores $1 < x$.

Para confirmarmos estas observações, vamos aplicar o teorema aicma, conforme comandos **Maple** abaixo:

```
> g:=D(f);
```

$$g := x \rightarrow 2\,x - 2$$

```
> solve(g(x)<0);
```

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(1))$$

```
> solve(g(x)>0);
```

$$\mathrm{RealRange}(\mathrm{Open}(1), \infty)$$

Considerando que a função é contínua em todo o conjunto dos reais, temos que:

f(x) é decrescente para valores de x no intervalo ($-\infty$; 1]
f(x) é crescente para valores de x no intervalo [1 ; ∞)

Exemplo 04: Determine os intervalos para os quais a função $f(x) = x^3 - 1$ é crescente e decrescente.

Solução: Vamos traçar o gráfico da função com apoio do **Maple** e analisarmos as regiões de crescimento e decrescimento.

```
> restart: with(plots): with(student):
> f:=x->x^3-1;
```

$$f := x \rightarrow x^3 - 1$$

```
> plot(f(x),x=-3..3,y=-3..3,color=blue,thickness=2);
```



Observamos, no gráfico acima, que a função aprresenta apenas uma raiz em $x_1 = 1$. Ainda, observamos que a função é sempre crescente.

Para confirmarmos estas observações, vamos aplicar o teorema aicma, conforme comandos **Maple** abaixo:

```
> g:=D(f); # avalia a derivada de f(x) e atribue a função
  g(x)
```

$$g := x \rightarrow 3\, x^2$$

```
> solve(g(x)>0);
```

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(0)), \mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

```
> solve(g(x)<0);
```

A avaliação do comando **Maple** [**solve(g(x) > 0**] nos indica que a função é contínua para x no intervalo ($-\infty$; 0] ou [0 ; ∞).

A avaliação do comando **Maple** [**solve(g(x)** < **0**)] nos indica que a função não encontrou nenhuma solução para esta inequação, ou seja, não existe nenhum valor para x, tal que a derivada da função $f(x)$ seja negativa.

Considerando que a função é contínua em todo o conjunto dos reais, temos que:

$$f(x) \text{ é crescente para valores para todo } x \text{ no intervalo } (-\infty\,;\,\infty)$$

Exemplo 05: Determine os intervalos para os quais a função $f(x) = x^3 + 3\,x^2 - 3$ é crescente e decrescente.

Solução:
Vamos traçar o gráfico da função com apoio do **Maple** e analisarmos as regiões de crescimento e decrescimento.

```
> restart: with(plots): with(student):
> f:=x->x^3+3*x^2-3;
```

$$f := x \rightarrow x^3 + 3\,x^2 - 3$$

```
> plot(f(x),x=-4..4,y=-4..4,color=blue,thickness=2);
```



Vamos aplicar o teorema aicma, conforme comandos **Maple** abaixo:

```
> g:=D(f);
```

$$g := x \rightarrow 3\,x^2 + 6\,x$$

```
> solve(g(x)<0);
```

RealRange(Open(-2), Open(0))

```
> solve(g(x)>0);
```

RealRange($-\infty$, Open(-2)), RealRange(Open(0), ∞)

Considerando que a função é contínua em todo o conjunto dos reais, temos que:

f(x) é decrescente para valores de x no intervalo [-2 ; 0]
f(x) é crescente para valores de x no intervalo ($-\infty$; -2] ou [0 ; ∞)

Concavidades e Pontos de Inflexão

Na seção anterior, vimos que a primeira derivada mostra-nos se uma função é crescente ou decrescente em um intervalo aberto. Porém, não temos informações sobre o grau de curvatura da função, ou ainda, se esta curva apresenta a abertura para cima (côncoca para cima) ou se a abertura da curva é para baixo (côncava para baixo).

O exemplo seguinte, mostra-nos uma sequência animada de gráficos da função $f(x) = x^2$e retas tangentes à curva da função. Neste exemplo, foi utilizado o comando **Maple** ***showtangent*** controlado por uma estrutura de repetição ***for .. do*** .

```
> restart: with(plots): with(student):
> f:=x->x^2-1;
```

$$f := x \rightarrow x^2 - 1$$

```
> for k from -3 to 3 do
  graf[k]:=showtangent(f(x),x=k,x=-4..4,y=-4..10,thickness=[
  1,2],color=[red,blue]):
  end do:
> display([seq(graf[k],k=-3..3)],insequence=true);
```



A animação acima, indica-nos que as retas tangentes à curva apresentam inclinações crescentes e a curva sempre está acima de suas retas tangentes, conforme figura seguinte.



Este outro exemplo, mostra-nos uma animação dos gráficos da função $f(x) = -x^2 + 1$ e retas tangentes à curva.

```
> g:=x->-x^2+1;
```

$$g := x \to -x^2 + 1$$

```
> for k from -3 to 3 do
  graf[k]:=showtangent(g(x),x=k,x=-4..4,y=-10..4,thickness=[
  1,2],color=[red,blue]):
  end do:
> display([seq(graf[k],k=-3..3)],insequence=true);
```



A animação acima, indica-nos que as retas tangentes à curva apresentam inclinações decrescentes e a curva sempre está abaixo de suas retas tangentes, conforme figura seguinte.



Das considerações anteriores, podemos concluir a definição seguinte:

Definição: Se f(*x*) é uma função derivável em um intervalo aberto, então f(*x*) é **côncava para cima**, neste intervalo, se f '(*x*) é **crescente** e **côncava para baixo**, neste intervalo, se f '(*x*) é **decrescente**.

Considerando que as inclinações das retas tangentes a uma curva de uma função derivável são os valores das derivadas da função, e considerando que a função apresente a segunda derivada, então podemos empregar o teorema para funções crescentes e decrescentes, visto anteriormente, o qual possibilita-nos apresentar o seguinte teorema:

Teorema 13: Seja uma função duas vezes derivável, em um intervalo aberto.

I - Se f ''(*x*) > 0 para todo *x* pertencente ao intervalo, então f(*x*) é côncava para cima, neste intervalo.

II - Se f "(x) < 0 para todo x pertencente ao intervalo, então f(x) é côncava para baixo, neste intervalo.

Exemplo 06: Seja a função $f(x) = x^3 + 3x^2 - 3$. Determine as regiões em que a função apresenta concavidade para cima e para baixo.

Solução:
No **Exemplo 05** anterior, analisamos as regiões de crescimento e decrescimento para esta função.

Para isto, traçamos o gráfico da função e avaliamos a sua derivada primeira.

```
> restart: with(plots): with(student):
> f:=x->x^3+3*x^2-3;
```

$$f := x \rightarrow x^3 + 3x^2 - 3$$

```
> plot(f(x),x=-4..4,y=-4..4,color=blue,thickness=2);
```



Determinação das suas derivadas de primeira e segunda ordens:

```
> g:=D(f); h:=D(g);
```

$$g := x \rightarrow 3\,x^2 + 6\,x$$

$$h := x \rightarrow 6\,x + 6$$

A função apresenta crescimento quando g(x) > 0 e decrescimento quando g(x) < 0 , ou seja:

```
> Crescimento=solve(g(x)>0);
```

$$\mathit{Crescimento} = (\mathrm{RealRange}(-\infty, \mathrm{Open}(-2)), \mathrm{RealRange}(\mathrm{Open}(0), \infty))$$

```
> Decrescimento=solve(g(x)<0);
```

$$\mathit{Decrescimento} = \mathrm{RealRange}(\mathrm{Open}(-2), \mathrm{Open}(0))$$

As regiões com concavidade para cima são dadas por:

```
> Côncava_para_cima=solve(h(x)>0);
```

$$\mathit{Côncava\_para\_cima} = \mathrm{RealRange}(\mathrm{Open}(-1), \infty)$$

```
> graf1:=plot(f(x),x=-1..4,y=-4..4,color=blue,thickness=2):
> graf2:=plot(f(x),x=-1..4,y=-4..4,filled=true,color=cyan):
> display(graf1,graf2);
```



As regiões com concavidade para baixo são dadas por:

```
> Côncava_para_baixo=solve(h(x)<0);
```

$$Côncava\_para\_baixo = \mathrm{RealRange}(-\infty, \mathrm{Open}(-1))$$

```
> graf1:=plot(f(x),x=-4..-1,y=-4..4,color=blue,thickness=2):
> graf2:=plot(f(x),x=-4..-1,y=-4..4,filled=true,color=cyan):
> display(graf1,graf2);
```



No **Exemplo 06** acima, observamos que o gráfico da função muda de concavidade para baixo para a concavidade para cima, no ponto $x = 1$. Este ponto apresenta uma característica especial, conforme definição seguinte:

Definição: Se $f(x)$ é uma função contínua em um intervalo aberto contendo um ponto $x = x_0$, e neste ponto o gráfico da função muda de concavidade, então dizemos que o ponto x_0 do domínio, ou o ponto $(x_0\ ;\ f(x_0))$ do gráfico, é um ponto de **inflexão** da função.

Das definições e teoremas anteriores, podemos concluir que os pontos de inflexão de uma função são os pontos em que esta função mude de uma função crescente para uma função decrescente ou vice-versa.

```
> graf1:=plot(f(x),x=-4..3,y=-4..4,color=blue,thickness=2):
> graf2:=plot(f(x),x=-4..-1,y=-4..4,filled=true,color=pink):
```

```
> graf3:=plot(f(x),x=-1..3,y=-4..4,filled=true,color=tan):
> graf4:=plot([[-1,0],[-1,f(-1)]],color=red,thickness=2):
> graf5:=plot([[-1,f(-1)],[0,f(-1)]],color=red,thickness=2):
> graf6:=textplot([-1,-1.2,"Ponto de Inflexão"],color=red):
> display(graf1,graf2,graf3,graf4,graf5,graf6);
```



Máximos e Mínimos de uma Função

Quando observamos os gráficos de funções, notamos que estes apresentam regiões de crescimento, decrescimento, e pontos onde a função atinge valores máximo ou mínimo. Estes pontos de máximo ou mínimo podem ser pontos absolutos, ou seja, o maior ou menor valor que a função assume, ou pontos relativos (ou locais) a uma região ou intervalo de análise. Nesta seção, analisaremos como a derivada de uma função pode ser utilizada para encontrarmos estes pontos.

Definição:

Dizemos que uma função f(x), definida em um intervalo aberto, tem um máximo relativo (ou local) em um ponto x_0 , se houver um f(x_0) > f(x), para todo x no intervalo. Da mesma forma, dizemos que uma função f(x), definida em um intervalo aberto, tem um mínimo relativo (ou local) em um ponto x_0 , se houver um f(x_0) < f(x), para todo x no intervalo.

Exemplo 01:

I - Seja a função $f(x) = x^2 - 1$ definida em um intervalo aberto. Vamos proceder a análise do seu gráfico.

```
> restart: with(plots): with(student):
> f:=x->x^2-1;
```

$$f := x \rightarrow x^2 - 1$$

```
> plot(f(x),x=-3..3,y=-2..4,color=blue,thickness=2);
```



Observamos que, no intervalo em análise, o gráfico da função apresenta um valor mínimo relativo (ou local), quando x = 0, porém não apresenta máximos relativos.

II - Seja a função $g(x) = -x^2 + 1$ definida em um intervalo aberto. Vamos proceder a análise do seu gráfico.

```
> g:=x->-x^2+1;
```

$$g := x \to -x^2 + 1$$

```
> plot(g(x),x=-3..3,y=-4..2,color=blue,thickness=2);
```



Observamos que, no intervalo em análise, o gráfico da função apresenta um valor máximo relativo (ou local), quando $x = 0$, porém não apresenta mínimos relativos.

III - Seja a função $h(x) = x^3 - 1$ definida em um intervalo aberto. Vamos proceder a análise do seu gráfico.

```
> h:=x->x^3-1;
```

$$h := x \to x^3 - 1$$

```
> plot(h(x),x=-3..3,y=-3..3,color=blue,thickness=2);
```



Observamos que, no intervalo em análise, o gráfico da função não apresenta valores mínimo nem máximo relativos.

IV - Seja a função $\mathrm{k}(x) = x^3 - 3\,x + 3$ definida em um intervalo aberto. Vamos proceder a análise do seu gráfico.

```
> k:=x->x^3-3*x+3;
```

$$k := x \to x^3 - 3\,x + 3$$

```
> plot(k(x),x=-4..4,y=-8..10,color=blue,thickness=2);
```



Observamos que, no intervalo em análise, o gráfico da função apresenta um valor máximo relativo (ou local), quando x = -1, um valor mínimo relativo (ou local), quando x = 1 .

V - Seja a função $t(x) = \mathrm{sen}(x)$ definida em um intervalo aberto. Vamos proceder a análise do seu gráfico.

```
> s:=x->cos(x);
```

$$s := x \rightarrow \cos(x)$$

```
> plot(s(x),x=-3*Pi..3*Pi,y=-2..2,color=blue,thickness=2);
```



Observamos que, no intervalo em análise, o gráfico da função apresenta valores máximos relativos para múltiplos de $2\,k\,\pi$, e apresenta valores mínimos relativos para múltiplos $k\,\pi$, onde k pertence ao conjunto dos inteiros.

Para a função do exemplo V, vamos traçar os gráfico das tangente nos pontos de máximo e mínimo relativos, fazendo uso do comando ***showtangent*** do **Maple**.

```
> graf1:=plot(k(x),x=-4..4,y=-8..8,color=blue,thickness=2):
> graf2:=showtangent(k(x),x=-1,x=-4..4,y=-8..8,color=[red,bl
  ack]):
> graf3:=showtangent(k(x),x=1,x=-4..4,y=-8..8,color=[red,bla
  ck]):
> display(graf1,graf2,graf3);
```



Observamos neste gráfico, que as inclinações das retas tangentes à curva é igual a zero, pois as tangentes estão paralelas ao eixo-*x*.

Definição: Seja f(x) uma função contínua definida em um intervalo aberto, contendo um ponto $x = x_0$. Definimos **ponto crítico** da função f(x) um ponto $x = x_0$, tal que x_0 seja um máximo ou mínimo relativo da função f(x).

Teorema 14: Seja f(x) uma função contínua definida em um intervalo aberto, contendo um ponto $x = x_0$. Se x_0 for um ponto crítico de f(x), então f '(x) = 0 ou f(x) não é derivável neste ponto.

Teorema 15 - Teorema da 1ª Derivada
Seja f(x) uma função contínua definida em um intervalo aberto, contendo um ponto crítico $x = x_0$.

I - Se f '(x) > 0 à esquerda de x_0 e f '(x) < 0 à direita de x_0, então f(x) tem um máximo relativo em x_0 .

II - Se f '(x) < 0 à esquerda de x_0 e f '(x) > 0 à direita de x_0, então f(x) tem um mínimo relativo em x_0 .

III - Se f '(x) tiver o mesmo sinal à esquerda e à direita de x_0 , então f(x) não tem máximo nem mínimo relativo em x_0 .

Teorema 16 - Teorema da 2ª Derivada
Seja f(x) uma função contínua definida em um intervalo aberto, contendo um ponto crítico $x = x_0$, e considerando que f(x) tenha a derivada de segunda ordem.

I - Se f '(x) = 0 e f "(x) > 0, então f(x) tem um mínimo relativo em x_0 .

II - Se f '(x) = 0 e f "(x) < 0, então f(x) tem um máximo relativo em x_0 .

III - Se f '(x) = 0 e f "(x) = 0, então o teste é inconclusivo, isto é, f(x) pode ter um mínimo ou máximo relativo ou nenhum dos dois em x_0 .

Exemplo 02: Seja a função $f(x) = 3x^5 - 5x^3$. Determine:

a) Região de crescimento e decrescimento da função;
b) Pontos de mínimo e máximos relativos da função;
c) Pontos de inflexão da função.

```
> restart: with(plots): with(student):
> f:=x->3*x^5-5*x^3;
```

$$f := x \to 3x^5 - 5x^3$$

Traçando o gráfico da função:

```
> plot(f(x),x=-2..2,y=-3..3,color=blue,thickness=2);
```



Determinando as derivadas de primeira e segunda ordens:

```
> g:=D(f); h:=D(D(f));
```

$$g := x \rightarrow 15\,x^4 - 15\,x^2$$

$$h := x \rightarrow 60\,x^3 - 30\,x$$

Encontrando regiões de crescimento e decrescimento:

```
> Crescimento=solve(g(x)>0);
```

$$Crescimento = (\mathrm{RealRange}(-\infty, \mathrm{Open}(-1)), \mathrm{RealRange}(\mathrm{Open}(1), \infty))$$

```
> Decrescimento=solve(g(x)<0);
```

$$Decrescimento = (\mathrm{RealRange}(\mathrm{Open}(-1), \mathrm{Open}(0)), \mathrm{RealRange}(\mathrm{Open}(0), \mathrm{Open}(1)))$$

Determinando pontos críticos:

```
> pontos_críticos=solve(g(x)=0);
```

$$pontos\_críticos = (0, 0, 1, -1)$$

Determinado pontos mínimo e máximo relativos:
Aplicando-se os pontos críticos à segunda derivada:

```
> 'h(-1)'=h(-1);'h(0)'=h(0);'h(1)'=h(1);
```

$$h(-1) = -30$$
$$h(0) = 0$$
$$h(1) = 30$$

Como em $x = -1$ a segunda derivada é negativa, então $x = -1$ é um ponto de máximo relativo.

Como em $x = 1$ a segunda derivada é positiva, então $x = 1$ é um ponto de mínimo relativo.

Como em $x = 0$ a segunda derivada é igual a zero, então neste ponto o teste da segunda derivada é inconclusivo. Deveremos analisar o sinal de $f'(x)$ à esquerda e à direita do ponto $x = 0$.

O comando **sign** do **Maple** analisa o signal de uma expressão, retornando -1 ou 1, se este for negativo ou positivo, respectivamente.

N.B.: Ver *help* do **Maple** para sintaxe e aplicações do comando **sign**.

```
> plot(sign(g(x)),x=-1..1);
```



Como não ocorre mudança de sinal da primeira derivada, então no ponto $x = 0$, não há nem um máximo nem um mínimo relativo.

Exercícios propostos:

I - Para cada função seguinte, defina a função no **Maple** e trace o seu gráfico. Utilizando-se de aplicações das derivadas, determine, se existirem:

a) os máximos relativos;
b) os mínimos relativos;
c) os pontos de inflexão;
d) regiões de crescimento e decrescimento;
e) regiões de concavidade para cima e para baixo.

01) $f(x) = (x^2 - 4)^2$

02) $f(x) = \dfrac{4x}{x^2 + 4}$

03) $f(x) = x\,\mathbf{e}^{x}$

04) $f(x) = x^2\,\mathbf{e}^{x}$

05) $f(x) = 2\,x\,\mathbf{e}^{(-3\,x)}$

Problemas de Máximos e Mínimos

Problema 01: Máximo volume de uma caixa
Uma caixa aberta é feita de um pedaço de papelão medindo 15 cm x 21 cm, cortando fora os quatro quadrados dos cantos e dobrando para cima os lados.



a) Seja V o volume da caixa que resulta quando os quadrados tiverem lados de comprimento *x*. Determine uma expressão (fórmula) para V como função de *x*.
b) Encontre o domínio da função.
c) Construa uma tabela de valores para x e $V(x)$, com *x* inteiros no intervalo de 0 até 8.

d Construa um gráfico para estimar a imagem de V.
e) Observando a tabela e o gráfico, estime com valores aproximados, qual o volume máximo da caixa e qual o valor correspondente para o lado x do quadrado, onde ocorre este máximo?
f) Encontre, analiticamente, fazendo uso dos conceitos de derivadas para máximo e mínimos relativos, o valor de x que fornece o máximo volume.
g) Qual é o valor para x quando o volume for 300 cm^3?
h) Qual é o intervalo de x para o qual volume está aumentando? E diminuindo?
i) Construa (monte) a caixa, observando a figura acima, e os valores encontrados nos itens anteriores. Comprove, experimentalmente, derramando água na caixa para verificar a capacidade máxima desta.

Obs.: $1\ cm^3 = 1\ ml$ (um centímetro cúbico é igual a um mililitro).

Solução:

```
> restart: with(plots):
```

a) Expressão

```
> V:=x->x*(15-2*x)*(21-2*x);
```

$$V := x \to x\,(15-2\,x)\,(21-2\,x)$$

```
> V:=unapply(expand(V(x)),x);
```

$$V := x \to 315\,x - 72\,x^2 + 4\,x^3$$

b) Domínio

$\mathbf{D} = \{\ x \text{ em } \mathbf{R}\ /\ \ 0 \le x \text{ and } x \le \dfrac{15}{2}\ \}$

c) Tabela

	A	B
1	x	V(x)
2	0	247
3	1	374
4	2	405
5	3	364
6	4	275
7	5	162
8	6	49
9	7	-40
10	8	

d) Gráfico

```
> plot(V(x),x=0..8,V=0..500);
```



e) Máximo

```
> Equção_Ponto_Máximo:=diff(V(x),x);
```

$$\mathit{Equção\_Ponto\_Máximo} := 315 - 144\,x + 12\,x^2$$

```
> solve(Equção_Ponto_Máximo=0);
```

$$6 + \frac{\sqrt{39}}{2},\ 6 - \frac{\sqrt{39}}{2}$$

```
> evalf(%);
```

$$9.122498999,\ 2.877501001$$

Considerando que o valor de x deve estar no intervalo $0 \le x,\ x \le \frac{15}{2}$, então:

Ponto_Máximo ==> { $x = 2,\ 8775$ }

f) Contradomínio

D = { x em **R** / $0 \le x$, $x \le 500$ }

g) V(x)=300

```
> fsolve(V(x)=300);
```

$$1.323109612, 4.754449907, 11.92244048$$

Valores de x para volume de 300 cm^3 .

V(300) ==> { $x = 1.323$ ou $x = 4.754$ }

Problema 02: Área máxima de um retângulo sob uma curva

Encontre as dimensões do retângulo, na figura abaixo, para que este tenha a maior área possível, com sua base sobre o eixo-*x* e os dois vértices localizados sobre a curva da parábola $y = 8 - x^2$.



Solução 1: sem uso de derivada

```
> restart:Digits:=4:
> with(plots):with(plottools):
> with(student):
```

Definição da função:

```
> f:=x->8-x^2;
```

$$f := x \rightarrow 8 - x^2$$

Chamando $A(x)$ como a área do retângulo em função da dimensão x, temos:
$A = 2\,x\,(8 - x^2)$.

```
> A:=x->2*x*(8-x^2);
```

$$A := x \rightarrow 2\,x\,(8 - x^2)$$

Traçando o gráfico $A(x)$ em intervalo apropriado:

```
> plot(A(x),x=0..3);
```



Observamos que para $x = 1.6$ nos fornece o ponto máximo para a curva. Logo, temos que $a = 1.6$

```
> f(1.6);
```

$$5.44$$

A altura b = f(1.6), logo $b = 5.4$

Assim, a dimensão do retângulo será: *base* = 2 . 1,6 = 3,2 e a *altura* = 5,4 .
Consequentemente, a área máxima será:

```
> A:=2*1.6*f(1.6);
```

$$A := 17.41$$

Como solução, encontramos o retângulo com dimensões de: 3,2 e 5,4 e área $A := 17.41$.

Solução 2: com uso de derivada

Utilizando-se das funções $f(x)$ e $A(x)$ acima desenvolvidas, e sabendo-se que, igualando-se a zero a derivada da função $A(x)$ nos dá um ponto crítico, então temos:

```
> f:=x->8-x^2; A:=x->2*x*(8-x^2);
```

$$f := x \to 8 - x^2$$

$$A := x \to 2\,x\,(8 - x^2)$$

```
> x=solve(D(A)(x)=0);
```

$$x = \left(-\frac{2\sqrt{6}}{3}, \frac{2\sqrt{6}}{3}\right)$$

```
> evalf(%);
```

$$x = (-1.633, 1.633)$$

Calculando-se a derivada 2a. no ponto *x*, então teremos um máximo ou mínimo, a saber:

```
> diff(A(x),x$2);
```

$$-12\,x$$

Considerando que *a* é uma dimensão, então temos que $x = a = 1.633$, nos dá:

```
> subs(x=1.633,%);
```

$$-19.60$$

Como o resultado acima é um valor negativo, então temos que no ponto $x = a = 1.633$ é um ponto de máximo.

Consequentemente, o valor para *b* será:

```
> b=evalf(f(1.633));
```

$$b = 5.333$$

Desta forma, teremos a área:

```
> Area_máxima=A(1.633);
```

$$Area\_máxima = 17.42$$

Problema 03: Volume máximo de um cone

Um cone é formado pela revolução de um triângulo retângulo, com o comprimento da hipotenusa de 3 cm, sobre uma mesa.
Qual é o máximo volume do cone que pode ser formado desta maneira?

Solução:
Os comandos **Maple** mostram uma representação gráfica do problema.

```
> restart:
```

```
> with(plots): with(plottools):
> with(student):
> graf1:=cone([0,0,4],sqrt(8),-4,color=gold):
> graf2:=plot3d(0,x=-3..3,y=-3..3,color=grey,grid=[5,10],style=patchnogrid):
> graf3:=pointplot3d([[0,0,0],[sqrt(8),0,0],[0,0,4],[0,0,0]],connect=true,color=black,thickness=2):
> display(graf1,graf2,graf3,scaling=constrained,style=wireframe,orientation=[45,60]);
```



Seja a o comprimento da base do triângulo retângulo e b a sua altura.
O problema nos dá que a hipotenusa do triângulo mede 3 cm.

O volume do cone é dado pela expressão: $V = \frac{\pi a^2 b}{3}$. Nós desejamos maximizar o volume.

Das relações do triângulo retângulo, e o Teorema de Pitágoras fornece-nos: $a^2 + b^2 = c^2$.

```
> Vol:=Pi*a^2*b/3;
```

$$Vol := \frac{\pi\, a^2\, b}{3}$$

```
> pitagoras:=c^2=a^2+b^2;
```

$$pitagoras := c^2 = a^2 + b^2$$

```
> pitagoras:=subs(c=3,pitagoras);
```

$$pitagoras := 9 = a^2 + b^2$$

```
> quadrado_a:=isolate(pitagoras,a^2);
```

$$quadrado\_a := a^2 = 9 - b^2$$

```
> V:=unapply(subs(a^2=rhs(quadrado_a),Vol),b);
```

$$V := b \to \frac{1}{3}\, \pi\, (9 - b^2)\, b$$

```
> plot(V(b),b=0..3);
```



Da observação do gráfico, vemos que o volume tem um máximo e que este máximo está situado para os valores de b entre 1.5 e 2.0, e que o máximo pode ser calculado por $\frac{d}{db}\mathrm{V}(b) = 0$.

```
> Diff('V(b)',b)=diff(V(b),b);
```

$$\frac{d}{db}\mathrm{V}(b) = -\frac{2\,\pi\, b^2}{3} + \frac{\pi\,(9 - b^2)}{3}$$

Assim, resolvendo-se esta equação, temos:

```
> b=solve(rhs(%)=0,b);
```

$$b = (\sqrt{3}, -\sqrt{3})$$

O cone tem o volume máximo quando o valor de $b = \sqrt{3}$, considerando que uma dimensão não pode ter valor negativo.
Logo, podemos calcular a dimensão de *a*, a saber:

```
> 'a'=solve(subs(b=sqrt(3),pitagoras),a);
```

$$a = (-\sqrt{6}, \sqrt{6})$$

Considerando que *a* não pode ser negativo, então temos que $a = \sqrt{6}$.

Desta forma, podemos caluclar o volume máximo, a saber:

```
> Volume_Maximo=subs(a=sqrt(6),b=sqrt(3),Vol);
```

$$Volume\_Maximo = 2\,\pi\,\sqrt{3}$$

Problema 04: Ângulo para área máxima de um triângulo

Um triângulo apresenta dois lados de comprimentos 5 e 8 unidades.
a) Qual é o ângulo entre os dois lados tais que a sua área seja máxima?
b) Qual é o perímetro do triângulo quando sua área é máxima?



Solução:

Definindo *a* e *b* como os lados do triângulo, *A* como a sua área e β como o ângulo entre os dois lados.

Desta forma, temos: $A = \frac{a\,b\,\sin(\beta)}{2}$. Nós pretendemos maximar A.

```
> restart:
> with(plots):
> with(student):
> Area:=a*b*sin(beta)/2;
```

$$Area := \frac{1}{2}\,a\,b\,\sin(\beta)$$

```
> A:=unapply(subs(a=5,b=8,Area),beta);
```

$$A := \beta \rightarrow 20\,\sin(\beta)$$

```
> plot(A(beta),beta=0..2*Pi);
```



Do gráfico acima, observamos que o máximo está entre 0 e π .

Podemos calcular o ponto critico, conforme a saber:

```
> Ponto_Critico=fsolve(D(A)(beta)=0,beta=0..Pi);
```

$$Ponto\_Critico = 1.570796327$$

Verificando o valor A''($\frac{\pi}{2}$):

```
> D(D(A))(Pi/2);
```

$$-20$$

O ponto crítico de A(β) entre 0 e π é: $\frac{\pi}{2}$.

Como A''($\frac{\pi}{2}$) < 0, então a área do triângulo é máxima quando o ângulo entre os lados dados é $\frac{\pi}{2}$ radianos.

Para calcularmos o perímetro há a necessidade de obtermos o outro lado *c*, que pode ser dado pela lei dos cossenos, a saber:

$c^2 = a^2 + b^2 - 2\,a\,b\cos(\beta)$

```
> cc:=sqrt(a^2+b^2-2*a*b*cos(beta));
```

$$cc := \sqrt{a^2 + b^2 - 2\,a\,b\cos(\beta)}$$

```
> c=evalf(subs(a=5,b=8,beta=Pi/2,cc));
```

$$c = 9.433981133$$

Desta forma, o perímetro *p*, será:

```
> p:=a+b+c;
```

$$p := a + b + c$$

```
> Perímetro=subs(a=5,b=8,c=9.43,p);
```

$$Perímetro = 22.43$$

Problema 05: Distância mínima de um ponto a uma curva

Dado um ponto **P** de coordenadas (a ; b) e o gráfico de uma função f(x), calcular a distância do ponto **P** ao gráfico da função.

Solução:

Por definição, a distância de P ao gráfico da função é a menor distância entre os pontos Q do gráfico e o ponto P. Então, podemos definir:

$$\text{distância} = \left| P - Q \right|.$$

Vamos considerar uma função $f(x) = \cos(x)$ e o ponto P = (2 ; 3). Um ponto do gráfico de $f(x)$ é Q = (x ; $\cos(x)$).
Desta forma a distância do ponto P a um ponto Q do gráfico é dada por:

```
> restart: with(plots): with(plottools): with(student):
> f:=x->cos(x);
```

$$f := x \to \cos(x)$$

```
> distancia:=sqrt((2-x)^2+(3-cos(x))^2);
```

$$distancia := \sqrt{(2-x)^2 + (3-\cos(x))^2}$$

Desejamos a menor distância. Logo, a menor distância será determinada quando a derivada da expressão acima se anula.
A seguir, apresentamos o gráfico da expressão da distância.

```
> plot(distancia,x=0..3,y=0..4,color=red);
```



O ponto em que a derivada da expressão distancia se anula é:

```
> Ponto_minimo:=diff(distancia,x);
```

$$Ponto\_minimo := \frac{1}{2}\,\frac{2\,x-4+2\,(3-\cos(x))\,\sin(x)}{\sqrt{(2-x)^2+(3-\cos(x))^2}}$$

Resolvendo a equação Ponto_minimo = 0, para x, em um intervalo de [0 ; 3], temos:

```
> x[min]:=fsolve(Ponto_minimo,x,0..3);
```

$$x_{min} := 0.6551969516$$

```
> y[min]:=f(x[min]);
```

$$y_{min} := 0.7929279378$$

A menor distância será determinada substituindo-se o valor x_{min} acima encontrado na expressão da distância. Logo:

```
> menor_distancia:=subs(x=x[min],distancia);
```

$$menor\_distancia := \sqrt{1.808495238+(3-\cos(0.6551969516))^2}$$

```
> Resposta=evalf(%);
```

$$Resposta = 2.584504271$$

Análise gráfica do problema:

```
> graf1:=plot(f(x),x=0..2*Pi,y=-1..4,color=red,thickness=2):
> graf2:=plot([ [x[min],y[min]], [2,3]
  ],color=black,linestyle=3,thickness=2):
> graf3:=disk([2,3],0.1,color=blue):
> display(graf1,graf2,graf3);
```



Problema 06: Minimizando área de uma lata cilíndrica
Uma lata cilíndrica deve ser confeccionada de tal forma que tenha a capacidade de 1 litro. Que dimensões deve ter esta lata de tal forma que seja usado o mínimo de material possível?

Solução:
Os comandos **Maple** abaixo demonstram a configuração da lata

```
> restart: with(plots): with(plottools):
> graf1:=cylinder([0,0,0],1,4):
> display(graf1);
```



Sejam r e h as medidas do raio e altura da lata, expressos em centímetros.
Desta forma, o volume da lata será expresso por: $V = \pi r^2 h$.

A superfície lateral da lata será expressa pela soma da área das base, que são circulares e a parede do cilindro, que corresponde a um retângulo. Logo, chamando A a área da superfície da lata, temos: $A = 2 \pi r^2 + 2 \pi r h$.

Implementando as equações no **Maple**, temos:

```
> V:=Pi*r^2*h;
```

$$V := \pi r^2 h$$

```
> Area:=2*Pi*r^2+2*Pi*r*h;
```

$$Area := 2 \pi r^2 + 2 \pi r h$$

Na primeira equação, vamos expressar a altura h em função do raio r e substituirmos o valor do volume.

```
> isolate(subs(V=1000),h);
```

$$h = \frac{1000}{\pi r^2}$$

Assim,

```
> h:=1000/(Pi*r^2);
```

$$h := \frac{1000}{\pi r^2}$$

```
> A:=unapply(Area,r); # substituindo a expressão de h em Area
```

$$A := r \rightarrow 2 \pi r^2 + \frac{2000}{r}$$

Considerando que o valor do raio é positivo e nosso objetivo é minimizarmos a expressão da Área, vamos traçar o gráfico da Área em função do raio e analisarmos o comportamento deste gráfico.

```
> plot(A(r),r=0..15,A=0..2000);
```



Uma análise do gráfico acima, observamos que para valores pequenos do raio (raio igual a 2 unidades) temos um valor da área crescente e alto, o mesmo acontece quando o valor do raio maior do que 10 unidades. Para o primeiro caso, teremos uma lata fina e alta e para o segundo caso, uma lata larga e baixa.

Considerando que a função $\mathrm{A}(r)$ é derivável, então a área deve apresentar um valor mínimo quando $\mathrm{A}'(r) = 0$.

```
> Amin:=diff(A(r),r);
```

$$Amin := 4\,\pi\,r - \frac{2000}{r^2}$$

```
> fsolve(Amin=0,r);
```

$$5.419260701$$

Logo, o valor do raio $r = 5.42$ cm . Substituindo este valor na expressão da altura, temos:

```
> 'h'=evalf(subs(r=5.42,h));
```

$$h = 10.83556481$$

Por conseguinte, temos $h = 10.84$. Observamos que $h = 2\,r$ fornece o mínimo de material na fabricação da lata.

Exercícios propostos:

01 - Encontre as dimensões do retângulo com área máxima que pode ser inscrito em um círculo com raio de 10 cm.

02 - Uma linha com 100 cm deve ser dividida em dois pedaços. Com um dos pedaços devemos construir uma circunferência e com o outro pedaço deve ser formado um quadrado. Qual deve ser o tamanho da linha destinada ao quadrado de tal forma que as áreas formadas pelo quadrado e pelo círculo sejam iguais? E qual o tamanho para que a soma das áreas do quadrado e do círculo sejam máxima?

03 - Encontre o ponto P situado sobre o gráfico da da hipérbole $x\,y = 1$, que está mais próximo da origem do sistema de coordenadas.

04 - Achar dois números positivos cuja soma seja 70 e cujo produto seja o maior possível.

05 - Uma horta retangular apresenta uma área de 216 m^2 . Quais as dimensões da horta a fim de que esta apresente a menor quantidade de cerca? Quantos metros lineares de cerca serão necessários?

Regra de L'Hospital

No estudo de Limites deparamo-nos com algumas indeterminações do tipo $\frac{0}{0}$, $\frac{\infty}{\infty}$, $0 \cdot \infty$ ou $\infty - \infty$. As soluções encontradas eram efetuadas através de uma análise gráfica ou aplicando-se fatorações ao numerador ou denominador da expressão, possibilitando-nos, desta forma, eliminar a indeterminação. Nesta seção, apresentaremos uma regra, através de um processo de derivações (sucessivas), o qual vem eliminar estas indeterminações, na obtenção de limites.

Indeterminações do tipo: 0/0

Teorema - Regra de L'Hospital

Sejam f(*x*) e g(*x*) funções deriváveis em um intervalo aberto, contendo um ponto $x = c$, exceto talvez em $x = c$, e que

$$\lim_{x \to c} f(x) = 0 \quad e \quad \lim_{x \to c} g(x) = 0$$

Se $\lim_{x \to c} \left[\frac{f'(x)}{g'(x)} \right] = L$ (isto é, se o limite existir) , ou se esse limite é ∞ ou $-\infty$, e $g'(x) \neq 0$, então

$$\lim_{x \to c} \left[\frac{f(x)}{g(x)} \right] = \lim_{x \to c} \left[\frac{f'(x)}{g'(x)} \right]$$

N.B.: A Regra de L'Hospital acima, também é válida no caso de limites quando:

$$\lim_{x \to c-} \left[\frac{f(x)}{g(x)} \right], \lim_{x \to c+} \left[\frac{f(x)}{g(x)} \right], \lim_{x \to (-\infty)} \left[\frac{f(x)}{g(x)} \right] \text{ou} \lim_{x \to \infty} \left[\frac{f(x)}{g(x)} \right]$$

Indeterminações do tipo: $\frac{\infty}{\infty}$

Teorema - Regra de L'Hospital

Sejam f(*x*) e g(*x*) funções deriváveis em um intervalo aberto, contendo um ponto $x = c$, exceto talvez em $x = c$, e que

$$\lim_{x \to c} f(x) = \infty \quad e \quad \lim_{x \to c} g(x) = \infty$$

Se $\lim_{x \to c} \left[\frac{f'(x)}{g'(x)} \right] = L$ (isto é, se o limite existir) , ou se esse limite é ∞ ou $-\infty$, e $g'(x) \neq 0$, então

$$\lim_{x \to c} \left[\frac{f(x)}{g(x)} \right] = \lim_{x \to c} \left[\frac{f'(x)}{g'(x)} \right]$$

N.B.: A Regra de L'Hospital acima, também é válida no caso de limites quando:

$$\lim_{x \to c-} \left[\frac{f(x)}{g(x)} \right], \lim_{x \to c+} \left[\frac{f(x)}{g(x)} \right], \lim_{x \to (-\infty)} \left[\frac{f(x)}{g(x)} \right] \text{ou} \lim_{x \to \infty} \left[\frac{f(x)}{g(x)} \right]$$

Exemplo 01: Encontre o limite de $f(x) = \dfrac{x^2 - 4}{x - 2}$, quando $x \to 2$, usando a Regra de L'Hospital e confira o resultado calculando o limite usando fatoração.

Solução 1: Regra de L'Hospital

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\dfrac{0}{0}$, pois:

$$\lim_{x \to 2} \frac{x^2 - 4}{x - 2} = \frac{2^2 - 2}{2 - 2} = \frac{0}{0}$$

Solução:

$$\lim_{x \to 2} \frac{x^2 - 4}{x - 2} = \lim_{x \to 2} \left[\frac{\frac{d}{dx}(x^2 - 4)}{\frac{d}{dx}(x - 2)} \right] = \lim_{x \to 2} \frac{2x}{1} = \lim_{x \to 2} 2x = 4$$

Solução 2: Usando fatoração

$$\lim_{x \to 2} \frac{x^2 - 4}{x - 2} = \lim_{x \to 2} \frac{(x - 2)(x + 2)}{x - 2} = \lim_{x \to 2} x + 2 = 4$$

Exemplo 02: Encontre $\lim\limits_{x \to 0} \dfrac{3x - \operatorname{sen}(x)}{x}$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\dfrac{0}{0}$, pois:

$$\lim_{x \to 0} \frac{3(0) - \operatorname{sen}(0)}{0} = \frac{0 - 0}{0} = \frac{0}{0}$$

Solução:

$$\lim_{x \to 0} \frac{3x - \operatorname{sen}(x)}{x} = \lim_{x \to 0} \left[\frac{\frac{d}{dx}(3x - \operatorname{sen}(x))}{\frac{d}{dx}x} \right] = \lim_{x \to 0} \frac{3 - \cos(x)}{1} = \frac{3 - 1}{1} = 2$$

Exemplo 3: Encontre $\lim\limits_{x \to 0} \dfrac{x}{\mathbf{e}^x - 1}$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\dfrac{0}{0}$, pois:

$$\lim_{x \to 0} \frac{x}{\mathbf{e}^x - 1} = \frac{0}{\mathbf{e}^0 - 1} = \frac{0}{1 - 1} = \frac{0}{0}$$

Solução:

$$\lim_{x \to 0} \frac{x}{\mathbf{e}^x - 1} = \lim_{x \to 0} \left[\frac{\dfrac{d}{dx} x}{\dfrac{d}{dx} (\mathbf{e}^x - 1)} \right] = \lim_{x \to 0} \frac{1}{\mathbf{e}^x} = \frac{1}{1} = 1$$

Exemplo 4: Encontre $\lim\limits_{x \to \infty} \dfrac{x}{\mathbf{e}^x}$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\dfrac{\infty}{\infty}$, pois:

$$\lim_{x \to \infty} \frac{x}{\mathbf{e}^x} = \frac{\infty}{\mathbf{e}^\infty} = \frac{\infty}{\infty}$$

Solução:

$$\lim_{x \to \infty} \frac{x}{\mathbf{e}^x} = \lim_{x \to \infty} \left[\frac{\dfrac{d}{dx} x}{\dfrac{d}{dx} (\mathbf{e}^x)} \right] = \lim_{x \to \infty} \frac{1}{\mathbf{e}^x} = \frac{1}{\infty} = 0$$

Exemplo 5: Encontre $\lim\limits_{x \to \infty} \dfrac{\mathbf{e}^x + 1}{x^3 + 1}$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\dfrac{\infty}{\infty}$, pois:

$$\lim_{x \to \infty} \frac{\mathbf{e}^x + 1}{x^3 + 1} = \frac{\infty + 1}{\infty^3 + 1} = \frac{\infty}{\infty}$$

Solução:

$$\lim_{x\to\infty} \frac{\mathbf{e}^x + 1}{x^3 + 1} = \lim_{x\to\infty} \left[\frac{\frac{d}{dx}(\mathbf{e}^x + 1)}{\frac{d}{dx}(x^3 + 1)} \right] = \lim_{x\to\infty} \frac{\mathbf{e}^x}{3\,x^2} = \frac{\mathbf{e}^\infty}{3\,\infty^2} = \frac{\infty}{\infty} \ .$$

Observamos que a indeterminação ainda continua. Assim sendo, devemos aplicar sucessivas vezes a Regra de L'Hospital, até retirarmos a indeterminação.

$$\lim_{x\to\infty} \frac{\mathbf{e}^x}{3\,x^2} = \lim_{x\to\infty} \frac{\frac{d}{dx}(\mathbf{e}^x)}{\frac{d}{dx}(3\,x^2)} = \lim_{x\to\infty} \frac{\mathbf{e}^x}{6\,x}$$

$$\lim_{x\to\infty} \frac{\mathbf{e}^x}{6\,x} = \lim_{x\to\infty} \frac{\frac{d}{dx}(\mathbf{e}^x)}{\frac{d}{dx}(6\,x)} = \lim_{x\to\infty} \frac{\mathbf{e}^x}{6} = \frac{\mathbf{e}^\infty}{6} = \infty \ .$$

Indeterminações dos tipos: $0 \,.\, \infty \ , \infty - \infty$

Estas formas de indeterminações podem ser removidas efetuando-se operações algébricas, de tal forma que possamos convertê-las nas indeterminações dos tipos $\frac{0}{0}$ ou $\frac{\infty}{\infty}$.

Exemplo 6: Encontre $\lim_{x\to\infty} \left[x \operatorname{sen}\left(\frac{1}{x}\right) \right]$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\infty \,.\, 0$, pois:

$$\lim_{x\to\infty} x \operatorname{sen}\left(\frac{1}{x}\right) = \infty \,.\, \left(\operatorname{sen}\left(\frac{1}{\infty}\right) \right) = \infty \,.\, 0$$

Solução:

Fazendo-se uma mudança de variável, $z = \frac{1}{x}$. Como $x \to \infty$, então $z \to 0$.

$$\lim_{x\to\infty} x \operatorname{sen}\left(\frac{1}{x}\right) = \lim_{z\to 0} \frac{\operatorname{sen}(z)}{z}$$

Neste caso, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\frac{0}{0}$.

Logo, devemos aplicar a Regra de L'Hospital.

$$\lim_{z \to 0} \frac{\operatorname{sen}(z)}{z} = \lim_{z \to 0} \frac{\frac{d}{dz}\operatorname{sen}(z)}{\frac{d}{dz} z} = \lim_{z \to 0} \frac{\cos(z)}{1} = \frac{1}{1} = 1$$

Exemplo 7: Encontre $\lim_{x \to 0} \left[\frac{1}{x} - \frac{1}{\operatorname{sen}(x)} \right]$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo $\infty - \infty$, pois:

$$\lim_{x \to 0} \left[\frac{1}{x} - \frac{1}{\operatorname{sen}(x)} \right] = \frac{1}{0} - \frac{1}{\operatorname{sen}(0)} = \infty - \infty$$

Solução:

Realizando-se operação algébrica, leva-nos a uma indeterminação do tipo $\frac{\infty}{\infty}$.

$$\lim_{x \to 0} \frac{\operatorname{sen}(x) - x}{x \operatorname{sen}(x)} = \lim_{z \to 0} \frac{0 - 0}{0 \,.\, 0} = \frac{0}{0}$$

Logo, devemos aplicar a Regra de L'Hospital.

$$\lim_{x \to 0} \frac{\operatorname{sen}(x) - x}{x \operatorname{sen}(x)} = \lim_{x \to 0} \frac{\frac{d}{dx}(\operatorname{sen}(x) - x)}{\frac{d}{dx}(x \operatorname{sen}(x))} = \lim_{x \to 0} \frac{\cos(x) - 1}{\operatorname{sen}(x) + x\cos(x)}$$

$$\lim_{x \to 0} \frac{\cos(x) - 1}{\operatorname{sen}(x) + x\cos(x)} = \lim_{x \to 0} \frac{\frac{d}{dx}(\cos(x) - 1)}{\frac{d}{dx}(\operatorname{sen}(x) + x\cos(x))} = \lim_{x \to 0} \frac{-\operatorname{sen}(x)}{\cos(x) + \cos(x) + x\operatorname{sen}(x)}$$

$$\lim_{x \to 0} \frac{-\operatorname{sen}(x)}{\cos(x) + \cos(x) + x\operatorname{sen}(x)} = \frac{0}{1 + 1 + (0 \,.\, 0)} = \frac{0}{2} = 0$$

Indeterminações do tipo: 0^0 , ∞^0 e 1^∞

Estes limites são originados a partir dos limites da forma $\lim_{x \to c} [\,f(x)^{g(x)}\,]$.

Nestes casos, geralmente, aplicamos logarítmos e em seguida aplicamos as regras de indeterminações vistas anteriormente.

Exemplo 8: Encontre $\lim_{x \to 0} \left[(1+x)^{\left(\frac{1}{x}\right)} \right]$, usando a Regra de L'Hospital.

Neste exemplo, se aplicarmos o limite diretamente, obtemos uma indeterminação do tipo 1^{∞}, pois:

$$\lim_{x \to 0} \left[(1+x)^{\left(\frac{1}{x}\right)} \right] = (1+0)^{\left(\frac{1}{0}\right)} = 1^{\infty}$$

Sabemos que este limite leva-nos ao número de Euler, ou seja: $\lim_{x \to 0} \left[(1+x)^{\left(\frac{1}{x}\right)} \right] = \mathbf{e}$.

Chamando $y = (1+x)^{\left(\frac{1}{x}\right)}$.

Aplicando-se logaritmos a expressão acima, obtemos: $\ln(y) = \ln\left((1+x)^{\left(\frac{1}{x}\right)} \right)$.

Assim, vamos aplicar o limite à expressão acima e em seguida as regras discutidas anteriormente.

$$\lim_{x \to 0} \ln(y) = \lim_{x \to 0} \ln\left((1+x)^{\left(\frac{1}{x}\right)} \right) = \lim_{x \to 0} \left[\frac{1}{x} \right] [\ln(1+x)]$$

$$\lim_{x \to 0} \ln(y) = \lim_{x \to 0} \left[\frac{1}{x} \right] [\ln(1+x)] = \lim_{x \to 0} \frac{\ln(1+x)}{x} = \frac{0}{0}$$

Aplicando-se Regra de L'Hospital, fornece-nos:

$$\lim_{x \to 0} \ln(y) = \lim_{x \to 0} \frac{\ln(1+x)}{x} = \lim_{x \to 0} \frac{\frac{d}{dx}\ln(1+x)}{\frac{d}{dx}x} = \lim_{x \to 0} \frac{1}{(1+x)\,1}$$

$$\lim_{x \to 0} \ln(y) = \lim_{x \to 0} \frac{1}{1+x} = 1$$

Considerando que podemos escrever $y = \mathbf{e}^{\ln(y)}$, desta forma:

$$\lim_{x \to 0} \left[(1+x)^{\left(\frac{1}{x}\right)} \right] = \lim_{x \to 0} [y] = \lim_{x \to 0} \mathbf{e}^{\ln(y)} = \lim_{x \to 0} [\ln(y)\,\mathbf{e}] = [\lim_{x \to 0} \ln(y)] [\lim_{x \to 0} \mathbf{e}]$$

$$\lim_{x \to 0} \left[(1+x)^{\left(\frac{1}{x}\right)} \right] = [\lim_{x \to 0} \ln(y)]\,[\lim_{x \to 0} \mathbf{e}] = 1 \,\mathbf{.}\, \mathbf{e} = \mathbf{e}$$

Exercícios propostos:

I - Calcule os limites sem usar a Regra de L'Hospital, e depois use-a para verificar os resultados:

a) $\lim_{x \to 2} \frac{x^2 - 4}{x^3 + 2x - 8}$

b) $\lim_{x \to 1} \frac{x^2 - 1}{x^3 - 1}$

c) $\lim_{x \to 0} \frac{\text{sen}(x)}{\text{tg}(x)}$

II - Usando a Regra de L'Hospital determine os limites:

a) $\lim_{x \to 0} \frac{\mathbf{e}^x - 1}{\text{sen}(x)}$

b) $\lim_{x \to 0} \frac{\text{tg}(x)}{x}$

c) $\lim_{x \to \infty} \frac{\ln(x)}{x}$

d) $\lim_{x \to \infty} \frac{\mathbf{e}^{(2x)}}{x^2}$

e) $\lim_{x \to \infty} x\,\mathbf{e}^{(-x)}$

f) $\lim_{x \to 0+} \frac{1 - \ln(x)}{\mathbf{e}^{\left(\frac{1}{x}\right)}}$

g) $\lim_{x \to \infty} \left(1 - \frac{3}{x}\right)^x$

III - Usando o **Maple**, defina a função e trace o seu gráfico. Estime os limites das funções abaixo. Verifique a sua solução usando a Regra de L'Hospital.

a) $\lim_{x \to \infty} \frac{\ln(\ln(x))}{\sqrt{x}}$

b) $\lim_{x \to 0+} x^x$

c) $\lim_{x \to 0+} \operatorname{sen}(x)^{\left(\frac{3}{\ln(x)}\right)}$

Tabela Geral de Derivadas

Apresentamos uma tabela geral de derivadas.
Nesta tabela, u e v são funções deriváveis em x, e a, b e c são constantes reais.

01 - $y = c$, então $y' = 0$

02 - $y = x$, então $y' = 1$

03 - $y = x^b$, então $y' = b\, x^{(b-1)}$

04 - $y = c\, u$, então $y' = c \,.\, u'$

05 - $y = u + v$, então $y' = u' + v'$

06 - $y = u\, v$, então $y' = u \,.\, v' + u' \,.\, v$

07 - $y = \frac{u}{v}$, então $y' = (u' \,.\, v - u \,.\, v') / v^2$

08 - $y = u^b$ (com $b \neq 0$), então $y' = b\, u^{(b-1)} \,.\, u'$;

09 - $y = \mathbf{e}^x$, então $y' = \mathbf{e}^x$

10 - $y = \mathbf{e}^u$, então $y' = \mathbf{e}^u \,.\, u'$

11 - $y = a^x$ (com $0 < a$ e $a \neq 1$) , então $y' = \ln(a)\, a^x$

12 - $y = a^u$ (com $0 < a$ e $a \neq 1$) , então $y' = \ln(a)\, a^u \,.\, u'$

13 - $y = \ln(x)$, então $y' = \frac{1}{x}$

14 - $y = \log_a(x)$ (com $0 < a$ e $a \neq 1$) , então $y' = \left[\dfrac{1}{\ln(a)}\right]\left[\dfrac{1}{x}\right]$

15 - $y = \log_a(u)$ (com $0 < a$ e $a \neq 1$) , então $y' = \left[\dfrac{1}{\ln(a)}\right]\left[\dfrac{1}{u}\right] . u'$

16 - $y = \ln(u)$, então $y' = \dfrac{1}{u} . u'$

17 - $y = u^v$ (com $0 < u$), então $y' = v . u^{(v-1)} . u' + u^v . \ln(u) . v'$

18 - $y = \text{sen}(x)$, então $y' = \cos(x)$

19 - $y = \text{sen}(u)$, então $y' = \cos(u) . u'$

20 - $y = \cos(x)$, então $y' = -\text{sen}(x)$

21 - $y = \text{sen}(u)$, então $y' = -\text{sen}(u) . u'$

22 - $y = \text{tg}(x)$, então $y' = [\sec(x)]^2$

23 - $y = \text{tg}(u)$, então $y' = [\sec(u)]^2 . u'$

24 - $y = \text{arcsen}(x)$, então $y' = \dfrac{1}{\sqrt{1 - x^2}}$

25 - $y = \text{arcsen}(u)$, então $y' = \dfrac{1}{\sqrt{1 - u^2}} . u'$

26 - $y = \arccos(x)$, então $y' = \dfrac{-1}{\sqrt{1 - x^2}}$

27 - $y = \arccos(u)$, então $y' = \dfrac{-1}{\sqrt{1 - u^2}} . u'$

28 - $y = \text{arctg}(x)$, então $y' = \dfrac{1}{1 + x^2}$

29 - $y = \arctan(u)$, então $y' = \dfrac{1}{1 + u^2} . u'$